

HPBooks

Arcade, Board & Educational Games in BASIC!

35
AMAZING
GAMES

For Your
Commodore
64

John Mihalik

35
Amazing Games
For Your
COMMODORE 64
John Mihalik

Executive Editor: Rick Bailey
Editorial Director: Theodore DiSante
Art Director: Don Burton
Book Design: Leslie Sinclair
Typography: Cindy Coatsworth, Michelle Claridge
Illustrations: Thomas A. Rozum
Programming Consultant: Ron Paludan

NOTICE: The information contained in this book is true and complete to the best of our knowledge. All recommendations are made without any guarantees on the part of the author or HPBooks. The author and HPBooks disclaim all liability in connection with the use of this information. Commodore 64 is a registered trademark of Commodore Business Machines, Inc.

Published by HPBooks®, P.O. Box 5367, Tucson, AZ 85703 602/888-2150
ISBN: 0-89586-316-2 Library of Congress Catalog No. 84-80013
©1984 Fisher Publishing, Inc. Printed in USA

35
Amazing Games
For Your
COMMODORE 64

ARCADE GAMES

Shoot The Rapids	8
<i>Stay on course in your kayak while traveling down a dangerous set of rapids.</i>	
Night Sky	10
<i>Invaders from outer space, coming fast and straight at you! Stop them before they crash into you.</i>	
Space Maneuvers	13
<i>Fly your spacecraft through crowded space traffic. Avoid collisions to score.</i>	
Missile Defense	15
<i>Defend your cities against missile attack by intercepting oncoming missiles.</i>	
Wallbangers	18
<i>Trap your opponent (the computer) by building walls on the screen that limit the computer's cursor motion.</i>	
Rear Gunner	20
<i>You're defending the rear of a space fighter. Be careful not to shoot other members of your squadron.</i>	
Le Mans	22
<i>Drive your race car through a twisting track. Avoid other cars, oil spills and track hazards.</i>	
Moon Lander	25
<i>Land your spaceship on the moon's surface by controlling thrust, speed and other factors.</i>	
Gambler	28
<i>Play the odds with the computer's one-armed bandit.</i>	
Orion	31
<i>You are at the controls of a starship simulator, practicing defense of your homeland against imperial drones.</i>	
Berserkotron	34
<i>You've gone to interview for a job as a computer-game tester but entered the Robot Room where visitors are not allowed. Run away if you can.</i>	
BOARD GAMES	
Shell Game	37
<i>Computer version of the classic sleight-of-hand game.</i>	
Pirates	40
<i>Find and battle the scourge of the sea. You're the captain looking for adventure, glory and stolen gold.</i>	
Master Code	44
<i>Computer selects a code. You crack it.</i>	
A Day At The Races	46
<i>Bet on a dog race. Watch the race and find out if you're the winner.</i>	
Lost Dutchman Mine	50
<i>You're looking for the mine's treasure. Find clues while escaping from ghosts.</i>	

Biorhythm	54
<i>Computer calculates and predicts your emotional, mental and physical levels for any date.</i>	
Queen Of The Nile	57
<i>Search for Cleopatra's lost treasure, but be careful of snake pits and mummies.</i>	
Thin Ice	60
<i>Go ice fishing late in the season, but watch out for what lurks under the ice.</i>	
Bomb Squad	62
<i>Locate and diffuse a bomb hidden in a huge skyscraper.</i>	
Utopia	66
<i>You're an island ruler beset with decisions. Will they help or hinder your reign?</i>	
Tic-Tac-Toe	71
<i>For two players.</i>	
Dice Racers	74
<i>Use computer dice to race your cars across the finish line.</i>	
Alien	77
<i>Your chance to visit Earth. Will you come as friend or foe?</i>	
EDUCATIONAL GAMES	
Hangman	85
<i>Home-computer version of the popular word game.</i>	
Dr. Sigmund	87
<i>Visit the famous analyst for ink-blot testing. Get professional advice.</i>	
Spatial Concepts	91
<i>A castle where objects can be placed in a variety of positions. For young children.</i>	
Mind Reader	95
<i>Can you guess the number the computer has in mind? Can the computer guess the one you have in mind?</i>	
Poetry	99
<i>Wouldn't you know it? Your computer is a poet!</i>	
No Free Lunch	101
<i>Cross a hazardous tide pool to reach your lunch by giving correct math answers.</i>	
Spelling Bee	104
<i>Watch the bumble bee spell out a scrambled word. Give the correct spelling to win.</i>	
Speed Reading	106
<i>Learn computerized reading skills for rapid information intake. Gain speed as your skill level improves.</i>	
Pre-School Counter	110
<i>Computer displays 1 to 10 symbols and asks "How many?" Child receives a happy or sad face for correct or incorrect answers.</i>	
Keyboard Shooting Gallery	112
<i>Try your hand at the computer shooting gallery. Learn the keyboard at the same time.</i>	
Translator	114
<i>Computer has foreign words and English equivalents for user's study of foreign language.</i>	

Introduction

This book stresses one thing—computers are fun! Even though you may use your computer for serious applications, you should not ignore its ability to give you hours of challenging entertainment. Computer games add that dimension to home-computing.

ABOUT THE 35 GAMES IN THIS BOOK

Some game books give you short programs and tell *you* to add scoring, graphics and other interesting things. This book is different. It consists of 35 *complete* games. They have color, graphics, sound, instructions to players, and scoring.

In addition to the program listing, each game has *suggestions* for possible programming variations. First enter the complete, working game. Try it as is. Then change it if you want to.

This book doesn't explain the rudiments of BASIC programming because I assume that you already have reference books to do that. What the book does supply is a variety of fun game programs that work.

I've divided the games into three categories: Arcade, Board and Educational. The first category has moving graphics, like video games that cost money to play. The second includes puzzle and logic games you might play on a board, such as Tic-Tac-Toe. The third category, educational games, includes word and number fun for children and older computer users.

How To Use—Each game begins with a description of game rules, number of players, display, objective, scoring and keys to press. Don't skip this section. Read it carefully.

Then enter the program listing into your computer *exactly* as shown. The listing is a reproduction of actual printout of a tested, debugged, working program. When you make typing mistakes—and you will—the game may not work as intended. Or, it may not work at all. You'll have to find your mistakes and fix them.

Play the game for a while. If you like it, save it on disk or tape. Otherwise, you'll have to enter the game each time you want to play it.

How To Change It—After you've become familiar with a game, consider changing or improving it. Use the comments to the side of the computer printout as suggestions on how you can simplify the game, make it more complicated, speed it up, slow it down or supply different data. Only *some* of the possibilities are shown. Have fun experimenting!

Although this isn't a book about programming, changing the games will help you learn more about BASIC programming. If you want to learn still more, get *How to Program Your Commodore 64, BASIC for Beginners* by Carl Shipman, also published by HPBooks. BASIC is a good programming language because it's easy to understand. Commands are in plain English, and rules are straightforward and uncomplicated.

Even if you don't know BASIC, you'll have good times with these games. If you already know some BASIC, all the better. This book can be a springboard to your own creative programming.

WHAT YOU NEED TO USE THIS BOOK

The games in this book are written specifically for the Commodore 64 computer. None of these games requires a lot of memory. Plenty of memory is left over for you to add your own features to any program in this book.


Other Items—You also need a b&w or color monitor. I recommend that you use a disk drive or cassette tape recorder to save the games. That way, you don't have to type them in each time you want to use them.

Some arcade games should be played with a joystick. It's an inexpensive accessory that makes these games more enjoyable.

What Else? —I assume that you know how to use your computer hardware and that you're familiar with your computer keyboard. I also assume that you're eager to have fun with your computer and to learn something about BASIC as painlessly as possible.

If you think that typing an amazing game into your computer isn't painless enough, there's an easier way. Simply send a check or money order for \$30 and your name and address to the address below. You'll receive a floppy disk of all 35 games in this book.

MIKO WARE
PO BOX 43911
Tucson, AZ 85733



For more information about programming your Commodore 64 computer in BASIC, get this book.

UNDERSTANDING PROGRAM PRINTOUTS

The program lines in this book are actual computer printouts of the working games. To make the lines easier for you to copy, a special notation system is used for graphics characters, color and cursor keys.

An underlined, upper-case letter means that you should hold down the SHIFT key while entering that letter. For example, if you see Z, shift while entering Z. A diamond will appear on the screen. An underlined, lower-case letter indicates that you should hold down the Commodore key (⌘) while entering that letter.

Some keys (+, -, £, @, *) will be underlined twice if the Commodore key (⌘) is to be held down; underlined once if the SHIFT key is to be held down.

When you see a word in brackets, it indicates using special control keys. For example, [DOWN] means press the cursor-up/down key. [DOWN 7] means press the cursor-up/down key seven times. The only exception is with spaces. A single space looks that way in the program line. If two or more spaces are required, an expression will be in brackets, such as [SPACE 2].

The table on the next page lists some more key combinations. For example, if the program shows [WHT] press the CTRL and WHT keys at the same time.

KEY-COMBINATION TABLE

Listing	Press Key(s)
[UP]	SHIFT-CURSOR UP/DOWN
[DOWN]	CURSOR UP/DOWN
[LEFT]	SHIFT-CURSOR LEFT/RIGHT
[RIGHT]	CURSOR LEFT/RIGHT
[HOME]	CLR HOME
[CLR]	SHIFT-CLR HOME
[RVSON]	CTRL-RVS ON
[RVSOFF]	CTRL-RVS OFF
[BLK]	CTRL-BLK
[WHT]	CTRL-WHT
[RED]	CTRL-RED
[CYN]	CTRL-CYN
[PUR]	CTRL-PUR
[GRN]	CTRL-GRN
[BLU]	CTRL-BLU
[YEL]	CTRL-YEL
[ORN]	⌘ 1
[BRN]	⌘ 2
[LT. RED]	⌘ 3
[DARK GRAY]	⌘ 4
[MED. GRAY]	⌘ 5
[LT. GRN]	⌘ 6
[LT. BLU]	⌘ 7
[LT. GRAY]	⌘ 8
[F-1]	f1
[F-2]	f2
[F-3]	f3
[F-4]	f4
[F-5]	f5
[F-6]	f6
[F-7]	f7
[F-8]	f8
[POUND]	£

A duplicate of this table is in the back of this book, page 117. Cut it out and glue it to a piece of stiff cardboard. Use it for a ready reference when entering program lines.

Shoot The Rapids

Journey down a winding river in your kayak. The farther you go, the more treacherous the river. This fast-paced game requires concentration and skill. Use a joystick or < and > keys to paddle safely between river banks.

If you crash, your score is displayed along with commentary. Press any key to play again or RUN/STOP to quit.

Alteration—To make the river wind more, increase the value of variable **A** in line 120 (maximum value is 0.5) and decrease variable **B** (minimum is 0.5).

Programming Note—Line 220 is a special trick that makes the screen display scroll backward.


```

100 REM SHOOT THE RAPIDS
110 HS=0: GOTO 420
120 R$="[BLU][RVSON][SPACE 7]":R=18
   :D=0:S=1024:CO=55296:L=1924:A=.4
   :B=.7
130 PRINT "[CLR]": POKE 53281,13
140 FOR X=1 TO 23: PRINT TAB( R)R$
   : NEXT
150 SO=54272: FOR Z=SO TO SO+24
   : POKE Z,0: NEXT Z: POKE SO+5,31
   : POKE SO+6,31: POKE SO+24,15
160 POKE SO+1,65: POKE SO,227
   : POKE SO+4,129:K= RND (4)
   : IF R<B THEN K=K+.1
170 IF K<A AND R>1 THEN R=R-1
180 IF R>30 THEN K=K-.1
190 IF K>B AND R<38- LEN (R$) THEN
   R=R+1
200 PRINT "[HOME][DOWN]" TAB( INT (
   RND (8)*39)) "[GRN]*"
210 PRINT "[HOME][DOWN]" TAB( R)R$
220 PRINT "[HOME][DOWN][LEFT][INST]"
   : POKE 218,156
230 POKE L+40,160
240 JY= PEEK (56320):JY=15-(JY AND 15)
   : IF JY=0 THEN JY= PEEK (56321)
   : JY=15-(JY AND 15)
250 IF PEEK (197)=47 OR (JY<7 AND
   JY>3) THEN L=L-1
260 IF PEEK (197)=44 OR (JY>7 AND
   JY<11) THEN L=L+1
270 IF PEEK (L)<>160 THEN 320
280 POKE L,193:D=D+.5: IF D>HS THEN
   HS=D
290 PRINT "[HOME][BLK]SCORE
   : " STR$ ( INT (D))+" HIGH
   : " STR$ ( INT (HS))+" "
300 IF D/50= INT (D/50) AND D<201
   THEN R$= LEFT$ (R$, LEN (R$)-1)
310 POKE SO+4,128: GOTO 160
320 POKE L,90: FOR T=1 TO 2000
   : NEXT T
330 PRINT "[HOME][BLK]SCORE
   : " STR$ ( INT (D))+" "
340 IF D>HS THEN HS=D

```

Alteration—Change the commentary in lines 350 to 380.

```
350 IF D<=50 THEN PRINT "YOUR PADDLE
      MUST HAVE BROKEN"
360 IF D<=200 AND D>50 THEN PRINT
      "SNAGGED A TREE LIMB"
370 IF D<=300 AND D>200 THEN PRINT
      "YOU MUST BE PART OTTER"
380 IF D>300 THEN PRINT "I SEE YOU
      HAVE DONE[SPACE 3]THIS BEFORE"
390 PRINT : PRINT "HIT ANY KEY TO
      RESTART": POKE 198,0
400 GET Q$: IF Q$="" THEN 400
410 GOTO 120
420 PRINT "[CLR]": PRINT "SHOOT
      THE RAPIDS!!!": PRINT
430 PRINT : PRINT "YOU ARE ABOUT
      TO SHOOT THE RAPIDS IN[SPACE 4]
      YOUR KAYAK."
440 PRINT : PRINT "CONTROL YOUR
      PADDLES WITH YOUR JOYSTICK[SPACE 2]
      OR THE < > KEYS."
450 PRINT : PRINT "KEEP BETWEEN
      THE RIVER BANKS.": PRINT
      : PRINT "HIT ANY KEY TO START."
460 POKE 198,0
470 GET Q$: IF Q$="" THEN 470
480 GOTO 120
```



Night Sky

It's time to defend against invaders from space. They're coming straight at you, and getting bigger! Distance is determined by the size of the incoming invader. If you don't stop an invader, it crashes into you.

You move your aim sight by using a joystick. The cursor keys will work too. Fire by pressing the joystick's fire button or the keyboard's space bar.


Alteration—Change the starting range by altering **SR=2000** in line 140.

Alteration—Change the range at which the invader becomes larger by changing **RA < 1500** in line 170 and **RA < 1000** in line 180.

```

100 REM NIGHT SKY
110 DIM D(15): GOSUB 510
120 PRINT "[CLR]"
130 FOR X=55296 TO 56296: POKE X,1
    : NEXT X
140 S=0:SR=2000
150 RA=SR:R=1
155 FOR X=55296 TO 56295: POKE X,1:
    NEXT X
160 EL=1144+ INT ( RND (8)*800)
    :AL=1524:AC=43
170 FOR X=0 TO 4: POKE EL+X,32: NEXT X
    :RA=RA-10: IF RA<1500 THEN R=2
180 IF RA<1000 THEN R=3
190 D= INT ( RND (8)*16):K=EL+D(D)
    : IF K>1144 AND K<1944 THEN EL=K

```



```
200 IF RA<10 THEN 390
210 POKE AL,32: FOR X=0 TO 4
   : POKE EL+X,EC(R,X): NEXT X
220 K= PEEK (197):J= PEEK (653)
230 PRINT "[HOME][WHITE][SPACE 4]
   SCORE=" STR$ (S)+"[SPACE 11]
   RANGE=" STR$ (RA)+"[SPACE 2]"
240 JY= PEEK (56321):FI=JY AND 16
   :JY=15-(JY AND 15)
250 POKE AL,32: IF AL>1104 AND (JY=1
   OR JY=5 OR JY=9 OR (K=7 AND J=1))
   THEN AL=AL-40
260 IF AL<1945 AND (JY=2 OR JY=6 OR
   JY=10 OR (K=7 AND J=0)) THEN
   AL=AL+40
270 IF AL>1104 AND ((JY>3 AND JY<7)
   OR (K=2 AND J=1)) THEN AL=AL-1
280 IF AL<1945 AND (JY>7 OR (K=2 AND
   J=0)) THEN AL=AL+1
290 IF FI>0 THEN 320
300 POKE SO+5,1: POKE SO+6,9
   : POKE SO+24,15
310 POKE SO+1,5: POKE SO,235
   : POKE SO+4,129: POKE SO+4,128
320 IF PEEK (AL)<>32 AND (FI=0 OR
   K=60) THEN 340
330 POKE AL,AC: GOTO 170
340 POKE N,7:S=S+RA: IF SR>200 THEN
   SR=SR-100
350 FOR LO=SO TO SO+24: POKE LO,0
   : NEXT
360 POKE SO+5,1: POKE SO+6,15
   : POKE SO+24,15
370 POKE SO+1,5: POKE SO,235
   : POKE SO+4,129: POKE SO+4,128
380 FOR T=1 TO 3000: NEXT T: POKE N,0
   : PRINT "[CLR]": GOTO 150
390 POKE N-1,2: POKE N,7
   : PRINT "[CLR][WHITE][DOWN 10]
   [SPACE 13]***CRASH***"
400 FOR LO=SO TO SO+24: POKE LO,0
   : NEXT
410 POKE SO+5,15: POKE SO+6,15
   : POKE SO+24,15
420 POKE SO+1,1: POKE SO,235
   : POKE SO+4,129: POKE SO+4,128
430 FOR X=1 TO 99: IF ( PEEK (N) AND
   15)=7 THEN POKE N,2: GOTO 450
440 POKE N,7
450 FOR T=1 TO 44: NEXT T,X
460 FOR T=1 TO 2000: NEXT : POKE 198,0
470 POKE N,0: POKE N-1,6
480 PRINT : INPUT "PLAY AGAIN";Q$
```

MORE ->

```
490 IF LEFT$(Q$,1)="Y" THEN 120
500 END
510 SO=54272:N=53281: POKE N,0
   : PRINT "[CLR][WHITE]"
520 PRINT "YOU ARE ABOUT TO PROTECT
   THE NIGHT SKY."
530 PRINT "INTERCEPT THE ENEMY AS
   IT APPROACHES."
540 PRINT : PRINT "MOVE YOUR AIM
   SIGHT WITH A JOYSTICK"
550 PRINT "OR THE CURSOR CONTROL
   KEYS."
560 PRINT "PRESS THE FIRE BUTTON
   OR SPACE BAR TO[SPACE 3]FIRE."
   : PRINT
570 PRINT "THE GAME ENDS WHEN THE
   ENEMY CRASHES[SPACE 4]INTO YOU."
   : PRINT
580 PRINT "GOOD LUCK."
590 FOR X=0 TO 7: READ D(X): NEXT X
600 FOR X=1 TO 3: FOR Y=0 TO 4
   : READ EC(X,Y): NEXT Y,X
610 PRINT : PRINT "HIT ANY KEY TO
   START."
620 GET Q$: IF Q$="" THEN 620
630 RETURN
640 DATA -41,-40,-39,-1,1,39,40,41
650 DATA 32,109,125,32,32,32,109,42,
   125,32,124,67,81,67,126
```



Space Maneuvers

You're the pilot of a spaceship on a secret mission, traveling through crowded space traffic. To score, avoid collisions with asteroids, mercenary drones and dangerous enemy spacecraft. The deeper you penetrate space, the more difficult your task! Use your joystick or < and > keys to maneuver left and right. Saving the galaxy requires 1000 points.


```

100 REM SPACE MANEUVERS
110 GOTO 350
120 DL=.1:P=0:S=1024:L=S+900
130 PRINT "[CLR][WHITE]"
    : POKE 53280,6: POKE 53281,0
140 PRINT "[HOME][DOWN]" TAB( INT (
    RND (7)*36))"W[SPACE 3]W[SPACE 3]
    W"
150 IF P>100 AND RND (3)>.8 THEN
    PRINT "[HOME][DOWN]" TAB( INT (
    RND (7)*39))"*[POUND]"

```

Alteration—Increase the number of drones by changing **RND(3)>.8** to **RND(3)>.4** in line 150.

MORE →



Alteration—Increase the number of enemy spacecraft by changing **RND(4)>.9** to **RND(4)>.6** in line 160.

Alteration—Change symbols for drones and enemy spacecraft in lines 150 and 160, respectively.

```
160 IF P>400 AND RND (4)>.9 THEN
    PRINT "[HOME][DOWN]" TAB( INT (
        RND (7)*38))"*Z[POUND]"
170 PRINT "[HOME][DOWN][LEFT][INST]"
    : POKE 218,156
180 P=P+1: IF P>999 THEN 300
190 PRINT "[HOME]SCORE
    : " STR$ ( INT (P))+" "
200 JY= PEEK (56320):JY=15-(JY AND 15)
    : IF JY=0 THEN JY= PEEK (56321)
    :JY=15-(JY AND 15)
210 POKE L+40,32: IF PEEK (197)=47
    OR (JY<7 AND JY>3) THEN L=L-1
220 IF L<1983 AND ( PEEK (197)=44 OR
    (JY>7 AND JY<11)) THEN L=L+1
230 IF PEEK (L)<>32 THEN 250
240 POKE L,65: POKE L+54272,1:
    GOTO 140
250 E=54272: FOR X=E TO E+24
    : POKE X,0: NEXT X: POKE E+5,15
    : POKE E+6,15: POKE E+24,15
260 POKE 53280,2: POKE E+1,5
    : POKE E,235: POKE E+4,129
    : FOR V=1 TO 250
270 IF F=0 THEN PRINT "[CLR][DOWN 7]
    [SPACE 11]***** CRASH *****"
    : POKE 53281,7:F=1: GOTO 290
280 PRINT "[CLR]": POKE 53281,2:F=0
290 NEXT V: POKE 53280,6
    : POKE 53281,0: POKE S+4,128
    : GOTO 310
300 PRINT "[CLR][DOWN 10] THE GALAXY
    IS SAVED!"
310 PRINT "[CLR]": PRINT "[SPACE 8]
    HIT ANY KEY TO RESTART"
320 PRINT : PRINT "[SPACE 13]SCORE
    = "P: POKE 198,0
330 GET Q$: IF Q$="" THEN 330
340 GOTO 120
350 PRINT "[CLR]": PRINT "PREPARE
    FOR A JOURNEY THROUGH SPACE."
360 PRINT "AVOID ASTEROIDS, DRONES,
    AND SPACESHIPS.USE JOYSTICK
    OR < > ";
370 PRINT "KEYS TO MOVE LEFT
    ORRIGHT.": PRINT
380 PRINT : PRINT "HIT ANY KEY TO
    RESTART": POKE 198,0
390 GET Q$: IF Q$="" THEN 390
400 GOTO 120
```

Alteration—If the galaxy is saved, add triumphant music in lines 300 to 309.

Missile Defense

You must save your cities by intercepting oncoming enemy missiles. Use your joystick to move your laser sight (+). The cursor keys will work too. Fire by hitting the joystick's fire button or the keyboard's space bar. Each shot costs you 1 point. Each hit gets you 100 points. Destroying a missile does not require a direct hit. A close shot may work.

```

100 REM  MISSILE DEFENSE
110 GOSUB 520:B$(3)="[YEL]vffb"
    :B$(2)="[DOWN][RED][RVSON]b
    [RVSOFF]cd[YEL]v"
    :B$(1)="[DOWN 2][RED]bcf"
    :B$(0)="[DOWN 3] [RVSON]v"
120 A$="[HOME][DOWN 19]"
130 FOR X=0 TO 8: READ EL(X),EC(X)
    : NEXT
140 FOR X=0 TO 6: READ CL(X),CH(X)
    : NEXT X
150 D=0:CT=4:P=80
160 PRINT "[CLR]"
170 POKE 53280,6: POKE 53281,14
    : FOR X=55296 TO 56175: POKE X,0
    : NEXT

```

Alteration—Increase the number of cities to five by changing the value of **CT** to 5 in line 150.

Alteration—Increase the points for destroying a missile to 150 by changing **D** in line 410 to **D=D+150**.

MORE →



```
180 FOR X=56176 TO 56295: POKE X,5
: NEXT : FOR X=1904 TO 2023
: POKE X,160: NEXT
190 CL=1864
200 C=CL+ INT ( RND (8)*30)
: IF P>0 THEN P=P-10
210 FOR X=0 TO 6: POKE C+CL(X),CH(X)
: NEXT X
220 ML=C-818:MC=81:MD=39
: IF RND (8)>.5 THEN MD=41:ML=ML-1
230 AL=C-70-60* RND (8)*7:AC=43
240 MI=ML:P=P-1
250 K= PEEK (197):J= PEEK (653)
: FOR T=0 TO P: NEXT T
260 PRINT "[HOME][WHITE][SPACE 4]
SCORE=" STR$ (D)+"[SPACE 11]
CITIES=" STR$ (CT)+"[SPACE 21]"
270 JY= PEEK (56321):FI=JY AND 16
:JY=15-(JY AND 15)
280 POKE AL,32: IF AL>1104 AND (JY=1
OR JY=5 OR JY=9 OR (K=7 AND J=1))
THEN AL=AL-80
290 IF AL<1745 AND (JY=2 OR JY=6 OR
JY=10 OR (K=7 AND J=0)) THEN
AL=AL+80
300 IF AL>1104 AND ((JY>3 AND JY<7)
OR (K=2 AND J=1)) THEN AL=AL-2
310 IF AL<1825 AND (JY>7 OR (K=2 AND
J=0)) THEN AL=AL+2
320 POKE ML,32:ML=ML+MD
: IF PEEK (ML)<>32 AND PEEK
(ML)<>AC THEN 440
330 POKE ML,MC: POKE AL,AC
: IF K=60 OR K=15 OR FI<>16 THEN
350
340 GOTO 250
350 FOR E=0 TO 8: POKE AL+EL(E),32
: NEXT E: IF D>0 THEN D=D-1
360 IF PEEK (ML)<>32 THEN 250
370 FOR E=0 TO 8: POKE ML+EL(E),EC(E)
: POKE ML+EL(E)+54272,7: NEXT
380 S=54272: FOR L=S TO S+24
: POKE L,0: NEXT : POKE S+5,15
: POKE S+6,15: POKE S+24,15
390 POKE S+1,5: POKE S,235
: POKE S+4,129: POKE S+4,128
: FOR T=0 TO 999: NEXT T
400 FOR E=0 TO 8: POKE ML+EL(E),32
: POKE ML+EL(E)+54272,0: NEXT
410 D=D+100
420 FOR X=0 TO 6: POKE C+CL(X),32
: NEXT X:V=0
430 GOTO 200
```

```
440 FOR X=0 TO 3: PRINT A$; TAB(
    C-CL)B$(X): NEXT
450 S=54272: FOR L=S TO S+24
    : POKE L,0: NEXT : POKE S+5,15
    : POKE S+6,15: POKE S+24,15
460 POKE S+1,5: POKE S,235
    : POKE S+4,129: POKE S+4,128
    : FOR T=0 TO 2000: NEXT T
470 CT=CT-1: IF CT>0 THEN 160
480 PRINT "[CLR][WHITE][DOWN 7]
    [SPACE 8]**ALL CITIES
    DESTROYED**": PRINT
    : PRINT "SCORE="D
490 END
500 DATA -41,85,-40,66,-39,73,-1,67,0,90,
    1,67,39,74,40,66,41,75
510 DATA 0,254,1,254,2,252,-40,108,-39,
    32,-38,97,-37,32
520 PRINT "[CLR][WHITE]": PRINT "SAVE
    YOUR CITY BY INTERCEPTING ON
    [SPACE 7]COMING MISSILES."
530 PRINT : PRINT "USE JOYSTICK
    OR CURSOR KEYS TO MOVE
    YOUR LASER SIGHT (+)"
540 PRINT : PRINT "PRESS FIRE BUTTON,
    RETURN KEY OR SPACE[SPACE 2]BAR
    TO FIRE."
550 PRINT : PRINT "HIT ANY KEY TO
    START"
560 GET Q$: IF Q$="" THEN 560
570 RETURN
```



Wallbangers

It's you against the computer as you both try to trap each other by building electrified walls. If you run into the computer's wall or the screen border, you get electrocuted, and vice-versa.


Use a joystick to move up, down, left or right. The cursor keys will work too, but the game as written is really too fast for them. Good luck!

```

100 REM WALLBANGERS
110 POKE 53281,0: PRINT "[CLR][WHITE]"
120 PRINT "YOU WILL TRY TO BUILD
    AN ELECTRIFIED"
130 PRINT "WALL AROUND ME. (*)"
140 PRINT "I WILL TRY TO DO THE SAME
    TO YOU."
150 PRINT "YOU CANNOT CROSS A WALL
    WITHOUT GETTING ZAPPED.": PRINT
160 PRINT "USE JOYSTICK OR CURSOR
    KEYS TO CHANGE[SPACE 4]
    DIRECTION."
170 D(0)=-40:D(1)=-1:D(2)=1:D(3)=40
180 PRINT : PRINT "HIT ANY KEY TO
    START."
190 GET Q$: IF Q$="" THEN 190
200 PRINT "[CLR]": FOR X=55296 TO
    56295: POKE X,1: NEXT X
210 FOR X=1024 TO 1063: POKE X,160
    : POKE X+960,160: NEXT X
220 FOR X=1064 TO 1944 STEP 40
    : POKE X,160: POKE X+39,160
    : NEXT X
230 L=1924:D=-40:EL=1124:ED=4:S=54272
    :N=100
240 FOR X=S TO S+24: POKE X,0: NEXT
250 POKE S+24,15
260 POKE S+1,19: POKE S+4,29
    : POKE S+5,19: POKE S+15,18
    : POKE S+24,15
270 K= PEEK (197):J= PEEK (653)
280 JY= PEEK (56321):FI=JY AND 16
    :JY=15-(JY AND 15)
290 IF JY=1 OR (K=7 AND J=1) THEN
    D=-40
300 IF JY=2 OR (K=7 AND J=0) THEN D=40
310 IF JY=4 OR (K=2 AND J=1) THEN D=-1
320 IF JY=8 OR (K=2 AND J=0) THEN D=1
330 IF PEEK (L+D)<>32 THEN 450
340 L=L+D: POKE L,42
350 IF RND (2)>.95 THEN ED= INT ( RND
    (3)*4)
360 POKE S+24,0
370 IF PEEK (EL+D(ED))=32 THEN 410
380 FOR X=0 TO 3: IF PEEK
    (EL+D(X))=32 THEN ED=X

```

Alteration—Change the symbols for the walls by changing **42** to **81** in line 340 and **250** to **160** in line 410.



```
390 NEXT X
400 IF PEEK (EL+D(ED))<>32 THEN 420
410 EL=EL+D(ED): POKE EL,250: GOTO 260
420 PRINT "[HOME]"; TAB( 17)"GOT
    ME!": FOR T=1 TO 2000: NEXT T
430 PRINT "[CLR]": PRINT
    : PRINT TAB( 15)"YOU GOT ME"
    : PRINT
440 GOTO 470
450 POKE 53280,2: PRINT "[HOME]";
    TAB( 13)"YOU GOT ZAPPED!"
    : FOR T=1 TO 2000: NEXT T
460 PRINT "[CLR]": PRINT
    : PRINT TAB( 13)"YOU GOT ZAPPED!"
    : PRINT : POKE 53280,6
470 POKE 198,0: INPUT "PLAY AGAIN";Q$
480 IF LEFT$(Q$,1)="Y" THEN 110
```

Rear Gunner

The computer shows you the view out the tail window of a Star Fighter traveling through space. From this vantage point, you will see enemy ships (Y and +), and members of your own squadron (U, V, W and X). You want to fire at and destroy the dangerous intruders, but be careful not to shoot at your allies!

Use the space bar to fire as soon as an asteroid or enemy ship appears. Your timing must be very good. You score 20 points if you hit an enemy. You lose 20 points if you destroy a friendly ship. You have two minutes to play.

```

100 REM REAR GUNNER
110 DIM T(11):HI=0:SO=54272
120 PRINT "[CLR]YOU ARE ABOUT TO
      TAKE YOUR PLACE IN THE
      TAILGUNNER POSITION."
130 PRINT "SHOOT DOWN THE ENEMY
      Y OR + FOR POINTS. AVOID
      SHOOTING DOWN MEMBERS"
140 PRINT "OF YOUR OWN SIDE
      : U V W X. YOU LOSE"
150 PRINT "POINTS FOR HITTING YOUR
      OWN FIGHTERS."
160 PRINT : PRINT "HIT ANY KEY TO
      START.": POKE 198,0
170 GET Q$: IF Q$="" THEN 170
180 POKE 53280,6: POKE 53281,0:SR=0
190 PRINT "[CLR]": FOR X=55296 TO
      56295: POKE X,1: NEXT
200 FOR X=1104 TO 2023: IF RND (8)<.1
      THEN POKE X,46
210 NEXT X: FOR X=1064 TO 1391 STEP
      41: POKE X,77: NEXT
220 FOR X=1103 TO 1392 STEP 39
      : POKE X,78: NEXT
      : FOR X=1392 TO 1415: POKE X,99
      : POKE X+280,100
230 NEXT : FOR X=1391 TO 1691 STEP 40
      : POKE X,103: POKE X+25,101
      : NEXT
240 FOR X=1984 TO 1689 STEP -39
      : POKE X,78: NEXT
      : FOR X=2023 TO 1716 STEP -41
      : POKE X,77: NEXT
250 FP=1523: FOR X=0 TO 4:T(X)=X+21
      : NEXT :T(5)=43: FOR X=6 TO 11
      :T(X)=32: NEXT
260 TI$="000000"
270 T= INT ( RND (8)*12)
      : POKE FP,T(T)
280 REM
290 IF PEEK (197)=60 AND PEEK (FP)<25
      THEN SR=SR-20: GOSUB 400

```

Alteration—Change the penalty for hitting a friend to 30 by changing the value of **SR** in line 290 to **SR=SR-30**.

Alteration—To change points scored when hitting an enemy to 35, change the value of **SR** in line 300 to **SR=SR+35**.

Alteration—To make the game last for three minutes, change **TI** in line 330 to **IF TI > 10800 THEN 360**.

```

300 IF PEEK (197)=60 AND ( PEEK
      (FP)=25 OR PEEK (FP)=43) THEN
      SR=SR+20: GOSUB 400
310 PRINT "[HOME]" TAB( 8)"[YEL]SCORE
      " STR$( SR)+"[SPACE 4]TIME
      " MID$( TI$,4,1)"
      : " RIGHT$( TI$,2)+" "
320 IF PEEK (197)<>60 OR PEEK (FP)=32
      THEN POKE SO+24,0
330 IF TI>7200 THEN 360
340 IF RND (8)>.9 THEN 270
350 GOTO 290
360 PRINT "[CLR][WHITE][DOWN 7]
      [SPACE 10]***GAME OVER***"
      : IF SR>HI THEN HI=SR
370 PRINT : PRINT "SCORE="SR
      : PRINT "HIGH ="HI
380 FOR X=1 TO 2000: NEXT X
390 GOTO 160
400 FOR X=SO TO SO+23: POKE X,0
      : NEXT : POKE SO+14,6
      : POKE SO+18,16: POKE SO+3,1
410 POKE SO+24,143: POKE SO+6,240
      : POKE SO+4,65:FR=5389
      : FOR T=0 TO 2
420 FQ=FR+144:HF= INT (FQ/256)
      :LF=FQ-HF*256: POKE SO+0,LF
      : POKE SO+1,HF
430 NEXT T: RETURN

```




Le Mans

Get your helmet and racing gloves ready! You are about to negotiate the challenging track at Le Mans. In this computer simulation, you use a joystick or < and > keys to steer. Avoid other cars, debris and running off the road.

Alteration—Make the track wider by adding more # symbols to string variable R\$ in line 130.

```

100 REM LE MANS
110 POKE 52,48: POKE 56,48: CLR
120 GOTO 410
130 R$="[MED.GRAY]#####":R=16
    :W= LEN (R$):A=5
140 PRINT "[CLR]": POKE 53281,0
    :L=1884
150 FOR I=1 TO 23: PRINT TAB( R)R$
    : NEXT
160 S=54272: FOR N=S TO S+24
    : POKE N,0: NEXT
170 POKE S+5,65: POKE S+6,65
    : POKE S+24,15: POKE S+1,1
    : POKE S,95: POKE S+4,129
180 K= RND (9): IF K<.3 AND R>1 THEN
    R=R-1
190 IF K>.7 AND R<38-W THEN R=R+1
    
```



Programming Note—Line 210 is a special trick that makes the screen display scroll backward.

Alteration—To decrease the amount of traffic, change M/500 in line 250 to M/1000.

```

200 PRINT "[HOME][DOWN 2]" TAB( R)R$
210 PRINT "[HOME][DOWN][LEFT][INST]"
    : POKE 218,156
220 MI= INT (M/10)/10: IF MI>HS THEN
    HS=MI
230 PRINT "[HOME][WHITE]MILES
    : " STR$ (MI)+"[SPACE 2]":M=M+1
240 IF RND (8)<M/3000 THEN PRINT "
    [HOME][DOWN 3]" TAB( R+ RND
    (9)*A+1)"[MED.GRAY]&"
250 IF RND (8)<M/500 THEN PRINT "
    [HOME][DOWN 3]" TAB( R+ RND
    (9)*A+1)"[MED.GRAY]%"
260 POKE L+40,35: POKE L+40+54272,12
    :JY= PEEK (56321):JY=15-(JY AND
    15)
270 IF PEEK (197)=47 OR (JY<7 AND
    JY>3) THEN L=L-1
280 IF PEEK (197)=44 OR JY>7 THEN
    L=L+1
290 IF PEEK (L)<>35 AND PEEK (L)<>36
    THEN 310
300 POKE L,36: POKE L+54272,1
    : GOTO 180
310 POKE L,35: POKE L+54272,2
320 S=54272: FOR N=S TO S+24
    : POKE N,0: NEXT
330 POKE S+5,25: POKE S+6,25
    : POKE S+24,15
340 POKE S+1,5: POKE S,235
    : POKE S+4,129
350 PRINT : PRINT "[HOME][DOWN 2]
    [WHITE][SPACE 11]*****CRASH*****
    **"
360 PRINT : PRINT "HIGH:"HS: PRINT
    : FOR T=0 TO 999: NEXT T
    : POKE S+24,0
370 PRINT "[SPACE 8]HIT ANY KEY
    TO RESTART": POKE 198,0
380 GET Q$: IF Q$="" THEN 380
390 M=0: GOTO 130
400 END
410 PRINT "[CLR][WHITE]": PRINT "YOU
    ARE ABOUT TO DRIVE AT LE MANS."
    : PRINT
420 PRINT "USE < TO STEER LEFT AND
    > TO STEER[SPACE 6]RIGHT OR
    USE JOYSTICK."
430 PRINT : PRINT "AVOID OTHER CARS
    AND DEBRIS.": PRINT
440 PRINT "PLEASE WAIT FOR ME TO
    SET UP."

```

MORE →

Programming Note—Lines 480 to 520 create a special set of graphic characters for this game.

```
450 POKE 56334, PEEK (56334) AND 254
    : POKE 1, PEEK (1) AND 251
460 C=12288: FOR I=0 TO 511
    : POKE I+C, PEEK (I+53248)
    : NEXT I: POKE 1, PEEK (1) OR 4
470 POKE 56334, PEEK (56334) OR 1
    : FOR T=1 TO 1000: NEXT
480 FOR I=280 TO 312: READ J
    : POKE I+C,J: NEXT I
490 HS=0: POKE 53272,( PEEK (53272)
    AND 240)+12
500 GOTO 130
510 DATA 255,255,255,255,255,255,255,
    255,66,0,66,195,195,66,0,66
520 DATA 189,129,165,231,231,165,129,
    189,239,215,85,170,85,171,247,
    239,215
```

Moon Lander

You must try to land your ship safely on the moon. Control the amount of thrust by pressing the number keys. The higher the thrust, the greater the deceleration and fuel use. If you run out of fuel, the engine won't work and the ship crashes.


To land safely, you must reach zero altitude at a rate between -5 and 5 meters per second.

```

100 REM MOON LANDER
110 GOSUB 490
120 PRINT "[HOME][YEL]ALTITUDE
      : " STR$ ( INT (A+.5))+" ",
      "METERS"
130 PRINT "VELOCITY:" STR$ ( INT
      (V+.5))+"[SPACE 2]","METERS/SEC"
140 PRINT "FUEL[SPACE 4]
      : " STR$ ( INT (F+.5))+" ","KG"
150 PRINT "THRUST[SPACE 2]
      : " STR$ ( INT (T*10+.5))+"%
      [SPACE 2]"
160 GET T$: IF T$<>" " THEN T= VAL
      (T$)
170 IF F<=0 THEN F=0:T=0
180 IF T$=" " THEN T=10
190 DV=V-G+(T*20000)/(M+F):F=F-T
200 A=A+(V+DV)/2:V=DV

```

MORE →




```
210 PRINT F$;: FOR X=1 TO 8
: PRINT TAB( 17)"[SPACE 3]"
: NEXT X
220 IF T>0 THEN PRINT F$"[YEL]+++"
230 IF T>1 THEN PRINT F$"[DOWN][RED]±
[YEL]±[RED]±"
240 IF T>2 THEN PRINT F$"[DOWN 2]
[RED]+++"
250 IF T>3 THEN PRINT F$"[DOWN 3]
[RED]+++"
260 IF T>4 THEN PRINT F$"[DOWN 4]
[RED]+++"
270 IF T>5 THEN PRINT F$"[DOWN 5]
[RED]+++"
280 IF T>6 THEN PRINT F$"[DOWN 6]
[RED]+++"
290 IF T>7 THEN PRINT F$"[DOWN 7]
[RED]+++"
300 IF A>5 THEN 120
310 V= ABS (V): IF V<10 THEN 410
320 POKE N-1,2: POKE N,7: PRINT "[CLR]
[WHITE][DOWN 10][SPACE 13]
***CRASH***"
330 FOR LO=SO TO SO+24: POKE LO,0
: NEXT
340 POKE SO+5,15: POKE SO+6,15
: POKE SO+24,15
350 POKE SO+1,1: POKE SO,235
: POKE SO+4,129: POKE SO+4,128
360 FOR X=1 TO 99: IF ( PEEK (N) AND
15)=7 THEN POKE N,2: GOTO 380
370 POKE N,7
380 FOR T=1 TO 44: NEXT T,X
390 FOR T=1 TO 2000: NEXT
: POKE 198,0
400 POKE N,0: POKE N-1,6
410 PRINT "[HOME][DOWN 6][WHITE]"
420 IF V<=5 THEN PRINT "PERFECT
LANDING!"
430 IF V>5 AND V<=10 THEN PRINT "HARD
LANDING"
440 IF V>10 THEN PRINT "CRASH
LANDING"
450 IF V>15 THEN PRINT "NO
SURVIVORS!"
460 PRINT : PRINT : PRINT
: INPUT "PLAY AGAIN";Q$
470 IF LEFT$(Q$,1)="Y" THEN 110
480 END
490 SO=54272:N=53281
500 POKE N,0: PRINT "[CLR][WHITE]"
510 PRINT TAB( 15)"[RVSON]MOON LANDER
[RVSOFF]": PRINT : PRINT
```

```

520 PRINT "TRY TO LAND YOUR SHIP
      FIVE METERS PER"
530 PRINT "SECOND OR LESS.": PRINT
540 PRINT "PRESS KEYS 0-9 TO SET
      THRUST."
550 PRINT "PRESS SPACE BAR FOR
      MAXIMUM": PRINT "THRUST."
560 PRINT : PRINT "HIT ANY KEY TO
      START."
570 GET Q$: IF Q$="" THEN 570
580 P$="[HOME][DOWN 5]":F$=P$+"[DOWN 10]
      [RIGHT 17]"
590 PRINT "[CLR]"
600 PRINT P$ TAB( 17)"HWUI "
610 PRINT TAB( 14)" [RVSON][POUND]
      [SPACE 5]*[RVSOFF]"
620 PRINT TAB( 14)" [RVSON] *[RVSOFF]
      [RVSON] [RVSOFF] [RVSON][POUND]
      "
630 PRINT TAB( 14)" [RVSON][SPACE 7]"
640 PRINT TAB( 14)" [RVSON][SPACE 3]
      [RVSOFF] [RVSON][SPACE 3]"
650 PRINT TAB( 14)" *[RVSON][SPACE 5]
      [RVSOFF][POUND]"
660 PRINT TAB( 14)"[RVSON][SPACE 9]"
670 PRINT TAB( 14)"[RVSON][SPACE 3]
      USA[SPACE 3]"
680 PRINT TAB( 13)"[RVSON]b[RVSOFF]
      iu[RVSON][SPACE 3][RVSOFF]ui
      b"
690 PRINT TAB( 12)"[RVSON]b[RVSOFF]iu
      [SPACE 2][RVSON][POUND] *[RVSOFF]
      [SPACE 2]uib"
700 PRINT TAB( 11)"[RVSON]b[RVSOFF]
      [SPACE 13]b"
710 PRINT TAB( 10)"c[RVSON]i[RVSOFF]
      [SPACE 13][RVSON]i[RVSOFF]v"
720 A=20000:M=8000:F=2000:G=10:V=0
      :T=0
730 RETURN

```

Alteration—Try changing the values for the following variables in line 720: **A** is starting altitude; **M** is mass of spaceship without fuel; **F** is amount of fuel; **G** is gravity, which affects the rate of fall.



Gambler

Play the one-armed bandit at the computer casino. The computer calculates payoff based on the amount bet—nickels, quarters or dollars. Jackpots are awarded for three of a kind. The computer plays until you've won \$100 or lost everything. To quit early, press the RUN/STOP key.


Alteration—Sound effects are due to lines 160, 560 and 770. Alter the sound by changing **N*5** in line 560 to **N*** any number from 1 to 10.

Alteration—Change the appearance of the slot machine by altering graphics and color symbols in lines 210-330.

```

100 REM GAMBLER
110 POKE 53281,1
120 D$="[HOME][DOWN 5]"
    :B$="[RIGHT 17]":H$=B$+"[RIGHT 11]"
    "
130 P$=D$+"[DOWN 6]"
140 R$(1)=P$+B$:R$(2)=R$(1)+"[RIGHT 3]"
    ":R$(3)=R$(2)+"[RIGHT 3]"
150 K$(1)="A":K$(2)="B":K$(3)="C"
160 FOR X=1 TO 6: READ S(X): NEXT
170 FOR X=1 TO 6: POKE S(X),0: NEXT
180 D=10
190 PRINT "[CLR][SPACE 4][BLK]WELCOME
    TO THE MICROCHIP CASINO!"
    : PRINT
200 PRINT : PRINT
210 PRINT TAB( 15)"[BLK][RVSON]
    [SPACE 10][RVSOFF]"
220 PRINT TAB( 13)"[BLK]([RVSON]
    [SPACE 12][RVSOFF])
230 PRINT TAB( 13)"[RVSON] [RED]bbbbbb
    bbbbbbb[BLK] [RVSOFF]"
240 PRINT TAB( 13)"[RVSON] [RED]bbb
    [BLK]BANDIT[RED]bbb[BLK] [RVSOFF]"
    "
250 PRINT TAB( 13)"[RVSON] [RED]bbbbbb
    bbbbbbb[BLK] [RVSOFF]"
260 PRINT TAB( 13)"[RVSON] [RED]bbbbbb
    bbbbbbb[BLK] [RVSOFF]"
270 PRINT TAB( 13)"[RVSON] [RED]b
    [BLU][SPACE 10][RED]b[BLK]
    [RVSOFF]"
280 PRINT TAB( 13)"[RVSON] [RED]b
    [BLU] [WHITE][SPACE 2][BLU]
    [WHITE][SPACE 2][BLU] [WHITE]
    [SPACE 2][BLU] [RED]b[BLK][SPACE 3]"
    "
290 PRINT TAB( 13)"[RVSON] [RED]b
    [BLU][SPACE 10][RED]b[BLK] "
300 PRINT TAB( 13)"[RVSON] [RED]bbbbbb
    bbbbbbb[BLK] "
310 PRINT TAB( 13)"[RVSON] [RED]bbbbbb
    bbbbbbb[BLK] "
320 PRINT TAB( 13)"[RVSON] [RED]bbbbbb
    bbbbbbb[BLK] "

```



```

330 PRINT TAB( 12)"[RVSON][SPACE 2]
    [RED]bbbbbbbbbbbbbb[BLK][SPACE 2]"
340 PRINT TAB( 11)"[RVSON][SPACE 18]"
350 PRINT : PRINT TAB( 14)"[RVSON]F-1
    [RVSOFF] NICKEL": PRINT
    : PRINT TAB( 14)"[RVSON]F-3
    [RVSOFF] QUARTER"
360 PRINT : PRINT TAB( 14)"[RVSON]F-5
    [RVSOFF] DOLLAR"
370 PRINT D$;H$" [RVSON] [RVSOFF]k"
    : PRINT H$" [RVSON]k[SPACE 2]"
380 FOR X=1 TO 4: PRINT H$" [RVSON]k"
    : NEXT X
390 PRINT H$" [RVSON] k"
400 FOR X=1 TO 3: PRINT R$(X)" [RVSON]
    [WHITE] [BLK]": NEXT X
410 PRINT "[HOME][DOWN 2][SPACE 8]YOU
    HAVE $" STR$(D)" LEFT.[SPACE 2]
    "
420 POKE 198,0
430 GET Q$: IF Q$="" THEN 430
440 B=0: IF Q$="[f1]" THEN B=.05
450 IF Q$="[f3]" THEN B=.25
460 IF Q$="[f5]" THEN B=1
470 IF B=0 OR B>D THEN 430
480 D=D-B
490 PRINT D$;H$" [SPACE 3]"
500 FOR X=1 TO 5: PRINT H$" [SPACE 3]"
    : NEXT X

```

MORE →

Alteration—Increase jackpot payoff by changing 50 in line 610 to a larger number.

```
510 PRINT P$;H$"[RVSON] k"  
520 FOR X=1 TO 3: PRINT H$" [RVSON]k"  
   : NEXT X  
530 PRINT H$"[RVSON]k[SPACE 2]"  
540 PRINT H$" [RVSON] [RVSOFF]k"  
550 C=0: POKE 54296,15  
560 N= INT ( RND (4)*20)+1  
   : FOR X=1 TO 6: POKE S(X),N*5  
   : NEXT  
570 IF N=20 AND RND (2)<.7 THEN C=C+1  
   :T$(C)=K$( INT ( RND (5)*3)+1)  
   : PRINT R$(C);T$(C)  
580 FOR X=1 TO 6: POKE S(X),0: NEXT  
590 IF C<3 THEN 560  
600 IF T$(1)<>T$(2) OR T$(1)<>T$(3)  
   OR T$(3)<>T$(2) THEN 690  
610 D=D+B* INT ( RND (8)*50)+1  
620 FOR Y=1 TO 7  
630 PRINT "[HOME][DOWN 3][SPACE 29]"  
640 FOR X=1 TO 6: POKE S(X),20: NEXT  
650 PRINT "[HOME][DOWN 3][SPACE 15]  
   [RVSON][RED] JACKPOT! [BLK]  
   [RVSOFF]"  
660 FOR T=1 TO 400: NEXT T  
670 FOR X=1 TO 6: POKE S(X),0  
   : NEXT X,Y  
680 PRINT "[HOME][DOWN 3][SPACE 33]"  
690 FOR T=0 TO 999: NEXT T  
700 PRINT P$;H$"[SPACE 3]"  
710 FOR X=1 TO 5: PRINT H$"[SPACE 3]"  
   : NEXT X  
720 IF D>100 THEN PRINT "[CLR] THE  
   'BANDIT' IS OUT OF MONEY!"  
   : GOTO 750  
730 IF D>0 THEN 370  
740 PRINT "[CLR]": PRINT "YOU ARE  
   BROKE!"  
750 PRINT : PRINT : INPUT "PLAY  
   AGAIN";Q$  
760 IF LEFT$(Q$,1)="Y" THEN 180  
770 DATA 54272,54273,54279,54280,  
   54286,54287
```


Orion

This is your chance to be a warrior, defending your cities by shooting down invading drones. You have three missile silos. Fire your interceptor missiles at the incoming drones by pressing keys 1, 2 or 3. Once you launch a missile, you can't launch another from that silo until the missile hits a drone or reaches maximum altitude. You lose when all of your buildings are destroyed.

```
100 REM ORION
110 GOSUB 470
120 PRINT "[CLR]": POKE 53280,0
    : POKE 53281,6: FOR X=55296 TO
    56176: POKE X,15: NEXT X
130 FOR X=56176 TO 56295: POKE X,7
    : NEXT X
140 FOR X=1904 TO 2023: POKE X,160
    : NEXT : Z=.1: DD=0: SR=8: V=0
```

MORE →





Alteration—Change the number and shapes of buildings by changing line 150.



```

150 PRINT A$;"[BLK][SPACE 3][RVSON]c
      [RVSOFF][SPACE 6][RVSON]vv
      [RVSOFF][SPACE 2]ik[SPACE 4]
      [RVSON]kv[RVSOFF]o[RVSON] v
      [RVSOFF][SPACE 4]d[RVSON]v
      [RVSOFF]ik"
160 FOR X=1 TO 3: POKE BS(X),30
      :BL(X)=BS(X):M(X)=0:F(X)=0: NEXT
170 Z=Z+.001
180 FOR X=1 TO 3: IF M(X)=1 THEN 210
190 J(X)=1344+ INT ( RND (8)*240)
      : IF RND (8)>Z THEN 250
200 POKE J(X)-1,M: POKE J(X),M
      : POKE J(X)+1,M:M(X)=1:D(X)=41
      : IF RND (8)>.5 THEN D(X)=39
210 IF PEEK (J(X))=M AND PEEK
      (J(X)+1)=M AND PEEK (J(X)-1)=M
      THEN 230
220 POKE J(X)-1,32: POKE J(X),32
      : POKE J(X)+1,32: POKE J(X)-1,32
      :M(X)=0: GOTO 250
230 POKE J(X)-1,32: POKE J(X),32
      : POKE J(X)+1,32:J(X)=J(X)+D(X)
      : IF J(X)>1903 THEN 370
240 POKE J(X),M: POKE J(X)+1,M
      : POKE J(X)-1,M
250 IF F(X)=1 THEN 280
260 GET F$:A= VAL (F$): IF A<1 OR A>3
      THEN 280
270 F(A)=1
280 IF PEEK (BL(X))<>30 AND
      (BL(X))<1864 THEN 390
290 POKE BL(X),32: IF F(X)=1 THEN
      BL(X)=BL(X)-40
300 IF BL(X)>1304 THEN 320
310 BL(X)=BS(X):F(X)=0
320 POKE BL(X),30: NEXT X
330 SR=0: FOR Y=1864 TO 1902
      : IF PEEK (Y)>90 THEN SR=SR+1
340 NEXT Y: PRINT "[HOME][WHITE]SCORE
      [SPACE 4]:"DD: PRINT "BUILDINGS
      : " STR$(SR)+" "
350 IF SR<1 THEN 430
360 GOTO 170
370 S0=54272: FOR Z=S0 TO S0+24
      : POKE Z,0: NEXT Z: POKE S0+5,15
      : POKE S0+6,15: POKE S0+24,15
380 POKE S0+1,3: POKE S0,245
      : POKE S0+4,129:M(X)=0: GOTO 250

```



```
390 SO=54272: FOR Z=SO TO SO+24
: POKE Z,0: NEXT Z: POKE SO+5,15
: POKE SO+6,15: POKE SO+24,15
400 POKE SO+1,5: POKE SO,235
: POKE SO+4,129: FOR Y=0 TO 8
: POKE BL(X)+EL(Y),EC(Y)
410 POKE 54272+BL(X)+EL(Y),7: NEXT Y
: FOR T=1 TO 50: NEXT T: DD=DD+SR
420 FOR Y=0 TO 8: POKE BL(X)+EL(Y),32
: POKE 54272+BL(X)+EL(Y),1
: NEXT Y: GOTO 310
430 PRINT "[HOME][DOWN 7][SPACE 9]
****CITY DESTROYED****": PRINT
: PRINT
440 FOR T=1 TO 2000: NEXT T
: POKE 198,0: PRINT "[SPACE 9]HIT
ANY KEY TO RESTART"
450 GET Q$: IF Q$="" THEN 450
460 GOTO 120
470 A$="[HOME][DOWN 21]"
480 FOR X=0 TO 8: READ EL(X),EC(X)
: NEXT
490 BS(1)=1913:BS(2)=1922:BS(3)=1931
:M=90:I=30
500 PRINT "[CLR]" TAB( 17);"ORION"
: PRINT : PRINT "[SPACE 2]DEFEND
YOUR CITY WITH INTERCEPTORS"
510 PRINT : PRINT "[SPACE 2]PRESS
KEYS 1-3 TO LAUNCH MISSILES."
: PRINT
520 PRINT "[SPACE 9]HIT ANY KEY
TO START"
530 GET Q$: IF Q$="" THEN 530
540 RETURN
550 DATA -41,85,-40,66,-39,73,-1,67,0,9
1,67,39,74,40,66,41,75
```

Berserkotron


You're applying for a job at a computer-game factory, but get lost in a room of berserk robots. Escaping through one of three exits is your only hope of survival.

This is made difficult because the robots chase you. Though berserk, they aren't stupid. Lines 210-270 give the robots the ability to track your position. However, you can use the barriers in the room to block a robot's progress. The game ends if one catches you.

```

100 REM BERSERKOTRON
110 GOSUB 510
120 JY= PEEK (56321):JY=15-(JY AND 15)
   :K= PEEK (197):J= PEEK (653):AL=L
130 POKE L,32: IF JY=1 OR JY=5 OR
   JY=9 OR (K=7 AND J=1) THEN
   AL=AL-40
140 IF JY=2 OR JY=6 OR JY=10 OR (K=7
   AND J=0) THEN AL=AL+40
150 IF (JY>3 AND JY<7) OR (K=2 AND
   J=1) THEN AL=AL-1
160 IF JY>7 OR (K=2 AND J=0) THEN
   AL=AL+1
170 IF PEEK (AL)<>160 THEN L=AL
180 IF PEEK (L)=R THEN 300
190 IF PEEK (L)=46 THEN 440
200 POKE L,I:LY= INT ((L-S)/40)
   :LX=L-S-LY*40
210 FOR X=1 TO 3:M=R(X): POKE M,32
220 RY= INT ((M-S)/40):RX=M-S-RY*40
230 IF RX<LX AND ( PEEK (M+1)=32 OR
   PEEK (M+1)=I) THEN M=M+1
240 IF RX>LX AND ( PEEK (M-1)=32 OR
   PEEK (M-1)=I) THEN M=M-1
250 IF RY>LY AND ( PEEK (M-40)=32 OR
   PEEK (M-40)=I) THEN M=M-40
260 IF RY<LY AND ( PEEK (M+40)=32 OR
   PEEK (M+40)=I) THEN M=M+40
270 R(X)=M: POKE M,R
280 IF M=L THEN 300
290 NEXT X: GOTO 120
300 REM CAUGHT
310 S0=54272: FOR X=0 TO 24
   : POKE S0+X,0: NEXT X
320 POKE S0+14,5: POKE S0+18,16
   : POKE S0+3,1: POKE S0+24,143
330 POKE S0+6,240: POKE S0+4,65
340 POKE 53280,2:F1=5389
   : FOR X=0 TO 119:F=F1+ PEEK
   (S0+27)*3.5
350 H= INT (F/256):L=F-H*256
360 IF INT (X/10)/2= INT ( INT
   (X/10)/2) THEN POKE 53280,2
   : POKE 53281,2

```



```

370 IF INT (X/10)/2<> INT ( INT
    (X/10)/2) THEN POKE 53280,14
    : POKE 53281,11
380 POKE 50,L: POKE 50+1,H: NEXT X
390 POKE 50+24,0
400 FOR X=1 TO 2000: NEXT X
410 PRINT "[CLR]": PRINT : PRINT
420 PRINT TAB( 12)"YOU WERE CAUGHT!"
    : PRINT : PRINT : PRINT
430 GOTO 480
440 REM SAFE
450 FOR X=1 TO 2000: NEXT X
460 PRINT "[CLR]": PRINT : PRINT
470 PRINT TAB( 14)"YOU MADE IT!"
    : PRINT : PRINT
480 POKE 198,0: PRINT TAB( 14)"PLAY
    AGAIN";: INPUT Q$
490 IF LEFT$(Q$,1)="Y" THEN 110
500 END
510 PRINT "[CLR][BLK]": POKE 53281,11
520 PRINT TAB( 14);"[RVSON]
    BESERKOTRON": PRINT
530 PRINT "ON THE WAY TO INTERVIEW
    FOR A JOB AS A"
540 PRINT "COMPUTER GAME TESTER,
    YOU MADE A WRONG TURN...": PRINT

```

MORE →

```
550 PRINT "YOU HAVE JUST ENTERED
      THE ROBOT ROOM."
560 PRINT "YOU ARE NOT WELCOME."
      : PRINT
570 PRINT "USE A JOYSTICK OR THE
      CURSOR KEYS TO"
580 PRINT "RUN THROUGH THE SIDE OR
      BOTTOM DOORS.": PRINT
590 PRINT "ENTER YOUR FIRST INITIAL."
600 GET I$: IF I$="" THEN 600
610 IF ASC (I$)<65 OR ASC (I$)>90
      THEN 600
620 PRINT "AVOID ROBOTS ([RVSON]R
      [RVSOFF]).": PRINT
630 PRINT "YOUR ONE ADVANTAGE IS
      YOU CAN GO THROUGH";
640 PRINT "THE BLACK CIRCLES AND
      THE ROBOTS CANNOT.": PRINT
650 PRINT "HIT ANY KEY TO START."
660 GET Q$: IF Q$="" THEN 660
670 PRINT "[CLR]": PRINT "[BLK]
      [RVSON][SPACE 37]"
680 FOR X=1 TO 9: PRINT " [RVSON]
      "; TAB( 37);"[RVSON] ": NEXT X
      : PRINT " ."; TAB( 37);"."
690 FOR X=1 TO 9: PRINT " [RVSON]
      "; TAB( 37);"[RVSON] ": NEXT X
700 PRINT " [RVSON][SPACE 19][RVSOFF]
      .[RVSON][SPACE 17]"
710 S=1106:B=81:R=146:I= ASC (I$)-64
720 BX=199
730 FOR X=55296 TO 56295: POKE X,0
      : NEXT X
740 FOR X=0 TO BX: POKE S+40* INT (
      RND (8)*19)+ INT ( RND (8)*35),B
      : NEXT X
750 R(1)=1428:R(2)=1458:R(3)=1764
      :L=S+18
760 FOR X=1 TO 3: POKE R(X),R: NEXT X
      : POKE L,I
770 RETURN
```

Alteration—Change the symbols for barriers and robots by changing the values of **B** and **R** respectively in line 710.

Alteration—Change the number of barriers by changing the value of **BX** in line 720. For example, **BX=50** would make the game more difficult; **BX=250** would make the game easier.

Shell Game

In this game, the computer will place a ball under one of three boxes. Then it shuffles the boxes around the screen. Watch carefully because you're asked to find the box that has the ball.

```

100 REM SHELL GAME
110 GOSUB 450
120 PRINT "[CLR]" TAB( 15) "[RVSON]
    SHELL GAME[RVSOFF]"
130 FOR X=7 TO 9: FOR Y=1 TO 3
    : PRINT L$(X);S$(Y): NEXT Y,X
140 L= INT ( RND (8)*3)+1
150 PRINT L$(L);B$
160 PRINT "[BLK][HOME][DOWN 2]HIT
    ANY KEY TO START"
170 GET Q$: IF Q$="" THEN 170
180 PRINT "[HOME][DOWN 2][SPACE 21]"
190 FOR X=7 TO 9: FOR Y=1 TO 3
    : PRINT L$(X);E$(Y): NEXT Y,X
200 FOR X=1 TO 3: FOR Y=1 TO 3
    : PRINT L$(X);S$(Y): NEXT Y,X
210 FOR X=1 TO 3:W(X)=1: NEXT X:M=1
220 FOR K=1 TO 9:C=0: IF W(K)=0 THEN
    290
230 H= INT ( RND (4)*4)+1:C=C+1
    : IF C>9 THEN 290
240 N=A(K,H): IF W(A(K,H))=1 THEN 230
250 W(K)=0:W(A(K,H))=1
260 FOR Y=1 TO 3: PRINT L$(K);E$(Y)
    : NEXT Y

```

Alteration—You may want to add sound effects similar to those used in **Gambler** while the boxes are being shuffled.

MORE →



Alteration—Change the amount of shuffling the computer does by altering line 300. For example, **IF M < 4 THEN 220** decreases shuffling; **IF M < 15 THEN 220** increases shuffling.

```

270 FOR Y=1 TO 3: PRINT L$(A(K,H));
    S$(Y): NEXT Y
280 IF L=K THEN L=A(K,H)
290 NEXT K:M=M+1
300 IF M<9 THEN 220
310 IF W(1)=1 AND W(2)=1 AND W(3)=1
    THEN 330
320 GOTO 220
330 FOR X=1 TO 3: PRINT L$(X);"[RVSON]
    "; STR$(X): NEXT X
340 PRINT "[BLK][HOME][DOWN 2]OK,
    WHICH BOX IS THE BALL UNDER?"
350 GET Q$: IF Q$="" THEN 350
360 Q=VAL(Q$): IF Q<1 OR Q>3 THEN
    350
370 IF Q=L THEN PRINT "YOU GOT IT!"
    : GOTO 390
380 PRINT "SORRY, IT WAS UNDER #"L
390 FOR X=1 TO 3: FOR Y=1 TO 3
    : PRINT L$(X);E$(Y): NEXT Y,X
400 FOR X=4 TO 6: FOR Y=1 TO 3
    : PRINT L$(X);S$(Y): NEXT Y,X
410 PRINT L$(L);B$
420 PRINT : INPUT "PLAY AGAIN";Q$
430 IF LEFT$(Q$,1)="Y" THEN 120
440 END
450 POKE 53281,1: PRINT "[CLR][BLK]"
460 PRINT "YOU ARE ABOUT TO PLAY
    [RVSON]SHELL GAME[RVSOFF]."
470 PRINT "I WILL PLACE A BALL (Q)
    UNDER ONE OF"
480 PRINT "THREE RED BOXES AND THEN
    SHUFFLE THE"
490 PRINT "BOXES AROUND.": PRINT
500 PRINT "WATCH CAREFULLY AND THEN
    TELL ME WHICH[SPACE 2]BOX THE
    BALL IS UNDER."
510 PRINT : PRINT "HIT ANY KEY TO
    CONTINUE."
520 GET Q$: IF Q$="" THEN 520
530 S$(1)="[RED][RVSON][SPACE 3]
    [RVSOFF]";E$(1)="[SPACE 3]"
540 S$(2)="[DOWN][RED][RVSON][SPACE 3]
    [RVSOFF]";E$(2)="[DOWN][SPACE 3]"
550 S$(3)="[DOWN 2][RED][RVSON][SPACE
    3][RVSOFF]";E$(3)="[DOWN 2][SPACE
    3]"
560 B$="[BLU][DOWN][RIGHT]Q"
570 L$(1)="[HOME][DOWN 12][RIGHT 12]"
580 L$(2)=L$(1)+"[RIGHT 6]"
590 L$(3)=L$(2)+"[RIGHT 6]"
600 L$(4)=L$(1)+"[DOWN 6]"
610 L$(5)=L$(2)+"[DOWN 6]"
620 L$(6)=L$(3)+"[DOWN 6]"

```



```
630 L$(7)=L$(1)+"[UP 6]"
640 L$(8)=L$(2)+"[UP 6]"
650 L$(9)=L$(3)+"[UP 6]"
660 FOR X=1 TO 9: FOR Y=1 TO 4
      : READ A(X,Y): NEXT Y,X
670 RETURN
680 DATA 2,4,7,1,1,3,5,8,2,6,9,3
690 DATA 1,4,5,4,4,2,6,5,5,3,6,6
700 DATA 1,7,8,7,7,2,9,8,8,9,3,9
```



Pirates

Batten down your hatches and head for the high seas. Can you recover the gold stolen by the pirate Bluebeard? You must search for five pirate ships hidden in the waters displayed on the screen. When you find a pirate ship, you can either fire on it or board it. In the ensuing battle you may lose men and cannons.

The game ends after five minutes or after you or all five pirate ships are sunk.

```
100 REM PIRATES
110 PRINT "[CLR][WHITE]": PRINT "AHOY
      THERE MATEY. AND WHAT WOULD
      BE"
120 INPUT "YOUR NAME NOW";N$: PRINT
130 PRINT "WELL, WELL, CAPTAIN "N$"
      IT IS.": PRINT
140 PRINT "AND I'M CAPTAIN
      BLUEBEARD.": PRINT
150 PRINT "I HAVE A FLEET OF 5
      FRIGATES HIDDEN IN[SPACE 2]THESE
      WATERS. CATCH ME IF
160 PRINT "YOU CAN.": PRINT
      : PRINT "YOUR SHIP IS REPRESENTED
      BY AN 'S'."
170 PRINT "MY SHIPS ARE REPRESENTED
      BY A 'P'."
180 PRINT "REEFS ARE REPRESENTED
      BY AN '*'.": PRINT
190 PRINT "YOUR SHIP HAS A CREW
      OF 200 MEN AND 50"
200 PRINT "CANNONS.": PRINT
      : PRINT "HIT ANY KEY"
210 GET Q$: IF Q$="" THEN 210
220 PRINT "[CLR]": PRINT "YOUR
      COMMAND OPTIONS ARE:": PRINT
230 PRINT " N = STEER NORTH"
      : PRINT " E = STEER EAST"
240 PRINT " S = STEER SOUTH"
      : PRINT " W = STEER WEST"
250 PRINT " F = FIRE CANNONS"
      : PRINT " B = BOARD": PRINT
260 PRINT "CANNONS WILL FIRE ON
      ALL SHIPS WITHIN[SPACE 3]TWO
      SPACES."
270 PRINT "TO BOARD A SHIP, IT MUST
      BE ONE SPACE[SPACE 3]AWAY."
280 PRINT : PRINT "YOU HAVE 5
      MINUTES"
290 PRINT : PRINT "HIT ANY KEY TO
      START."
300 GET Q$: IF Q$="" THEN 300
310 LX= INT (40* RND (8))
      :LY= INT ( RND (8)*20)+3:S=1024
      :M=200:C=50:G=0:K=0
```


```

320 FOR X=0 TO 4
330 PX(X)= INT (40* RND (8))
    :PY(X)= INT ( RND (8)*20)+3
    :M(X)=100:C(X)=20
340 IF PX(X)=LX AND PY(X)=LY THEN 330
350 NEXT X: PRINT "[CLR]"
    :L=S+LX+40*LY: POKE L,19
    : FOR X=0 TO 24
360 I= INT ( RND (4)*800)+S+120
    : IF PEEK (I)<>32 THEN 360
370 POKE I,42: NEXT X
375 FOR X=55296 TO 56295: POKE X,1:
    NEXT X
380 TI$="000000"
390 L=S+LX+40*LY: POKE L,19
400 C$=" ": GET C$
410 PRINT "[HOME]MEN:" STR$ (M)+"
    [SPACE 2]CANNONS:" STR$ (C)+"
    [SPACE 2]GOLD:" STR$ (G)+"[SPACE 2]
    "
420 FOR X=0 TO 4:D(X)= SQR
    ((PX(X)-LX)^2+(PY(X)-LY)^2)
430 IF C(X)=<0 OR M(X)=<0 THEN 620
440 IF D(X)>6 THEN POKE
    S+PX(X)+PY(X)*40,32: GOTO 620
450 POKE S+PX(X)+PY(X)*40,16
460 IF D(X)>2.9 THEN 560
470 IF C$<>"F" THEN 530

```

Alteration—Change the range at which pirate ships are spotted by altering **D(X)** in line 440. For example, **IF D(X) > 4** will increase the range to four spaces.

MORE →



Alteration—Changing the equation for **C(X)** in line 520 changes the damage done by your cannons. For example, **C(X)=C(X)-INT(RND(7)*.2*C)** will do less damage.

Alteration—Changing the equation for **G** in line 610 changes the amount of gold recovered from a captured pirate ship. For example, **G=G+INT(RND(8)*8000)** will increase the average amount of gold recovered.


Alteration—Changing **TI** in line 710 to **IF TI>10800 THEN 890** decreases the time limit to three minutes.

```

480 PRINT "[HOME][DOWN]FIRING CAPTAIN
      "N$!"
490 FOR CA=1 TO C/5:SO=54272
      : FOR LO=SO TO SO+24: POKE LO,0
      : NEXT
500 POKE SO+5,9: POKE SO+6,15
      : POKE SO+24,15
510 POKE SO+1,5: POKE SO,235
      : POKE SO+4,129: FOR TT=1 TO 40
      : NEXT TT,CA
520 C(X)=C(X)-INT(RND(7)*.3*C)
      :M(X)=M(X)-INT(RND(8)*.2*C)
530 IF RND(2)>.5 THEN C=C-INT(RND
      (7)*.3*C(X)):M=M-INT(RND
      (8)*.1*C(X))
540 IF C(X)<=0 THEN PRINT "[HOME]
      [DOWN]PIRATE SHIP SUNK[SPACE 13]"
      : FOR T=1 TO 2000: NEXT T:K=K+1
550 IF C(X)<=0 THEN PRINT "[HOME]
      [DOWN][SPACE 19]"
560 IF D(X)>1.5 THEN 620
570 IF C$="B" THEN PRINT "[HOME]
      [DOWN]BOARDING CAPTAIN "N$!"
580 IF C$="B" THEN M(X)=M(X)-INT
      (RND(7)*.5*M)
590 IF M(X)>0 THEN M=M-INT(RND
      (7)*.1*M(X))
600 IF M(X)<=0 THEN PRINT "[HOME]
      [DOWN]PIRATE SHIP CAPTURED.
      [SPACE 13]": FOR T=1 TO 4000
      : NEXT T
610 IF M(X)<=0 THEN PRINT "[HOME]
      [DOWN][SPACE 22]":G=G+INT(RND
      (8)*5000):K=K+1
620 IF C(X)<=0 OR M(X)<=0 THEN POKE
      S+PX(X)+PY(X)*40,32
630 NEXT X: IF K=5 THEN 730
640 PRINT "[HOME][DOWN][SPACE 31]"
650 IF M<=0 OR C<=0 THEN 790
660 POKE L,32: IF C$="N" AND LY>2
      THEN LY=LY-1
670 IF C$="S" AND LY<22 THEN LY=LY+1
680 IF C$="E" AND LX<40 THEN LX=LX+1
690 IF C$="W" AND LX>1 THEN LX=LX-1
700 IF PEEK(S+LX+40*LY)=42 THEN 840
710 IF TI>18000 THEN 890
720 GOTO 390
730 PRINT "ALL PIRATES ELIMINATED"
740 FOR T=1 TO 3000: NEXT T
750 PRINT "[CLR]"
760 PRINT TAB(10)"ALL PIRATES
      ELIMINATED."
770 PRINT : PRINT "YOU GOT"G"GOLD
      PIECES": PRINT "CAPTAIN "N$".

```

```
780 GOTO 940
790 PRINT "YOU ARE SUNK"
800 FOR T=1 TO 3000: NEXT T
810 PRINT "[CLR]"
820 PRINT "YOU ARE SUNK CAPTAIN
      "N$". "
830 GOTO 940
840 PRINT "YOU RAN AGROUND."
850 FOR T=1 TO 3000: NEXT T
860 PRINT "[CLR]"
870 PRINT "YOU RAN AGROUND CAPTAIN
      "N$". "
880 GOTO 940
890 PRINT "TIME HAS EXPIRED"
900 FOR T=1 TO 3000: NEXT T
910 PRINT "[CLR]"
920 PRINT TAB( 14)"TIME HAS EXPIRED."
930 PRINT : PRINT "YOU GOT"G"GOLD
      PIECES": PRINT "CAPTAIN "N$". "
940 POKE 198,0: PRINT
      : PRINT TAB( 10)"PLAY AGAIN
      (Y/N)";
950 INPUT Q$
960 IF LEFT$(Q$,1)="Y" THEN 310
```



Master Code

This short program is based on a popular board game. The computer invites you to break a secret code. When the four squares appear on the screen, you enter one number of code at a time. The number is between 1 and 5. A number *may* be repeated.

A correct answer is indicated by an asterisk in that square. You have four tries to break the code. You win when you guess all four numbers in the code.

```

100 REM MASTER CODE
110 M$="[HOME][DOWN 5]":L$=M$+"[DOWN 3]
    [RIGHT 19]"
120 POKE 53281,1: PRINT "[CLR][BLK]"
    : PRINT TAB( 14)"MASTER CODE"
    : PRINT : PRINT
130 PRINT "I AM CATHY CRYPTIC, A
    CYBERNETIC CODE[SPACE 3]
    GENERATING PROGRAM."
140 PRINT "YOU ARE ABOUT TO ATTEMPT
    TO BREAK"
150 PRINT "MY CODE.": PRINT
160 PRINT "I WILL GENERATE FOUR
    SQUARES. EACH"
170 PRINT "SQUARE WILL REPRESENT
    ONE NUMBER:[SPACE 7]1,2,3,4,
    OR 5.": PRINT
180 PRINT "YOU WILL HAVE FOUR TRIES
    TO OBTAIN THE[SPACE 2]CORRECT
    SEQUENCE.": PRINT
190 PRINT "NUMBERS MAY BE USED MORE
    THAN ONCE.": PRINT : PRINT "HIT
    ANY KEY TO START."
200 GET Q$: IF Q$="" THEN 200
210 PRINT "[CLR]": PRINT TAB(
    14)"MASTER[SPACE 2]CODE": PRINT
    : PRINT
220 E= INT ( RND (8)*5)+1
230 B= INT ( RND (8)*5)+1
240 C= INT ( RND (8)*5)+1
250 D= INT ( RND (8)*5)+1
260 PRINT TAB( 13)"A[SPACE 3]B[SPACE 3]
    C[SPACE 3]D"
270 PRINT TAB( 13)"[RVSON][SPACE 2]
    [RVSOFF][SPACE 2]++[SPACE 2]
    [RVSON]+ [RVSOFF][SPACE 2][RVSON]
    [SPACE 2][RVSOFF]"
280 PRINT TAB( 13)"[RVSON][SPACE 2]
    [RVSOFF][SPACE 2]++[SPACE 2]
    [RVSON]+ [RVSOFF][SPACE 2][RVSON]
    ++[RVSOFF]"
290 FOR X=1 TO 4: PRINT L$;X
300 IF W=E THEN 340
310 PRINT L$: PRINT : PRINT TAB( 13)"
    [SPACE 2]A";: INPUT W

```

Alteration—Make the game more difficult by changing the **5** in lines 220-250 to a larger number. This increases the range of numbers that the computer can use in its code. Also, indicate the new range by changing the **PRINT** statement in line 170.

Alteration—Make the game easier by increasing the number of tries the user gets. For example, changing **X** in line 290 to **FOR X=1 TO 5: PRINT L\$;X** allows the user five tries.

```

320 IF W=E THEN PRINT M$; TAB( 13)"
      [RVSON]**[RVSOFF]"
      : PRINT TAB( 13)"[RVSON]**
      [RVSOFF]"
330 PRINT L$: PRINT : PRINT TAB( 13)"
      [SPACE 7]"
340 IF V=B THEN 380
350 PRINT L$: PRINT : PRINT TAB( 13)"
      [SPACE 2]B";: INPUT V
360 IF V=B THEN PRINT M$; TAB( 17)"
      [RVSON]**[RVSOFF]"
      : PRINT TAB( 17)"[RVSON]**
      [RVSOFF]"
370 PRINT L$: PRINT : PRINT TAB( 13)"
      [SPACE 9]"
380 IF Y=C THEN 420
390 PRINT L$: PRINT : PRINT TAB( 13)"
      [SPACE 2]C";: INPUT Y
400 IF Y=C THEN PRINT M$; TAB( 21)"
      [RVSON]**[RVSOFF]"
      : PRINT TAB( 21)"[RVSON]**
      [RVSOFF]"
410 PRINT L$: PRINT : PRINT TAB( 13)"
      [SPACE 6]"
420 IF Z=D THEN 460
430 PRINT L$: PRINT : PRINT TAB( 13)"
      [SPACE 2]D";: INPUT Z
440 IF Z=D THEN PRINT M$; TAB( 25)"
      [RVSON]**[RVSOFF]"
      : PRINT TAB( 25)"[RVSON]**
      [RVSOFF]"
450 PRINT L$: PRINT : PRINT TAB( 13)"
      [SPACE 6]"
460 IF W=E AND V=B AND Y=C AND Z=D
      THEN 520
470 NEXT X
480 PRINT : PRINT : PRINT TAB( 6)"YOU
      DID NOT BREAK THE CODE."
490 PRINT : PRINT TAB( 8)"IT WAS
      :";E" "B" "C" "D
500 PRINT : PRINT TAB( 7)"HIT ANY
      KEY TO PLAY AGAIN."
510 GOTO 200
520 PRINT : PRINT "[SPACE 4]YOU
      BROKE THE CODE IN"X"TRIES!!!"
530 GOTO 500

```



A Day At The Races

You've gone to the dogs—the dog track that is. You and the computer can choose among four dogs racing toward the finish line.


The program is complete with user-friendly features such as an introduction, a racing program, fun graphics and easy restart. It works great as is, but it's also fun to personalize.

```
100 REM A DAY AT THE RACES
110 M=500: POKE 53281,11
120 C$="[SPACE 5]"
130 C$(1,1)="[BLK][SPACE 3][RVSON]vc"
: C$(1,2)="N[RVSON] [RVSOFF]1
[RVSON] ": C$(1,3)=" [RVSON]c
[RVSOFF] [RVSON]c"
140 C$(2,1)="[YEL][SPACE 3]f"
: C$(2,2)="N[RVSON] [RVSOFF]2
[RVSON] i": C$(2,3)=" [RVSON]c
[RVSOFF] [RVSON]c"
150 C$(3,1)="[BRN] ) b"
: C$(3,2)=" [RVSON] [RVSOFF]3
[RVSON] i": C$(3,3)=" k k"
160 C$(4,1)="[WHITE][SPACE 3][RVSON]
vc": C$(4,2)="(+4+"
: C$(4,3)=" [RVSON]c[RVSOFF]
[RVSON]c[RVSOFF]"
170 PRINT "[CLR][WHITE]"
: PRINT "WELCOME TO UNCLE LOUIS'
DOGTRACK, FINEST";
180 PRINT "DOGTRACK IN ARIZONA."
: PRINT
190 PRINT "I HOPE YOU'RE FEELING
LUCKY TODAY."
200 PRINT "I HAVE SOME OF THE FINEST
PUPS IN THE"
210 PRINT "MICROCHIP WORLD JUST
READY TO RUN THEIR PAWS OFF."
220 PRINT : PRINT "SO MAKE YOURSELF
AT HOME IN THE BOX[SPACE 5]
SEAT.": PRINT
230 PRINT "BY THE WAY YOU CAN CALL
ME UNCLE LOUIS."
240 INPUT "WHAT IS YOUR NAME";N$
: PRINT
250 PRINT N$", WOULD YOU LIKE A
PROGRAM";: INPUT P$
260 IF LEFT$(P$,1)<>"Y" THEN 470
270 PRINT "[CLR]1) [RVSON]MOLLY
[RVSOFF]: HUSKY MIXED BREED
CAN REALLY"
280 PRINT "[SPACE 3]HAUL...HAS WON
16 OUT OF 24"
```

Alteration—Change the names and descriptions of the dogs in lines 270-460 and 500 to match the names and characteristics of family and neighborhood dogs.


```
290 PRINT "[SPACE 3]PROFESSIONAL
    RACES. THIS 7 YEAR OLD"
300 PRINT "[SPACE 3]HAS EXPERINCE,
    CONFIDENCE AND CLASS.";
310 PRINT "[SPACE 4]A SURE BET IF
    SHE DOESN'T STOP ALONG";
320 PRINT "[SPACE 4]THE TRACK FOR
    A SNACK.": PRINT "[SPACE 3]ODDS
    : 2-1": PRINT
330 PRINT "2) [RVSON]FELICE[RVSOFF]
    THIS 8 YEAR OLD SHAGGY HAS"
340 PRINT "[SPACE 3]BEEN KNOWN TO
    FALL ASLEEP IN THE"
350 PRINT "[SPACE 3]MIDDLE OF THE
    TRACK. BUT WATCH OUT."
360 PRINT "[SPACE 3]THIS SLEEPER
    IS FULL OF SURPRISES."
    : PRINT "[SPACE 3]ODDS: 10-1"
    : PRINT
370 PRINT "[SPACE 3]HIT ANY KEY
    FOR NEXT PAGE": PRINT
380 GET Q$: IF Q$="" THEN 380
390 PRINT "3) [RVSON]BAD DOG[RVSOFF]
    THIS GUY CAN RUN..."
400 PRINT "[SPACE 3]UNFORTUNATELY,
    HE HAS BEEN KNOWN TO"
410 PRINT "[SPACE 3]THROW A FEW
    RACES...HIS LICENSE IS"
420 PRINT "[SPACE 3]ON PROBATION."
    : PRINT "[SPACE 3]ODDS: 4-1"
    : PRINT
430 PRINT "4) [RVSON]BILLY[RVSOFF]
    THIS PUP HAS SPEED, PEP, AND
    [SPACE 2]";
```

MORE →




Alteration—The variables in line 510 control the relative speeds of the dogs. The lower the number, the faster the dog. For example, changing **O(1)** to **O(1)=.85** will make dog #1, Molly, faster.

```

440 PRINT "[SPACE 4]VITALITY...IF
      HE RUNS IN THE RIGHT"
450 PRINT "[SPACE 3]DIRECTION. HE
      IS A REAL THREAT."
460 PRINT "[SPACE 3]FIRST PROFESSIONAL
      RACE.": PRINT "[SPACE 3]ODDS
      : 7:1"
470 PRINT : PRINT : PRINT TAB( 7)"HIT
      ANY KEY TO START": PRINT
480 GET Q$: IF Q$="" THEN 480
490 PRINT "[CLR]": PRINT "PLACE
      YOUR BET BY SELECTING NUMBER."
      : PRINT
500 D$(1)="MOLLY":D$(2)="FELICE"
      :D$(3)="BAD DOG":D$(4)="BILLY"
510 O(1)=.95:O(2)=.98:O(3)=.96
      :O(4)=.97
520 FOR X=1 TO 4: PRINT X") "D$(X)
      : PRINT : NEXT X
530 INPUT "WHICH DOG";S
540 IF S>4 OR S<1 THEN 490
550 PRINT : PRINT "YOU HAVE $"M
560 PRINT "HOW MUCH DO YOU WANT
      TO BET";: INPUT B
570 IF B>M THEN PRINT "SORRY YOU
      DO NOT HAVE CREDIT": GOTO 550
580 A$(1)="[HOME][DOWN 2]"
      :A$(2)=A$(1)+"[DOWN 5]"
      :A$(3)=A$(2)+"[DOWN 5]"
590 FOR X=1 TO 4:B$(X)="[RIGHT]"
      : NEXT X
600 A$(4)=A$(3)+"[DOWN 5]"
610 T$="[DOWN 4][RVSON][BLK][SPACE 36]
      [RVSOFF]"
620 PRINT "[CLR]"; TAB( 15)"[RVSON]
      RACE TRACK[RVSOFF]": PRINT
630 FOR X=1 TO 22: PRINT TAB( 36)"
      [RED][RVSON] [RVSOFF]": NEXT X
640 PRINT A$(1);T$: PRINT A$(2);T$
      : PRINT A$(3);T$: PRINT A$(4);T$
650 FOR X=1 TO 4: PRINT A$(X)
      : FOR Y=1 TO 3: PRINT B$(X);C$(X,
      Y): NEXT Y,X
660 FOR X=1 TO 4: PRINT A$(X)
      : FOR Y=1 TO 3: PRINT B$(X);C$
      : NEXT Y
670 IF RND (8)>.6 THEN B$(X)=B$(X)+"
      [RIGHT]"
680 IF RND (8)>O(X) THEN
      B$(X)=B$(X)+"[RIGHT]"
690 PRINT A$(X): FOR Y=1 TO 3
      : PRINT B$(X);C$(X,Y): NEXT Y

```

```
700 IF LEN (B$(X))>30 THEN PRINT "[HOME][DOWN][WHITE]#"X;D$(X)"
      WINS": GOTO 720
710 NEXT X: GOTO 650
720 FOR T=1 TO 3000: NEXT T
      : PRINT "[CLR]": PRINT
      : PRINT TAB( 7)"#"X;D$(X)" WINS!"
      : PRINT
730 PRINT TAB( 7);N$" BET ON #"S;
      D$(S): PRINT
740 IF S=X THEN 770
750 PRINT TAB( 7)N$" LOSES $"B:M=M-B
      : IF M<=0 THEN PRINT TAB( 7)N$"
      IS BROKE!": END
760 GOTO 820
770 IF S=1 THEN B=B*2
780 IF S=2 THEN B=B*10
790 IF S=3 THEN B=B*4
800 IF S=4 THEN B=B*7
810 PRINT TAB( 7)N$" WINS $"B:M=M+B
820 PRINT : PRINT : INPUT "PLAY
      AGAIN";P$: IF LEFT$( P$,1)="Y"
      THEN PRINT : GOTO 470
830 PRINT : PRINT : PRINT TAB( 7)N$"
      LEFT THE TRACK WITH $"M
```



Lost Dutchman Mine

Here's a game based on an Arizona folk tale about the disappearance of a Dutch prospector in the Superstition mountains. The game is an adventure maze with more than one screen display. You move through the first maze picking up gold pieces (.) and searching for clues (?) to the hidden treasure of the mine. But you're constantly being chased by ghosts of lost prospectors. The game ends if one catches you.

When you touch the correct clue, you've found the secret treasure stash. You then find yourself in a new room. The treasure is in clear sight, but it's protected by the Dutchman's ghost. If you reach the treasure, you receive 10 times the amount of gold collected in the first maze.

```

100 REM  LOST DUTCHMAN MINE
110 GOSUB 710
120 SR=0:GL=2020
130 PRINT "[HOME][RIGHT 14]SCORE:"SR
140 JY= PEEK (56321):JY=15-(JY AND 15)
   :K= PEEK (197):J= PEEK (653):AL=L
150 POKE L,32: IF JY=1 OR JY=5 OR
   JY=9 OR (K=7 AND J=1) THEN
   AL=AL-40
160 IF JY=2 OR JY=6 OR JY=10 OR (K=7
   AND J=0) THEN AL=AL+40
170 IF (JY>3 AND JY<7) OR (K=2 AND
   J=1) THEN AL=AL-1
180 IF JY>7 OR (K=2 AND J=0) THEN
   AL=AL+1
190 IF PEEK (AL)<>160 THEN L=AL
200 IF PEEK (L)=G THEN 380
210 IF PEEK (L)=46 THEN SR=SR+1
220 IF PEEK (L)=CL THEN SR=SR+100
   : PRINT "[HOME]FIND ROOM"
   : FOR T=1 TO 999: NEXT
230 IF PEEK (L)=CL THEN PRINT "[HOME]
[SPACE 10]"
240 IF L=RL THEN 520
250 IF L=GL THEN 630
260 POKE L,I:LY= INT ((L-S)/40)
   :LX=L-S-LY*40
270 NG=3: IF RL=2020 THEN NG=1
280 FOR X=1 TO NG:M=G(X): POKE M,32
290 GY= INT ((M-S)/40):GX=M-S-GY*40
300 IF GX<LX AND ( PEEK (M+1)<50 OR
   PEEK (M+1)=I) THEN M=M+1
310 IF GX>LX AND ( PEEK (M-1)<50 OR
   PEEK (M-1)=I) THEN M=M-1
320 IF GY>LY AND ( PEEK (M-40)<50 OR
   PEEK (M-40)=I) THEN M=M-40
330 IF GY<LY AND ( PEEK (M+40)<50 OR
   PEEK (M+40)=I) THEN M=M+40
340 G(X)=M: POKE M,G: IF G=65 THEN
   G=88: GOTO 360
350 IF G=88 THEN G=65
360 IF M=L THEN 380


```

```

370 NEXT X: GOTO 130
380 SO=54272: FOR X=0 TO 24
   : POKE SO+X,0: NEXT X
390 POKE SO+14,5: POKE SO+18,16
   : POKE SO+3,1: POKE SO+24,143
400 POKE SO+6,240: POKE SO+4,65
410 POKE 53280,2:F1=5389
   : FOR X=0 TO 119:F=F1+ PEEK
     (SO+27)*3.5
420 H= INT (F/256):L=F-H*256
430 IF INT (X/10)/2= INT ( INT
     (X/10)/2) THEN POKE 53280,2
   : POKE 53281,2
440 IF INT (X/10)/2<> INT ( INT
     (X/10)/2) THEN POKE 53280,14
   : POKE 53281,1
450 POKE SO,L: POKE SO+1,H: NEXT X
460 POKE SO+24,0
470 FOR X=1 TO 2000: NEXT X
480 PRINT "[CLR]": PRINT : PRINT
490 PRINT "YOU LOSE!": PRINT "MY
     GHOST GOT YOU!"
500 PRINT : PRINT "HA HA HA HA HA!"
510 GOTO 680
520 PRINT "[CLR]": PRINT "[BLK]
     [RVSON][SPACE 18]"
530 FOR X=1 TO 12: PRINT " [RVSON]
     "; TAB( 18);"[RVSON] ": NEXT X
540 PRINT "[BLK] [RVSON][SPACE 18]"
550 S=1106:B=160:BX=49

```

MORE →



```

560 FOR X=0 TO BX: POKE S+40* INT (
      RND (8)*12)+ INT ( RND (8)*17),B
      : NEXT X
570 FOR X=55296 TO 56295: POKE X,0
      : NEXT X
580 GL=S+40* INT ( RND (8)*12)+ INT (
      RND (8)*17): POKE GL,164
590 G=88: FOR X=1 TO 3:G(X)=1513
      : NEXT X:L=S+8
600 FOR X=1 TO 3: POKE G(X),G: NEXT X
      : POKE L,I
610 RL=2020
620 FOR T=1 TO 2500: NEXT T: GOTO 130
630 FOR X=1 TO 2000: NEXT X
640 PRINT "[CLR]": PRINT : PRINT
650 PRINT TAB( 11)"YOU FOUND THE
      GOLD!": PRINT : PRINT
660 PRINT TAB( 11)"FINAL SCORE:"SR*10
670 PRINT : PRINT
680 POKE 198,0: PRINT TAB( 14)"PLAY
      AGAIN";: INPUT Q$
690 IF LEFT$(Q$,1)="Y" THEN 110
700 END
710 PRINT "[CLR][BLK]": POKE 53281,1
720 PRINT "[SPACE 5][RVSON]LOST
      DUTCHMAN MINE ADVENTURE": PRINT
730 PRINT "SOME TIME AGO, IN
      ARIZONA'S SUPERSTITION";
740 PRINT "MOUNTAINS, THE GOLD OF
      THE LOST"
750 PRINT "DUTCHMAN WAS LOST IN A
      CAVE.": PRINT
760 PRINT "AS YOU SEARCH THE CAVES
      BE SURE TO"
770 PRINT "AVOID THE GHOSTS (X). "
      : PRINT
780 PRINT "COLLECT DOTS FOR POINTS
      AND THE"
790 PRINT "MYSTERIOUS CLUES ([RVSON]?
      [RVSOFF]) TO HELP YOU ON[SPACE 5]
      THE WAY": PRINT
800 PRINT "USE A JOYSTICK OR THE
      CURSOR KEYS TO[SPACE 4]MOVE."
810 PRINT : PRINT "ENTER YOUR FIRST
      INITIAL."
820 GET I$: IF I$="" THEN 820
830 IF ASC (I$)<65 OR ASC (I$)>90
      THEN 820
840 PRINT "[CLR]": PRINT "[BLK]
      [RVSON][SPACE 37]"
850 FOR X=1 TO 19: PRINT " [RVSON]
      "; TAB( 37);"[RVSON] ": NEXT X
860 PRINT " [RVSON][SPACE 37]"

```

Alteration—Change the number of barriers by changing the value of **BX** in line 880.

Alteration—Change the number of gold pieces in the first maze by changing the value of **C** in line 880.

```
870 S=1106:B=160:G=146:CL=191
    :I=ASC(I$)-64
880 BX=249:C=99
890 FOR X=55296 TO 56295:POKE X,0
    :NEXT X
900 FOR X=0 TO BX:POKE S+40*INT(
    RND(8)*19)+INT(RND(8)*35),B
    :NEXT X
910 FOR X=0 TO C:POKE S+40*INT(
    RND(8)*19)+INT(RND(8)*35),46
    :NEXT X
920 FOR X=0 TO 3:RL=S+40*INT(RND
    (8)*19)+INT(RND(8)*35)
    :POKE RL,CL:NEXT X
930 G=88:G(1)=1428:G(2)=1458:G(3)=1764
    :L=S+18
940 FOR X=1 TO 3:POKE G(X),G:NEXT X
    :POKE L,I
950 RETURN
```

Biorhythm

Some experts say that our lives are governed by three regular cycles: physical (23 days), emotional (28 days) and intellectual (33 days). This fun program calculates your current status on each cycle, telling you about yourself today, yesterday or tomorrow!

First, you're instructed to enter any date for which you want your biorhythm calculated. Then you enter your date of birth. Months are spelled out. Days and years are entered as numbers.

The program calculates how many days you've been alive between the first date entered and your date of birth. Then it calculates your biorhythm and displays results for each cycle—physical, emotional and intellectual—in a bar chart. If you're in the first half of the cycle, you're in the up phase. The second half of the cycle is the down phase. Results are accurate to within one day for any two dates in the 20th Century.

```


100 REM  BIORHYTHM
110 DIM M$(12)
120 FOR X=1 TO 12: READ M$(X): NEXT X
130 PRINT "[CLR][WHITE]": PRINT "THE
      THEORY OF BIORHYTHMS STATES
      THAT"
140 PRINT "EACH PERSON HAS THREE
      PSYCHOLOGICAL[SPACE 5]CYCLES
      WHICH ARE RELATED TO"
150 PRINT "THEIR PHYSICAL, EMOTIONAL,
      AND[SPACE 10]INTELLECTUAL
      STATES."
160 PRINT : PRINT "THESE CYCLES
      ARE SAID TO BEGIN AT BIRTH."
170 PRINT "THE PHYSICAL IS 23 DAYS
      LONG."
180 PRINT "THE EMOTIONAL IS 28 DAYS
      LONG."
190 PRINT "THE INTELLECTUAL IS 33
      DAYS LONG.": PRINT
200 PRINT "THE FIRST HALF OF THE
      CYCLE IS SAID TO[SPACE 2]BE
      AN UP TIME AND THE SECOND ";
210 PRINT "HALF IS[SPACE 4]SAID
      TO BE A DOWN TIME."
220 PRINT "THE MID POINT IS CALLED
      A CRITICAL DAY."
230 FOR T=1 TO 1500: NEXT T: PRINT
      : PRINT
240 INPUT "ENTER BIRTH MONTH (SPELL
      IT)";M1$
250 FOR M1=1 TO 12: IF LEFT$(M1$,3)=
      LEFT$(M$(M1),3) THEN 270
260 NEXT M1: PRINT "NO SUCH MONTH"
      : GOTO 240
270 INPUT "ENTER BIRTH DAY";D1
      : IF D1>31 THEN 270
280 INPUT "ENTER BIRTH YEAR";Y1
290 PRINT : PRINT

```



```
300 INPUT "TODAY'S MONTH (SPELL  
IT)";M$  
310 FOR M=1 TO 12: IF LEFT$(M$,3)=  
LEFT$(M$(M),3) THEN 330  
320 NEXT M: PRINT "NO SUCH MONTH"  
: GOTO 300  
330 INPUT "TODAY'S DAY";D  
: IF D>31 THEN 330  
340 INPUT "TODAY'S YEAR";Y  
350 F=(Y-Y1)*365.25:H=(M-M1)*30.42  
:J=D-D1  
360 X= INT (F+H+J)-2  
370 P=(X/23- INT (X/23))*23  
380 E=(X/28- INT (X/28))*28  
390 I=(X/33- INT (X/33))*33  
400 PRINT "[CLR] BIORHYTHM FOR "M$";  
D;","Y: PRINT  
410 PRINT : PRINT "YOU HAVE BEEN  
ALIVE"X"DAYS.": PRINT
```

MORE →



Alteration—Change the graphics character in the bar chart from a square to any other character. Do this in lines 430, 450 and 470.

Alteration—Add a routine that interprets the results of the biorhythm calculation. For example, a statement might say IT LOOKS LIKE A GREAT DAY FOR YOU EMOTIONALLY AND INTELLECTUALLY. PHYSICALLY, YOU'RE IN TROUBLE!

```
420 PRINT "[RVSON]PHYSICAL CYCLE"  
    : PRINT "[SPACE 4]";  
430 FOR X=0 TO P: PRINT "[RVSON]  
    ";; NEXT X: PRINT "[LEFT 3]"  
    INT (P+.5)  
440 PRINT : PRINT : PRINT "[RVSON]  
    EMOTIONAL CYCLE": PRINT "[SPACE 4]"  
    ;  
450 FOR X=0 TO E: PRINT "[RVSON]  
    ";; NEXT X: PRINT "[LEFT 3]"  
    INT (E+.5)  
460 PRINT : PRINT : PRINT "[RVSON]  
    INTELLECTUAL CYCLE"  
    : PRINT "[SPACE 4]";  
470 FOR X=0 TO I: PRINT "[RVSON]  
    ";; NEXT X: PRINT "[LEFT 3]"  
    INT (I+.5)  
480 PRINT : PRINT : PRINT "HIT ANY  
    KEY FOR NEW BIORHYTHM"  
490 GET Q$: IF Q$="" THEN 490  
500 GOTO 290  
510 DATA JAN,FEB,MAR,APR,MAY,JUN,JUL,  
    AUG,SEP,OCT,NOV,DEC
```

Queen Of The Nile

Journey through Cleopatra's tomb in search of ancient treasure. Your time is limited to two minutes, and the dangers are many. But clever action will bring you rewards.

Control your progress through the burial maze by using a joystick or the cursor keys to move yourself (U) to the treasure (\$). By pressing the fire button or space bar, you can blast through walls with dynamite. But be careful because your supply of dynamite is limited. You may destroy treasure too!

If you run into a snake pit (S) or mummy (M), the game ends.

Alteration—Decrease or increase the amount of dynamite by changing the value of **D** in line 130.


Alteration—Increase the number of walls by changing line 160 to **IF RND(9)>.4 THEN POKE X,160: POKE X+CO,0**

```

100 REM  QUEEN OF THE NILE
110 FOR X=0 TO 8: READ L(X): NEXT X
120 GOSUB 530
130 PRINT "[CLR][WHITE]":CO=54272
    :UL=1884:UC=21:D=9:G=0:FL=0
140 FOR X=1064 TO 1103: POKE X,160
    : POKE X+CO,0: NEXT X
150 FOR X=1104 TO 1863: IF RND (8)>.8
    THEN POKE X,46: POKE X+CO,7
160 IF RND (7)>.5 THEN POKE X,160
    : POKE X+CO,0
170 NEXT X
180 FOR X=0 TO 9:M= INT ( RND
    (9)*600)+1144: POKE M,13
    : POKE M+CO,1: NEXT X

```

MORE →



```

190 FOR X=0 TO 9:S= INT ( RND
      (9)*600)+1144: POKE S,19
      : POKE S+CO,5: NEXT X
200 FOR J=1 TO 6
210 TR= INT ( RND (9)*480)+1144:F=0
      : FOR X=0 TO 8:A= PEEK (TR+L(X))
      : IF A=13 OR A=19 THEN F=1
220 NEXT X: IF F=1 THEN 210
230 TR(J)=TR: POKE TR,36
      : POKE TR+CO,7: NEXT J
240 TI$="000000"
250 K= PEEK (197):J= PEEK (653)
260 PRINT "[HOME]TIME:" MID$ (TI$,4,
      1):" RIGHT$ (TI$,2)"[SPACE 4]
      GOLD $" STR$ (G);
270 PRINT "[SPACE 2]DYNAMITE " STR$
      (D)+" "
280 JY= PEEK (56321):FI=JY AND 16
      :JY=15-(JY AND 15):TL=UL
      : POKE UL,32
290 IF UL>1064 AND (JY=1 OR JY=5 OR
      JY=9 OR (K=7 AND J=1)) THEN
      UL=UL-40: GOTO 330
300 IF UL<1904 AND (JY=2 OR JY=6 OR
      JY=10 OR (K=7 AND J=0)) THEN
      UL=UL+40: GOTO 330
310 IF UL>1064 AND ((JY>3 AND JY<7)
      OR (K=2 AND J=1)) THEN UL=UL-1
320 IF UL<1944 AND (JY>7 OR (K=2 AND
      J=0)) THEN UL=UL+1
330 P= PEEK (UL): IF P=160 THEN UL=TL
340 IF P=13 THEN E$="MY MUMMY GOT
      YOU!!!":FL=1: GOTO 470
350 IF P=19 THEN E$="YOU FELL INTO
      A SNAKE PIT!!!":FL=1: GOTO 470
360 IF P=46 THEN G=G+100
370 IF P=36 THEN G=G+ INT ( RND
      (8)*10)*1000+10000
380 IF TI>7200 THEN E$="TIME HAS
      EXPIRED": GOTO 470
390 IF RND (8)<.2 THEN M= INT ( RND
      (9)*600)+1144: IF PEEK (M)<>32
      THEN 390
400 POKE M,160: POKE M+CO,0
410 POKE UL,UC: POKE UL+CO,1
      : IF D>0 AND (K=60 OR K=15 OR
      FI<>16) THEN 430
420 GOTO 250
430 FOR E=0 TO 8: POKE UL+L(E),32
      : NEXT E:D=D-1
440 S=54272: FOR L=S TO S+24
      : POKE L,0: NEXT : POKE S+5,15
      : POKE S+6,15: POKE S+24,15

```


Alteration—Changing the equation for **G** in line 370 changes the value of the treasure. For example, **G=G+INT(RND(8)*20)*1000+20000** makes the treasure more valuable.

Alteration—Changing **TI** in line 380 to **TI>10800** allows a three-minute time limit.

```

450 POKE S+1,5: POKE S,235
      : POKE S+4,129: POKE S+4,128
      : POKE UL,UC: FOR T=0 TO 2000
      : NEXT T
460 GOTO 250
470 FOR T=1 TO 2000: NEXT T
      : POKE 198,0: PRINT "[CLR]"
      : PRINT : PRINT TAB( 20- LEN
        (E$)/2);E$
480 IF FL=1 THEN 500
490 PRINT : PRINT "[SPACE 2]YOU
      COLLECTED $"G"WORTH OF GOLD!"
      : PRINT
500 PRINT : PRINT TAB( 13)"PLAY
      AGAIN ";: INPUT Q$
510 IF LEFT$(Q$,1)="Y" THEN 130
520 END
530 POKE 53281,12: PRINT "[CLR]
      [WHITE]": PRINT TAB( 12)"QUEEN
      OF THE NILE"
540 PRINT : PRINT "USE JOYSTICK
      OR CURSOR KEYS TO MOVE."
550 PRINT "PRESS FIRE BUTTON OR
      SPACE BAR TO BLAST THROUGH
      WALLS.": PRINT
560 PRINT "YOU HAVE TWO MINUTES
      TO GATHER TREASURE."
570 PRINT "SYMBOLS:": PRINT
      : PRINT " [YEL]$( [WHITE] =
      TREASURE": PRINT
580 PRINT " [YEL].[WHITE] = GOLD
      PIECE": PRINT : PRINT " M =
      MUMMY": PRINT
590 PRINT " [GRN]$( [WHITE] = SNAKE
      PIT": PRINT : PRINT " [BLK]
      [RVSON] [RVSOFF][WHITE] = WALL"
      : PRINT
600 PRINT : PRINT "HIT ANY KEY TO
      START"
610 GET Q$: IF Q$="" THEN 610
620 RETURN
630 DATA -41,-40,-39,-1,0,1,39,40,41

```



Thin Ice


Don your wool cap and grab your pole and auger because it's time to do some ice fishing on Ted's pond. Due to some recent warm weather, there are spots of thin ice. Avoid them because unusual dangers lurk below. Get as many fish as you can before time runs out or you fall in. Hint: Pick up fish with your mouth.

```

100 REM THIN ICE
110 POKE 53281,1: POKE 53280,6
   : DIM F(16,20)
120 PRINT "[CLR][BLK]WELCOME TO
   TED'S POND": PRINT
130 PRINT "THE POND IS FROZEN OVER
   FOR THE FIRST"
140 PRINT "TIME THIS SEASON.": PRINT
150 PRINT "YOU ARE INVITED TO ICE
   FISH."
160 PRINT "BE CAREFUL THOUGH, THIS
   WARM SPELL"
170 PRINT "MIGHT HAVE CAUSED SOME
   SPOTS OF"
180 PRINT "[RVSON]THIN ICE[RVSOFF]. "
   : PRINT
190 PRINT "SKATE BY USING THE CURSOR
   KEYS.": PRINT
200 PRINT "YOU GET POINTS BY PICKING
   UP FROZEN"
210 PRINT "FISH (A) AND BY NOT
   GETTING INJURED."
220 PRINT : PRINT "HIT ANY KEY TO
   START."
230 GET Q$: IF Q$="" THEN 230
240 P$(1)="[BLK] W":P$(2)="[RED]<+>"
   :P$(3)="[BLK][RVSON]v[RVSOFF]
   [RVSON]c"
250 E$="[SPACE 3]"
260 V$="[HOME]":H$="[RIGHT]"
270 PRINT "[CLR]": FOR X=1 TO 15
   : PRINT " ";
280 FOR Y=1 TO 20:F(X,Y)=0
290 IF RND (8)<.3 THEN PRINT "[BLU]
   A";:F(X,Y)=1: GOTO 310
300 PRINT " ";
310 IF RND (8)<.01 THEN F(X,Y)=2
320 IF RND (8)<.004 THEN F(X,Y)=3
   : PRINT "[LEFT]*";
330 IF RND (8)<.001 AND F(X,Y)=1
   THEN F(X,Y)=4
340 NEXT Y: PRINT " ": NEXT X
350 PRINT V$: FOR I=1 TO 3
   : PRINT H$:P$(I): NEXT I
360 X=1:Y=2:S=0:TI$="000000"
370 GET Q$

```

Alteration—Change the graphics symbol in line 290 to create a different kind of fish.



Alteration—Add a splashing sound when the user falls through the ice, line 550.

```

380 PRINT "[HOME][DOWN 4][RIGHT 21]
      TIME : ";
390 PRINT MID$( TI$,4,1)
      : " RIGHT$( TI$,2)
400 PRINT V$: FOR I=1 TO 3
      : PRINT H$;E$: NEXT
410 IF Q$="[DOWN]" AND X<15 THEN
      V$=V$+Q$:X=X+1
420 IF Q$="[UP]" AND X>1 THEN V$=
      LEFT$( V$, LEN (V$)-1):X=X-1
430 IF Q$="[RIGHT]" AND Y<20 THEN
      H$=H$+Q$:Y=Y+1
440 IF Q$="[LEFT]" AND Y>1 THEN H$=
      LEFT$( H$, LEN (H$)-1):Y=Y-1
450 PRINT V$: FOR I=1 TO 3
      : PRINT H$;P$(I): NEXT I
460 F=F(X,Y): IF F=1 THEN S=S+10
      :F(X,Y)=0
470 IF F=2 THEN PRINT "[HOME]THE
      ICE NEARLY BROKE!":S=S+500
      : FOR T=1 TO 1000: NEXT T
480 IF F=2 THEN PRINT "[HOME]1500
      POINT BONUS.[SPACE 5]"
      : FOR T=1 TO 1000: NEXT T
490 IF F=2 THEN PRINT "[HOME][SPACE 17]
      ":F(X,Y)=0
500 IF F=4 THEN 630
510 IF F=3 THEN 550
520 PRINT "[HOME][DOWN 3][RIGHT 22]
      SCORE:" STR$( S)
530 IF TI>7200 THEN 590
540 GOTO 370
550 PRINT "[CLR]": POKE 53281,6
      : FOR T=1 TO 999: NEXT T
560 POKE 53281,1: PRINT : PRINT
570 PRINT "YOU JUST BECAME SHARK
      MEAL!"
580 S=0: GOTO 670
590 PRINT "[CLR]": PRINT : PRINT
600 PRINT "TIME HAS EXPIRED"
610 FOR T=1 TO 999: NEXT T
620 GOTO 670
630 PRINT "[CLR]": PRINT : PRINT
640 PRINT "YOU JUST CAUGHT A SCHOOL
      OF FISH!"
650 PRINT : PRINT "IT'S A MIRACLE!
      YOU WIN!"
660 S=S*3
670 PRINT : PRINT : PRINT
680 PRINT "FINAL SCORE="S
690 POKE 198,0
700 PRINT : INPUT "PLAY AGAIN";Q$
710 IF LEFT$( Q$,1)="Y" THEN 260

```

Bomb Squad

You're the expert working against the clock, trying to find a ticking bomb hidden in a million-room tower! You have 250 seconds to find the bomb. Each move takes exactly 10 seconds. To help, you have a bomb-sensitive device that gives a stronger readout as you approach the bomb.

You determine the location of a room by entering values from 0 to 99 for length, width and height. Your device then gives a readout that indicates how close you are to the bomb. 10,000 is the highest readout. A correct value for length, width or height is indicated by double asterisks in place of the number entered.

If you don't find the bomb in 250 seconds, it explodes. If you do find the bomb, you have an additional 100 seconds to defuse it. Guess the three-number combination in the same way you found the room. Enter values from 1-10.

Alteration—If you make the game easier or more difficult, be sure to change the instructions in lines 110 to 180. Also, to make the game less violent, change its name to **Treasure Hunt**. Change the statements that give the user instructions and commentary. The new statements should discuss finding a hidden treasure chest. When the chest is found, the user has to open the padlock.

Alteration—To make the game easier, try decreasing the number of rooms by changing **INT(RND(8)*100)** for variables **A**, **B** and **C** to **INT(RND(8)*50)** in line 220.

```

100 REM  BOMB SQUAD
110 PRINT "[CLR]YOU ARE ABOUT TO
      ACCEPT A DANGEROUS[SPACE 5]
      MISSION."
120 PRINT : PRINT "A BOMB IS HIDDEN
      IN ONE ROOM OF A[SPACE 7]MILLION
      ROOM TOWER."
130 PRINT : PRINT "ENTER ROOM NUMBER
      BY LENGTH, WIDTH, AND HEIGHT
      (0-99)."

```



```
270 PRINT "[SPACE 10]BOMB DETECTOR
[HOME]"
280 PRINT " L[SPACE 3]W[SPACE 3]H
[SPACE 3]TIME[SPACE 3]BEACON"
: PRINT
290 IF A=L THEN 330
300 PRINT "[HOME][DOWN 5]"
: INPUT "LENGTH";L
310 PRINT "[HOME][DOWN 6][SPACE 12]"
: IF L>99 OR L<0 THEN 300
320 PRINT "[HOME][DOWN 3]" STR$ (L)+"
"
330 IF B=W THEN 370
340 PRINT "[HOME][DOWN 5]"
: INPUT "WIDTH";W
350 PRINT "[HOME][DOWN 6][SPACE 12]"
: IF W>99 OR W<0 THEN 300
360 PRINT "[HOME][SPACE 4][DOWN 3]"
STR$ (W)+" "
370 IF C=H THEN 410
380 PRINT "[HOME][DOWN 5]"
: INPUT "HEIGHT";H
390 PRINT "[HOME][DOWN 6][SPACE 12]"
: IF H>99 OR H<0 THEN 300
400 PRINT "[HOME][SPACE 8][DOWN 3]"
STR$ (H)+" "
410 PRINT "[HOME][SPACE 12][DOWN 3]"S
```

MORE →




Alteration—To make the game easier, give yourself more time to find the bomb by changing **S** in line 510 to **S=S+10: IF S<300 THEN 290.**

```

420 PRINT "[HOME][DOWN 10]"; TAB( J) "
    "
430 BE=1E4-( ABS (A-L)+ ABS (B-W)+ ABS
    (C-H))*29
440 J= INT (BE/333.3333+.5)
    : PRINT "[HOME][DOWN 10]"; TAB(
    J)"^"
450 PRINT "[HOME][SPACE 19][DOWN 3]
    "BE
460 PRINT "[HOME][DOWN 3]";
470 IF A=L THEN PRINT TAB( 1)"[RVSON]
    **[RVSOFF]";
480 IF B=W THEN PRINT TAB( 5)"[RVSON]
    **[RVSOFF]";
490 IF C=H THEN PRINT TAB( 9)"[RVSON]
    **[RVSOFF]"
500 IF A=L AND B=W AND C=H THEN 550
510 S=S+10: IF S<260 THEN 290
520 GOSUB 900
530 PRINT "BOMB WAS AT": PRINT A;"
    "; " "; B; " "; C
540 END
550 PRINT "[CLR][DOWN 9][SPACE 14]
    [RVSON]BOMB LOCATED[RVSOFF]"
560 V= INT ( RND (8)*10)+1
    :N= INT ( RND (8)*10)+1
    :M= INT ( RND (8)*10)+1
570 FOR TT=1. TO 2000: NEXT
580 PRINT "[CLR]YOU HAVE LOCATED
    THE BOMB.": PRINT
590 PRINT "YOU MUST NOW DEFUSE IT
    ENTER THE CORRECTCOMBINATION."
    : PRINT
600 PRINT "EACH PART IS FROM 1 TO
    10. A CORRECT[SPACE 4]NUMBER
    IS DICTATED BY A [RVSON]**
    [RVSOFF]."

```

```
720 PRINT "[HOME][SPACE 8][DOWN 3]"B
730 IF C=M THEN 770
740 PRINT "[HOME][DOWN 5]"
    : INPUT "LEFT";C
750 PRINT "[HOME][DOWN 6][SPACE 12]"
    : IF C>10 OR C<1 THEN 740
760 PRINT "[HOME][SPACE 17][DOWN 3]"C
770 PRINT "[HOME][SPACE 25][DOWN 3]"S
780 PRINT "[HOME][DOWN 3]";
790 IF A=V THEN PRINT TAB( 1)"[RVSON]
    **[RVSOFF]";
800 IF B=N THEN PRINT TAB( 9)"[RVSON]
    **[RVSOFF]";
810 IF C=M THEN PRINT TAB( 18)"
    [RVSON]**[RVSOFF]"
820 IF A=V AND B=N AND C=M THEN 880
830 S=S+10: IF S<110 THEN 650
840 GOSUB 900
850 PRINT "THE COMBINATION WAS"
    : PRINT V;" ";N;" ";M: PRINT
860 PRINT "CONDOLENCES WILL BE SENT
    TO YOUR FAMILY."
870 END
880 PRINT : PRINT : PRINT "[SPACE 5]
    BOMB DEFUSED"
890 END
900 N=53281: POKE N-1,2: POKE 53281,7
    : PRINT "[CLR][WHITE][DOWN 10]
    [SPACE 13]***KABOOM***"
910 S=54272: FOR L=S TO S+24
    : POKE L,0: NEXT
920 POKE S+5,15: POKE S+6,15
    : POKE S+24,15
930 POKE S+1,5: POKE S,235
    : POKE S+4,129: POKE S+4,128
940 FOR X=1 TO 99: IF ( PEEK (N) AND
    15)=7 THEN POKE N,2: GOTO 960
950 POKE N,7
960 FOR T=1 TO 44: NEXT T,X
970 POKE N-1,14: POKE N,6
980 RETURN
```



Utopia

As ruler of an island nation, you're beset with important decisions. The goal is prosperity for you and your people. You try to increase national wealth, personal wealth, population and natural resources. The computer acts as your aid, giving you up-to-the-minute information and choices. Your reign can last up to 10 years, providing a revolution doesn't force a change.

Alterations—You can change the opening parameters by altering the values for the following variables in line 110: **P** is starting population; **M** is the treasury; **F** is the original number of fish; **T** is number of years.


```

100 REM  UTOPIA
110 P=5000:M=2000000:F=5000:S=10000
    :T=10:T=10
120 PRINT "[CLR][WHITE]HELLO, I AM
    MODEL 81882 OF THE"
130 PRINT "OMNISCIENT MICRO ANALYSIS
    RESOURCE"
140 PRINT "DIVISION. YOU MAY CALL
    ME OMAR.": PRINT
150 PRINT "PLEASE IDENTIFY YOURSELF
    (NAME)": INPUT N$
160 PRINT "[CLR]WELCOME TO UTOPIA,
    "N$".": PRINT
170 PRINT "AS BENIGN RULER OF UTOPIA,
    YOU MUST"
180 PRINT "GUIDE YOUR CITIZENS
    THROUGH THE"
185 PRINT "DECISIONS OF STATE."
    : PRINT
190 PRINT "AS YOUR CONSULTING AID,
    I WILL PROVIDE[SPACE 2]YOU WITH
    DATA.": PRINT
200 PRINT "YOU CAN GET A STATUS
    REPORT BY HITTING[SPACE 3]S
    DURING THE GAME.": PRINT
210 PRINT "UTOPIA IS AN ISLAND THAT
    IS LOCATED IN[SPACE 2]THE
    TROPICS.": PRINT
220 PRINT "YOU WILL RULE A MAXIMUM
    OF TEN YEARS."
230 PRINT "YOU MUST CONTROL THE USE
    OF NATURAL[SPACE 5]RESOURCES
    WELL.": PRINT
240 PRINT "GOOD LUCK YOUR HIGHNESS."
250 PRINT "MAY THE NAME OF "N$" BE
    A BLESSING ON OUR ISLAND."
260 PRINT : PRINT "HIT ANY KEY TO
    CONTINUE."
270 GET Q$: IF Q$="" THEN 270
280 GOSUB 1090
290 FOR T=10 TO 0 STEP -1
300 INPUT "WHAT IS YOUR SALARY THIS
    YEAR"; X
310 M=M-X:R=R+X: PRINT
    : PRINT "YOUR PERSONAL WEALTH
    IS NOW $"R

```

```
320 IF R>=(( RND (8)*3)+1)*M THEN
  GOSUB 1160
330 PRINT : PRINT "[RVSON]REPORT
  [RVSOFF] FROM MINISTER OF FISH"
  : PRINT "EMMY YELLOWTAIL."
340 PRINT : PRINT "YOUR HIGHNESS
  THIS IS IT:" : PRINT
350 A= INT ( RND (3)*3)+1
360 IF A=1 THEN PRINT "ALL IS WELL.
  THE FISH THRIVE. THERE IS[SPACE
  3]PLENTY TO FEED ALL."
370 IF A=1 THEN F=F+2000:P=P+5000
  :M=M+10000
380 IF A=2 THEN PRINT "WE NEED MORE
  BOATS. THEY COST $10000[SPACE 4]
  EACH."
390 IF A=2 THEN PRINT "HOW MANY WILL
  YOU BUY";: GOSUB 890:F=500*X
400 IF A=3 THEN PRINT "THE SEALS
  ARE EATING MOST OF THE FISH."
  : GOSUB 910
410 IF P<1000 THEN GOSUB 1160
420 PRINT : PRINT "HIT ANY KEY TO
  CONTINUE"
430 GET Q$: IF Q$="" THEN 430
440 PRINT "[CLR]": GOSUB 1090
450 PRINT : PRINT "[RVSON]REPORT
  [RVSOFF] FROM MINISTER OF FARMS"
  : PRINT "STEVEN SOIL."
```

MORE →



```
460 X= INT ( RND (3)*3)+1
470 IF X=1 THEN GOSUB 620
480 IF X=2 THEN GOSUB 650
490 IF X=3 THEN GOSUB 700
500 IF P=>8000 AND M=>8000000 AND
    F>8000 AND R>1500000 THEN 540
510 PRINT : PRINT "HIT ANY KEY TO
    CONTINUE"
520 GET Q$: IF Q$="" THEN 520
530 PRINT "[CLR]": NEXT T
540 FOR X=1 TO 10: PRINT "[CLR]"
    : PRINT
550 FOR TD=1 TO 200: NEXT TD
560 PRINT TAB( 16)"[RVSON][SPACE 7]
    [RVSOFF]"
570 PRINT TAB( 16)"[RVSON]YOU WIN
    [RVSOFF]"
580 PRINT TAB( 16)"[RVSON][SPACE 7]
    [RVSOFF]"
590 FOR TD=1 TO 200: NEXT TD,X
600 PRINT : PRINT : PRINT N$" IS
    MADE RULER FOR LIFE!"
610 PRINT : PRINT : END
620 PRINT "ALL CROPS ARE DOING WELL."
    :P=P+2000:M=M+100000
630 FOR TD=1 TO 1000: NEXT TD
640 RETURN
650 PRINT "EXCESS SUGAR CANE HAS
    BEEN SOLD FOR[SPACE 5]1000000."
660 PRINT "WHAT PERCENTAGE SHOULD
    GO INTO YOUR"
670 INPUT "POCKET (0-100)";X
680 M=M+1E6-(1E6*X/100):R=R+1E6*X/100
690 RETURN
700 PRINT "HEAVY FLOODS FROM A
    TYPHOON HAVE WIPED[SPACE 2]OUT
    MANY CROPS."
710 PRINT "WHAT WILL YOU DO?": PRINT
720 PRINT "A) REPLANT NEW CROPS."
    : PRINT "B) LIVE OFF STORED GRAIN"
730 PRINT "C) BUY CROPS FROM ABROAD."
    : PRINT
740 INPUT C$: IF C$<>"A" AND C$<>"B"
    AND C$<>"C" THEN 740
750 IF C$="A" THEN M=M- INT (200000*
    RND (8)):P=P- INT (100* RND (8))
760 IF C$="B" THEN P=P- INT (2000*
    RND (8)):M=M- INT ( RND (8)*10000)
770 IF C$="C" THEN M=M- INT ( RND
    (3)*400000)
780 IF M<0 THEN M=0: GOSUB 1160
790 IF P<100 THEN P=100
800 RETURN
810 FOR X=1 TO 5: PRINT "[CLR]"
```

```
820 FOR TD=1 TO 200: NEXT TD
830 PRINT : PRINT TAB( 10)"DEATH
      TO "N$""!!!"
840 FOR TD=1 TO 200: NEXT TD,X
850 I=I+ INT ( RND (4)*P/2)
      :R= INT (R* RND (3))
860 IF I>=P*.8 THEN PRINT "YOU HAVE
      BEEN OVERTHROWN."
      : PRINT "YOU LOSE.": STOP
870 IF RND (8)<.7 THEN PRINT "YOU
      HAVE MAINTAINED POWER.": PRINT
      : RETURN
880 PRINT "YOU HAVE RULED CRUELLY
      AND ARE BANISHED FROM UTOPIA."
      : STOP
890 INPUT X: IF X=0 THEN GOSUB 1160
900 M=M-X*10000: RETURN
910 PRINT : PRINT "WHAT WILL YOU
      DO?": PRINT
920 PRINT "A) KILL THE SEALS"
      : PRINT "B) GIVE UP FISHING"
930 PRINT "C) LET NATURE HANDLE IT"
      : PRINT "D) PANIC"
940 PRINT : INPUT A$: IF A$<>"A" AND
      A$<>"B" AND A$<>"C" AND A$<>"D"
      THEN 940
950 IF A$="D" THEN 910
960 IF A$="A" THEN I=I+ INT (1000*
      RND (2))
970 IF A$="C" THEN I=I+ INT (500* RND
      (7)):F=F- INT ( RND (4)*3000)
980 IF F<0 THEN F=0
990 IF A$<>"B" THEN RETURN
1000 PRINT : PRINT "HOW WILL YOU
      FEED THE PEOPLE?"
1010 PRINT : PRINT "A) LET THEM EAT
      CAKE.": PRINT "B) BUY THEM
      IMPORTED FOOD."
1020 PRINT "C) GROW MORE FOOD."
      : PRINT
1030 INPUT B$: IF B$<>"A" AND B$<>"B"
      AND B$<>"C" THEN 1030
1040 IF B$="A" THEN GOSUB 1160
1050 IF B$="B" THEN M=M-P*250
1060 IF B$="C" THEN M=M-P*200
      :P= INT (P*.95)
1070 IF M<0 THEN M=0: GOSUB 1160
1080 RETURN
1090 PRINT "[CLR][RVSON]STATUS REPORT
      : ISLAND OF UTOPIA[RVSOFF]"
      : PRINT
1100 PRINT : PRINT "POPULATION:",P
      : PRINT "TREASURY: ",M
```

MORE →

```
1110 PRINT "FISH:[SPACE 5]",F
      : PRINT "PERSONAL WEALTH:",R
1120 PRINT "TIME LEFT:",T"YEARS"
      : PRINT "INSURGENTS:",I
1130 PRINT : PRINT "HIT ANY KEY TO
      CONTINUE."
1140 GET Q$: IF Q$="" THEN 1140
1150 PRINT "[CLR]": RETURN
1160 PRINT "[CLR]THE INSURGENTS LED
      BY THE GREAT"
1170 PRINT "CATHY COURAGEOUS HAVE
      REBELLED."
1180 PRINT : PRINT "WHAT WILL YOU
      DO?": PRINT
1190 PRINT "A) GO INTO EXILE."
      : PRINT "B) TURN OVER ALL CASH"
      : PRINT "C) RESIST"
1200 PRINT : INPUT D$: IF D$<>"A" AND
      D$<>"B" AND D$<>"C" THEN 1200
1210 IF D$="A" THEN 1260
1220 IF D$="B" THEN M=M+R:K=R:R=0
      : IF K<.1*R THEN GOSUB 810
1230 IF D$="C" THEN GOSUB 810
1240 I=I+ INT ( RND (8)*P/5)
1250 RETURN
1260 IF A$="A" THEN PRINT "THE SEALS
      BLOCK YOUR ESCAPE.": STOP
1270 PRINT "YOU ESCAPE BUT YOU STILL
      LOSE THE GAME."
```


Tic-Tac-Toe


This is a classic game of strategy for two players. The computer provides the playing board and acts as master of ceremonies. It checks the progress of the game after each move and either announces the winner or determines whether the game is a tie.

```

100 REM TIC-TAC-TOE
110 DIM M(9),K(9),T(7,2)
120 FOR W=1 TO 9: READ M(W): NEXT W
130 FOR W=0 TO 7: FOR J=0 TO 2
: READ T(W,J): NEXT J,W
140 FOR W=1 TO 9:K(W)=0: NEXT W
150 DATA 1319,1323,1327,1479,1483,
1487,1639,1643,1647
160 DATA 1,2,3,4,5,6,7,8,9,1,4,7,2,5,
8,3,6,9,1,5,9,3,5,7
170 PRINT "[CLR][WHITE]"
: PRINT "HELLO. YOU ARE ABOUT
TO PLAY TIC-TAC-TOE": PRINT
180 INPUT "WHO ARE YOU";A$: PRINT
: PRINT A$, "YOU MAY PLAY WITH
A FRIEND."
190 PRINT "I WILL ANNOUNCE THE GAME."
: PRINT : PRINT "WHO DO YOU
WANT TO PLAY WITH";
200 INPUT B$: PRINT "[CLR]"
210 PRINT A$ AND "B$", "YOU CAN
TAKE"
220 PRINT "TURNS BY ENTERING X OR
O IN THE PLAYING GRID.": PRINT
230 PRINT "EACH PLAYER MAY ENTER
ONCE PER TURN.": PRINT
240 PRINT "THE PLAYER WHO GETS THREE
IN A ROW WINS.": PRINT
250 PRINT "A ROW MAY BE HORIZONTAL,
VERTICAL ORSPACE 3DIAGONAL."
260 PRINT : PRINT "THE FIRST PLAYER
USES AN X."
270 PRINT : PRINT "THE SECOND PLAYER
USES AN O.": PRINT : PRINT

```

MORE →



```
280 PRINT : PRINT "<<<<<[SPACE 2]
PRESS ANY KEY TO CONTINUE[SPACE 2]
>>>>>"
290 GET Q$: IF Q$="" THEN 290
300 PRINT "[CLR]": PRINT "WHEN YOU
SEE THE SCREEN, YOU WILL SEE"
310 PRINT "NUMBERS IN THE BOXES
WHERE THE LETTERS[SPACE 2]WILL
GO.": PRINT
320 PRINT "TO PLACE YOUR LETTER,
SIMPLY TYPE THE[SPACE 3]NUMBER
WHICH IS LOCATED ";
330 PRINT "WHERE YOU WANT[SPACE 2]
YOUR LETTER TO GO, AND PRESS
[RVSON]RETURN[RVSOFF]. "
340 PRINT : PRINT : PRINT "WHO WILL
GO FIRST, "A$" OR "B$";
350 INPUT Z$: IF Z$<>A$ AND Z$<>B$
THEN 340
360 IF Z$=A$ THEN F$=A$:S$=B$
: GOTO 380
370 S$=A$:F$=B$
380 PRINT "[CLR][DOWN 5]"
385 FOR X=55296 TO 56295: POKE X,1:
NEXT X
390 FOR W=1 TO 3: PRINT TAB( 14)"
[SPACE 3][RVSON] [RVSOFF][SPACE 3]
[RVSON] [RVSOFF]": NEXT W
400 PRINT TAB( 14)"[RVSON][SPACE 11]
[RVSOFF]"
410 FOR W=1 TO 3: PRINT TAB( 14)"
[SPACE 3][RVSON] [RVSOFF][SPACE 3]
[RVSON] [RVSOFF]": NEXT W
420 PRINT TAB( 14)"[RVSON][SPACE 11]
[RVSOFF]"
430 FOR W=1 TO 3: PRINT TAB( 14)"
[SPACE 3][RVSON] [RVSOFF][SPACE 3]
[RVSON] [RVSOFF]": NEXT W
440 FOR W=1 TO 9: POKE M(W),W+48
: NEXT W
450 T=0:FL=0
460 PRINT "[HOME]"F$", ENTER YOUR
MOVE";: INPUT M$
470 PRINT "[HOME][SPACE 38]"
480 M= VAL (M$): IF M>9 OR K(M)<>0
THEN 460
490 POKE M(M),24:K(M)=M:T=T+1
: GOSUB 580: IF FL=1 THEN 550
500 PRINT "[HOME]"S$", ENTER YOUR
MOVE";: INPUT M$
510 PRINT "[HOME][SPACE 38]"
520 M= VAL (M$): IF M>9 OR K(M)<>0
THEN 500
```

```
530 POKE M(M),15:K(M)=M:T=T+1
   : GOSUB 580: IF FL=1 THEN 550
540 GOTO 460
550 PRINT "[HOME]"W$: PRINT
   : INPUT "PLAY AGAIN";Q$
560 IF LEFT$(Q$,1)="Y" THEN 140
570 END
580 FOR W=0 TO 7:A= PEEK (M(T(W,0)))
   :B= PEEK (M(T(W,1)))
   :C= PEEK (M(T(W,2)))
590 IF A=24 AND B=24 AND C=24 THEN
   W$=F$+" WINS!":FL=1: GOSUB 630
600 IF A=15 AND B=15 AND C=15 THEN
   W$=S$+" WINS!":FL=1: GOSUB 630
610 IF FL=0 AND T=9 THEN FL=1
   :W$="TIE GAME!"
620 NEXT W: RETURN
630 FOR I=0 TO 2: POKE M(T(W,I)),
   PEEK (M(T(W,I)))+128: NEXT I
   : RETURN
```

Alteration—Add music when the winner is announced by extending the subroutine beginning at line 630.

Dice Racers

Here is a dice-simulation game using a computer-generated pair of dice. Two players take turns "rolling" the dice. Each player uses the die values to move two cars toward a finish line. The first player with both cars over the line wins.

Alteration—Use the dice routine for any board game requiring dice. Use lines 110, 490-510 and 800-820.

```

100 REM DICE RACERS
110 FOR X=1 TO 6: FOR Y=1 TO 3
    : READ K$(X,Y): NEXT Y,X
120 POKE 53281,0: PRINT "[CLR][WHITE]
    "; TAB( 16)"[RVSON]DICE RACE"
    : PRINT
130 PRINT "THIS GAME IS FOR TWO
    PLAYERS.": PRINT
140 INPUT "PLAYER ONE'S NAME";N1$
150 INPUT "PLAYER TWO'S NAME";N2$
    : PRINT
160 PRINT "YOU EACH HAVE TWO CARS
    TO DRIVE BY"
170 PRINT "MOVING THEM ACCORDING
    TO THE THROW"
180 PRINT "OF THE DICE. YOU CAN
    MOVE ONE CAR THE"
190 PRINT "VALUE OF BOTH DICE OR
    SPLIT THE DICE"
200 PRINT "AND MOVE BOTH CARS."
    : PRINT
210 PRINT "YOU CANNOT MOVE MORE
    THAN THE VALUE OF[SPACE 2]THE
    DICE.": PRINT
220 PRINT "THE FIRST PLAYER TO HAVE
    BOTH CARS"
230 PRINT "(H) ACROSS THE FINISH
    LINE WINS.": PRINT
240 PRINT "AFTER THE DICE ROLL,
    THE COMPUTER WILL"
250 PRINT "ASK YOU HOW MUCH YOU
    WANT TO MOVE CAR"
260 PRINT "ONE AND THEN HOW MUCH
    YOU WANT TO MOVE[SPACE 2]CAR
    TWO."
270 PRINT "BE SURE TO ENTER A ZERO
    IF YOU DO NOT[SPACE 3]WANT
    TO MOVE A CAR."
280 PRINT : PRINT "HIT ANY KEY TO
    START"
290 GET Q$: IF Q$="" THEN 290
300 R1$="[RVSON][MED.GRAY][SPACE 6]
    [WHITE] [MED.GRAY][SPACE 6]"
310 R2$="[RVSON][MED.GRAY][SPACE 13]"
320 PRINT "[CLR]" TAB( 14);R1$
330 PRINT TAB( 14)"[RVSON][MED.GRAY]
    *F-I-N-I-S-H*"

```


```

340 FOR X=1 TO 10: PRINT TAB( 14);R1$
      : PRINT TAB( 14);R2$: NEXT X
350 IF LEN (N1$)>6 THEN N1$= LEFT$
      (N1$,6)
360 IF LEN (N2$)>6 THEN N2$= LEFT$
      (N2$,6)
370 PRINT "[WHITE]" TAB( 14)N1$;
      TAB( 21)N2$"[HOME]"
380 FOR X=1 TO 2: FOR Y=1 TO 2
      :D$(X,Y)="[HOME][DOWN 21]"
      : NEXT Y,X
390 J$="[HOME][DOWN 15]"
400 R$(1,1)="[RIGHT 16]"
410 R$(1,2)=R$(1,1)+"[RIGHT 2]"
      :R$(2,1)=R$(1,1)+"[RIGHT 6]"
      :R$(2,2)=R$(2,1)+"[RIGHT 2]"
420 P=1:C=1
430 FOR X=1 TO 2: FOR Y=1 TO 2
      : PRINT D$(X,Y);R$(X,Y);"[RVSON]"
      [MED.GRAY]H": NEXT Y,X
440 F2=0: IF LEN (D$(2,1))<2 AND LEN
      (D$(2,2))<2 THEN F2=1
450 F1=0: IF LEN (D$(1,1))<2 AND LEN
      (D$(1,2))<2 THEN F1=1
460 IF F1=1 OR F2=1 THEN 740
470 PRINT "[HOME]": PRINT "[WHITE]"
      PLAYER"P:D1= INT ( RND (8)*6)+1
      :D2= INT ( RND (4)*6)+1
480 PRINT : PRINT
490 FOR X=1 TO 3: PRINT "[SPACE 2]"
      "K$(D1,X): NEXT X: PRINT : PRINT
500 FOR X=1 TO 3: PRINT "[SPACE 2]"
      "K$(D2,X): NEXT X: PRINT : PRINT
510 D=D1+D2
520 PRINT J$; "[WHITE]HOW MANY"
      : PRINT "CAR # 1"

```

Alteration—Change the program so players have to hit the finish line exactly. If a car goes too far, it must start over.

MORE →




```
530 INPUT Q$
540 Q= VAL (Q$): IF Q>D THEN Q=D
550 IF Q=0 THEN 600
560 PRINT D$(P,1);R$(P,1);"[RVSON]
[MED.GRAY] "
570 D$=D$(P,1): IF LEN (D$)<Q+1 THEN
Q= LEN (D$)-1
580 D$= LEFT$ (D$, LEN (D$)-Q)
:D$(P,1)=D$
590 PRINT D$(P,1);R$(P,1);"[RVSON]
[MED.GRAY]H"
600 D=D-Q
610 PRINT J$;: FOR X=1 TO 3
: PRINT "[SPACE 7]": NEXT X
620 PRINT J$;"[WHITE]HOW MANY"
: PRINT "CAR # 2"
630 INPUT Q$
640 Q= VAL (Q$): IF Q>D THEN Q=D
650 IF Q=0 THEN 700
660 PRINT D$(P,2);R$(P,2);"[RVSON]
[MED.GRAY] "
670 D$=D$(P,2): IF LEN (D$)<Q+1 THEN
Q= LEN (D$)-1
680 D$= LEFT$ (D$, LEN (D$)-Q)
:D$(P,2)=D$
690 PRINT D$(P,2);R$(P,2);"[RVSON]
[MED.GRAY]H"
700 FOR T=0 TO 999: NEXT T
710 PRINT "[HOME]": FOR X=1 TO 18
: PRINT "[SPACE 8]": NEXT X
720 IF P=1 THEN P=2: GOTO 470
730 P=1: GOTO 440
740 PRINT "[CLR][WHITE]"
750 IF F1=1 AND F2=1 THEN PRINT N1$"
AND "N2$" TIED!": GOTO 780
760 IF F1=1 THEN PRINT N1$" WINS!"
770 IF F2=1 THEN PRINT N2$" WINS!"
780 PRINT : PRINT : INPUT "PLAY
AGAIN";Q$
790 IF LEFT$ (Q$,1)="Y" THEN 120
800 DATA "[RVSON][SPACE 3]","[RVSON]
Q ","[RVSON][SPACE 3]","[RVSON]Q
[SPACE 2]","[RVSON][SPACE 3]","
[RVSON][SPACE 2]Q"
810 DATA "[RVSON]Q[SPACE 2]","[RVSON]
Q ","[RVSON][SPACE 2]Q","[RVSON]
Q Q","[RVSON][SPACE 3]","[RVSON]Q
Q"
820 DATA "[RVSON]Q Q","[RVSON] Q
","[RVSON]Q Q","[RVSON]QQQ","
[RVSON][SPACE 3]","[RVSON]QQQ"
```

Alien

Here's your chance to be the alien! From your space capsule, watch Earth come into view. Choose your options carefully. Will you invade or come in peace?

```
100 REM ALIEN
110 POKE 53281,0
120 PRINT "[CLR][WHITE]COMMUNICATION
:": PRINT : PRINT
130 PRINT TAB( 4)"[BLU][RVSON][SPACE 24]
"
140 PRINT TAB( 4)"[BLU][RVSON]
[RVSOFF][WHITE][SPACE 2].[SPACE 6]
.[SPACE 6].[SPACE 3].[BLU]
[RVSON] "
150 PRINT TAB( 4)"[BLU][RVSON]
[RVSOFF][WHITE][SPACE 4].[SPACE 9]
.[SPACE 3].[SPACE 3][BLU][RVSON]
[RVSOFF][WHITE][SPACE 2]ON
BOARD"
160 PRINT TAB( 4)"[BLU][RVSON]
[RVSOFF][WHITE] .[SPACE 7][RVSON]
v[SPACE 2]c[RVSOFF][SPACE 7].
[BLU][RVSON] [RVSOFF][WHITE]
[SPACE 2]SYSTEMS:"
```

MORE →



```
170 PRINT TAB( 4)"[BLU][RVSON]
   [RVSOFF][SPACE 7][LT.BLU][RVSON]v
   [RVSOFF]f[SPACE 2]±[RVSON]
   c[RVSOFF][SPACE 7][BLU][RVSON]
   "
180 PRINT TAB( 4)"[BLU][RVSON]
   [RVSOFF][WHITE][SPACE 2].[SPACE 3]
   [LT.BLU][RVSON]k[SPACE 2][RVSOFF]
   [WHITE]++[LT.BLU] bd[WHITE]±
   [LT.BLU]k .[SPACE 3].[BLU][RVSON]
   "
190 PRINT TAB( 4)"[BLU][RVSON]
   [RVSOFF][SPACE 6][LT.BLU][RVSON]
   [WHITE]++[LT.BLU]i[RVSOFF]
   if[RVSON]v[WHITE]++[RVSOFF]
   [SPACE 6][BLU][RVSON] "
200 PRINT TAB( 4)"[BLU][RVSON]
   [RVSOFF][WHITE][SPACE 2]..[SPACE 2]
   [LT.BLU][RVSON]k [WHITE]++
   [RVSOFF]±[LT.BLU]c[RVSON] [WHITE]
   ±[LT.BLU] [RVSOFF]k[WHITE][SPACE 3]
   .[SPACE 2][BLU][RVSON] "
210 PRINT TAB( 4)"[BLU][RVSON]
   [RVSOFF][SPACE 7][LT.BLU][RVSON]f
   [SPACE 2][RVSOFF] [WHITE]±
   [LT.BLU]d[RVSON] d[RVSOFF][SPACE 7]
   [BLU][RVSON] "
220 PRINT TAB( 4)"[BLU][RVSON]
   [RVSOFF][WHITE] .[SPACE 2].
   .[SPACE 2][LT.BLU][RVSON]i
   [WHITE]++[LT.BLU]i[RVSOFF][WHITE]
   [SPACE 4].[SPACE 4][BLU][RVSON]
   "
230 PRINT TAB( 4)"[BLU][RVSON]
   [RVSOFF][WHITE][SPACE 3].[SPACE 3]
   .[SPACE 4].[SPACE 4].[SPACE 2].
   [BLU][RVSON] "
240 PRINT TAB( 4)"[BLU][RVSON]
   [RVSOFF][WHITE][SPACE 2]. .
   [SPACE 2].[SPACE 2].[SPACE 5].
   [SPACE 5][BLU][RVSON] "
250 PRINT TAB( 4)"[BLU][RVSON][SPACE 24]
   ": PRINT
260 PRINT "[WHITE]ROCKET ORIENTATION
   ": PRINT
270 PRINT " [RVSON][SPACE 3]f[RVSOFF]
   [SPACE 2]SPEED:"
280 PRINT " [RVSON] [RED] [WHITE]
   U[RVSOFF]"
290 PRINT " [RVSON] [RED] [WHITE]
   E[RVSOFF][SPACE 2]WARP DRIVE:"
   : PRINT " [RVSON] [RED] [WHITE]
   L[RVSOFF]"
```




```

300 PRINT " [RVSON] [RED] [WHITE]
      [SPACE 2][RVSOFF][SPACE 2]ALERT
      STATUS:"
310 D1$="[HOME][DOWN 17]"
      :D2$=D1$+"[DOWN 2]"
      :D3$=D2$+"[DOWN 2]"
      :D4$=D3$+"[DOWN 2]"
320 R$="[RIGHT 20]":K$="[HOME][RIGHT 15]"
      "
330 S$="[HOME][DOWN 8]" + R$ + "[RIGHT 10]"
      "
340 B$="[SPACE 25]":B1$="[SPACE 9]"
350 PRINT "[HOME]"
360 PRINT D1$;R$;"EARTH"
370 PRINT D2$;R$;"SUBLIGHT"
380 PRINT D3$;R$;"OFF"
390 PRINT D4$;R$;"[GRN]GREEN"
400 PRINT S$;"[WHITE]EARTH ON"
      : PRINT S$;"[DOWN]MONITOR."
410 FOR T=1 TO 2000: NEXT T
420 PRINT K$;B$: PRINT S$;B1$
      : PRINT S$;"[DOWN]";B1$
430 PRINT S$;"A) INVADE"
440 PRINT S$;"[DOWN]B) FLEE"
450 PRINT S$;"[DOWN 2]C) PEACE"
460 PRINT K$;"WHY DO YOU COME
      (LETTER)?"
470 GET C$: IF C$="" THEN 470
480 IF C$="A" THEN 520
490 IF C$="B" THEN 990
500 IF C$="C" THEN 1560
510 GOTO 470
520 PRINT K$;B$: PRINT K$;"WOULD
      YOU RECONSIDER ?"
530 PRINT S$;B1$: PRINT S$;"[DOWN]";
      B1$: PRINT S$;"[DOWN 2]";B1$
540 PRINT S$;"[RVSON]Y[RVSOFF] YES"
      : PRINT S$;"[DOWN][RVSON]N
      [RVSOFF] NO"
550 GET Q$: IF Q$="" THEN 550
560 IF Q$="Y" THEN 420
570 PRINT K$;"WE MUST DEFEND
      OURSELVES!"
580 SO=54272: FOR L=0 TO 24
      : POKE SO+L,0: NEXT L
      : POKE SO+14,5: POKE SO+18,16
590 POKE SO+3,1: POKE SO+24,143
      : POKE SO+6,240: POKE SO+4,65
600 FOR X=0 TO 99: PRINT D4$;R$;B$"
      [HOME]": POKE SO+24,143
610 IF INT (X/10)/2= INT ( INT
      (X/10)/2) THEN 630

```

MORE →

Alteration—Add sound when the earthlings shoot at you, line 650.

Alteration—Change the probability of being destroyed by changing line 730. For example, **IF RND(8) < .5 AND S > 3 THEN 960** will increase your chance of destruction.

Alteration—Change the probability of Earth surrendering by changing **A** in line 910. For example, **IF A > .9** will increase the probability that Earth would surrender.

```

620 PRINT D4$;R$;"[RVSON][RED][SPACE 2]
    [WHITE]RED ALERT![RED][SPACE 2]
    [WHITE][RVSOFF][HOME]"
    : POKE SO+24,0
630 POKE SO,23: POKE SO+1,43
640 NEXT X: POKE SO+24,0
650 PRINT K$;B$: PRINT K$;"THEY
    ARE FIRING!"
660 PRINT S$;B1$: PRINT S$;"[DOWN]";
    B1$: PRINT S$"[DOWN 2]";B1$
670 PRINT S$;"A) SHIELD"
    : PRINT S$;"[DOWN]B) FLEE"
680 GET Q$: IF Q$="" THEN 680
690 IF Q$="A" THEN 720
700 IF Q$="B" THEN 990
710 GOTO 680
720 S=S+1
730 IF RND (8)<.4 AND S>3 THEN 960
740 PRINT K$;B$: PRINT K$;"SHIELD
    INTACT"
750 FOR T=1 TO 1500: NEXT T
760 PRINT S$;B1$: PRINT S$;"[DOWN]";
    B1$
770 PRINT S$;"A) ATTACK"
    : PRINT S$;"[DOWN]B) FLEE"
    : PRINT S$;"[DOWN 2]C) PEACE"
780 PRINT K$;B$: PRINT K$;"WHAT
    NOW ?"
790 GET Q$: IF Q$="" THEN 790
800 IF Q$="A" THEN 840
810 IF Q$="B" THEN 990
820 IF Q$="C" THEN 1560
830 GOTO 790
840 PRINT K$;B$: PRINT K$;"WHICH
    CITY";: INPUT N$
850 PRINT K$;B$
860 A= RND (8)
870 IF A<.3 THEN PRINT K$;"ATTACK
    FAILED": GOTO 890
880 PRINT K$;N$" DESTROYED!"
890 FOR T=1 TO 1500: NEXT T
900 PRINT K$;B$
910 IF A>.95 THEN PRINT K$"EARTH
    SURRENDERS! YOU WIN!": STOP
920 IF A>.05 THEN 570
930 PRINT "[HOME]I CANNOT LET YOU
    KILL INNOCENTS! YOUR"
940 PRINT "LIFE SUPPORT WAS TURNED
    OFF. YOU LOSE!"
950 STOP
960 GOSUB 1620: PRINT "[HOME]YOUR
    SHIELDS DID NOT HOLD OUT!"
970 PRINT TAB( 11);"*** YOU LOSE!
    ***"

```

```
980 STOP
990 PRINT S$;B1$; PRINT S$;"[DOWN]";
    B1$; PRINT S$;"[DOWN 2]";B1$
1000 PRINT K$;B$
1010 PRINT "[HOME][DOWN 4]"
1020 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE][SPACE 4].[SPACE 6]
    .[SPACE 5].[SPACE 2]. [BLU]
    [RVSON] "
1030 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE] .[SPACE 3].
    [SPACE 3]. .[SPACE 4]. .[SPACE 2]
    .[BLU][RVSON] "
1040 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE][SPACE 4].[SPACE 2]
    .[SPACE 2].[SPACE 4].[SPACE 2].
    . [BLU][RVSON] "
1050 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE] .[SPACE 6].
    [SPACE 5]... .[SPACE 3][BLU]
    [RVSON] "
1060 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE].[SPACE 2].
    [SPACE 7].[SPACE 3].[SPACE 3].
    [SPACE 2][BLU][RVSON] "
1070 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE][SPACE 2].[SPACE 3]
    .[SPACE 2].[SPACE 3].[SPACE 2].
    [SPACE 3]. [BLU][RVSON] "
1080 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE].[SPACE 6].
    [SPACE 3].[SPACE 7].[SPACE 2]
    [BLU][RVSON] "
1090 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE][SPACE 3].[SPACE 2]
    .[SPACE 7]. .[SPACE 3]..[BLU]
    [RVSON] "
1100 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE] ..[SPACE 2].
    [SPACE 2]. .[SPACE 2].[SPACE 3].
    .[SPACE 2][BLU][RVSON] "
1110 PRINT TAB( 4)"[BLU][RVSON]
    [RVSOFF][WHITE]. . . [SPACE 2].
    [SPACE 6].[SPACE 5][BLU][RVSON]
    "
1120 PRINT K$;"[WHITE]ENTER X
    COORDINATE";: INPUT X$
1130 X= VAL (X$)
1140 PRINT K$;B$
1150 PRINT K$;"ENTER Y COORDINATE";
    : INPUT Y$
1160 Y= VAL (Y$)
1170 PRINT K$;B$
```

MORE →

```

1180 PRINT K$;"PRESS 'W' TO WARP
      OUT"
1190 GET Q$: IF Q$<>"W" THEN 1190
1200 PRINT D1$;R$;"HYPERSPACE"
1210 PRINT D3$;R$;"ON[SPACE 2]"
1220 PRINT D4$;R$;"GREEN[SPACE 10]
      [HOME]"
1230 PRINT "[HOME][DOWN 3]"
      : FOR J=1 TO 11
1240 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE]*****
      *****[BLU][RVSON] "
1250 NEXT J
1260 FOR W=0 TO 9: FOR T=1 TO 200
      : NEXT T
1270 PRINT D2$;R$;"[WHITE]WARP "
      STR$ (W)+"[SPACE 5]"
1280 NEXT W
1290 FOR W=9 TO 0 STEP -1
      : FOR T=1 TO 200: NEXT T
1300 PRINT D2$;R$;"[WHITE]WARP "
      STR$ (W)+"[SPACE 5]"
1310 NEXT W
1320 IF INT ( RND (7)*20)=Y AND INT (
      RND (8)*20)=X THEN 1530
1330 PRINT "[HOME][DOWN 3]"
1340 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE][SPACE 2].[SPACE 6]
      .[SPACE 6].[SPACE 3]. [BLU]
      [RVSON] "
1350 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE][SPACE 4].[SPACE 9]
      .[SPACE 3].[SPACE 3][BLU][RVSON]
      [RVSOFF]"
1360 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE] .[SPACE 7][RED]
      [RVSON]v[SPACE 2]c[RVSOFF][SPACE 7]
      . [BLU][RVSON] [RVSOFF]"
1370 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][SPACE 7][RED][RVSON]v
      [RVSOFF][SPACE 3]u[RVSON] c
      [RVSOFF][SPACE 7][BLU][RVSON]
      "
1380 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE][SPACE 2].[SPACE 3]
      [RED][RVSON]k[SPACE 2][RVSOFF]
      [YEL]++[RED][SPACE 2]k[YEL]k
      [WHITE] .[SPACE 3].[BLU][RVSON]
      "
1390 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][SPACE 6][RED][RVSON]
      [YEL]u[RED]b [RVSOFF] ii[RVSON]v
      [YEL]++[RVSOFF][SPACE 6][BLU]
      [RVSON] "

```

Alteration—Change the appearance of the planet Omega by changing **PRINT** statements in lines 1340-1440.

```
1400 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE][SPACE 2]..
      [SPACE 2][RED][RVSON][SPACE 2]
      [YEL]++-[RVSOFF]++[RED] [RVSON]
      [YEL]++[RED][SPACE 2][RVSOFF]
      [WHITE][SPACE 3].[SPACE 2][BLU]
      [RVSON] "
1410 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE][SPACE 2].[SPACE 3]
      [RED][RVSON]k [YEL]+++[RED]
      [RVSON] [YEL]++[RED] [RVSOFF]k
      [WHITE][SPACE 3].[SPACE 2][BLU]
      [RVSON] "
1420 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][SPACE 7][RED][RVSON]f
      [SPACE 2][RVSOFF] [YEL]++[RED]v
      [RVSON] d[RVSOFF][SPACE 7][BLU]
      [RVSON] "
1430 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE] .[SPACE 2].
      .[SPACE 2][RED][RVSON]i [YEL]++
      [RED]i[RVSOFF][WHITE][SPACE 4].
      [SPACE 4][BLU][RVSON] "
1440 PRINT TAB( 4)"[BLU][RVSON]
      [RVSOFF][WHITE][SPACE 3].[SPACE 3]
      .[SPACE 4].[SPACE 4].[SPACE 2].
      [BLU][RVSON] "
1450 PRINT D1$;R$;"[WHITE]OMEGA
      SYSTEM"
1460 PRINT D2$;R$;"SUBLIGHT[SPACE 4]"
1470 PRINT D3$;R$;"OFF"
1480 PRINT D4$;R$;"[GRN]GREEN"
1490 PRINT S$;"[WHITE]OMEGA ON"
      : PRINT S$;"[DOWN]MONITOR."
1500 PRINT K$;B$
1510 PRINT K$;"WELCOME TO OMEGA
      5."
1520 PRINT "SAFE ESCAPE!": STOP
1530 GOSUB 1620
1540 PRINT : PRINT : PRINT "[WHITE]
      YOU HIT A STAR! YOU LOSE!"
1550 STOP
1560 PRINT K$;B$
1570 PRINT K$;"WELCOME TO PLANET
      EARTH."
1580 FOR X=1 TO 6
1590 PRINT K$;"[DOWN]";B$
      : FOR T=1 TO 200: NEXT T
1600 PRINT K$;"[DOWN]";"[RVSON]YOU
      WIN": FOR T=1 TO 200: NEXT T
1610 NEXT X: STOP
```

MORE →

```
1620 N=53281: POKE N-1,2: POKE 53281,7
      : PRINT "[CLR][WHITE][DOWN 8]
      [SPACE 12]***DESTROYED***"
1630 S=54272: FOR L=S TO S+24
      : POKE L,0: NEXT
1640 POKE S+5,15: POKE S+6,15
      : POKE S+24,15
1650 POKE S+1,5: POKE S,235
      : POKE S+4,129: POKE S+4,128
1660 FOR X=1 TO 99: IF ( PEEK (N)
      AND 15)=7 THEN POKE N,2
      : GOTO 1680
1670 POKE N,7
1680 FOR T=1 TO 44: NEXT T,X
1690 POKE N-1,14: POKE N,6
1700 RETURN
```

Hangman

Sharpen your word skills by playing this classic game with your computer. The computer will think of a word. You guess it, letter by letter. A correct letter will be inserted into the appropriate space. Each wrong letter adds to the hangman. You are allowed 12 wrong guesses before being hung. When the hangman is complete, the whole word is displayed. The computer gives you 10 words. Your score is based on the length of correctly identified words and the number of missed letters.

```

100 REM HANGMAN
110 POKE 53281,0: PRINT "[CLR][WHITE]
    THE COMPUTER THINKS OF A WORD.
    YOU TRY[SPACE 3]TO GUESS IT."
120 PRINT "TWELVE WRONG GUESSES
    BUILD THE HANGMAN."
130 PRINT : PRINT "HIT ANY KEY"
140 GET Q$: IF Q$="" THEN 140
150 SE=0: FOR PL=1 TO 10
160 PRINT : PRINT "[CLR]WORD #"PL
170 CT=0:NW=100
180 READ W$: IF CT>24 AND RND
    (8)<1/NW THEN RESTORE : GOTO 210
190 CT=CT+1: IF W$="END" THEN
    RESTORE : GOTO 170
200 GOTO 180
210 G$=" "
220 PRINT "[CLR]": PRINT : PRINT
    : PRINT : PRINT : PRINT : PRINT
225 FOR X=55296 TO 56295: POKE X,1:
    NEXT X
230 PRINT "[SPACE 10]iiiiii"
240 PRINT "[SPACE 10]_[SPACE 3][RVSON]
    "
250 PRINT "[SPACE 10]_[SPACE 3][RVSON]
    "
260 FOR A=1 TO 7: PRINT TAB( 14)"
    [RVSON] ": NEXT A
270 FOR A=1 TO 5: PRINT "[SPACE 4]
    [RVSON][SPACE 7]q*w[SPACE 5]"
    : NEXT A
280 PRINT "[HOME][RIGHT][DOWN]";
    : FOR X=1 TO LEN (W$): PRINT "-";
    : NEXT
290 C=0:I=0
300 PRINT "[HOME][DOWN 2]"
    : PRINT "ENTER A LETTER:"
310 POKE 198,0
320 GET L$: IF L$="" THEN 320
330 PRINT "[HOME][DOWN 3][SPACE 17]"
340 K=0: FOR A=1 TO LEN (G$)
    : IF L$= MID$ (G$,A,1) THEN K=1
350 NEXT A: IF K=1 THEN 300

```



MORE →

```

360 G$=G$+L$
370 PRINT "[HOME][DOWN 4]";G$
380 F=1: FOR X=1 TO LEN (W$)
390 IF L$= MID$ (W$,X,1) THEN PRINT "
[HOME]": PRINT TAB( X);L$;:F=0
:C=C+1
400 NEXT X
410 IF C= LEN (W$) THEN 470
420 IF F=0 THEN 300
430 I=I+1: READ M,L: POKE L,M
: POKE L+30720,2
440 IF I<12 THEN 300
450 PRINT "[HOME]": PRINT "YOU'RE
HUNG[SPACE 2]": PRINT "THE
WORD WAS:": PRINT W$
460 GOTO 490
470 PRINT "[HOME][RIGHT 5]
**SUCCESS**"
480 SE=SE+5* LEN (W$)-I: PRINT
: PRINT "YOUR SCORE ="SE
490 FOR T=1 TO 4000: NEXT
500 NEXT PL
510 RESTORE
520 PRINT "[CLR]": PRINT "YOUR FINAL
SCORE="SE: PRINT
530 INPUT "PLAY AGAIN (Y/N)";Q$
: IF LEFT$ (Q$,11)="Y" THEN 100
540 DATA 87,1434,102,1474,102,1514,
78,1473,77,1475
550 DATA 101,1513,103,1515,160,1554,
221,1594,221,1634,108,1633,123,
1635
560 DATA THE,AND,THAT,FOR,WITH,WAS,
HIS,NOT,BUT,HAVE,YOU,WHICH,ARE,
HER,HAD,FROM,THIS
570 DATA THEY,THEIR,SHE,HAS,WERE,
BEEN,HIM,ONE,SO,WILL,THERE,WHO,
WHEN,WHAT,YOUR,MORE
580 DATA WOULD,THEM,SOME,THAN,MAY,
UPON,ITS,OUT,INTO,OUR,THESE,MAN,
LIKE,SHALL,GREAT,NOW
590 DATA SUCH,SHOULD,OTHER,ONLY,ANY,
THEN,ABOUT,THOSE,CAN,MADE,WELL
600 DATA OLD,MUST,SAID,TIME,EVEN,NEW,
COULD,VERY,MUCH,OWN,MOST,MIGHT,
FIRST,AFTER,YET,TWO
610 DATA END

```

Alteration—A simple vocabulary of commonly used English words is in **DATA** statements 560 to 600. You can replace or extend this vocabulary by adding or deleting **DATA** statements. The last word in the last **DATA** statement must be **END**. It is used as a marker to identify the end of the list.

Depending on the average length of the words, you can store a few thousand words in a Commodore 64. If you increase the list to more than 100 words, increase the value of **NW** in line 170 to equal the number of words in the list.

Alteration—Design the word list for a certain user. For example, make a list of foreign words for the person learning a foreign language. Or, make a list of simple words for a child learning how to spell.

Dr. Sigmund


Find out more about yourself by having the great psychiatrist Dr. Sigmund analyze you. Once on the couch, you are shown computer-generated ink blots. You enter your first impression about what you see and get instantly analyzed in the classic fashion. You get 14 analyses per visit. There are no scores and no losers. This is a great game for a party.

Alteration—Change the ink-blot patterns by changing the graphics symbols in string variable **K\$**, line 110. To simplify the pattern, add more spaces to **K\$**.

```

100 REM SIGMUND
110 C$="[BLK][WHITE][RED][CYN][PUR]
    [GRN][BLU][YEL]":K$="[pound]
    Qi=k": POKE 53280,6: POKE 53281,0
120 PRINT "[CLR]": PRINT "HELLO,
    I AM ANNA, SECRETARY TO DR."
130 PRINT "SIGMUND. IS THIS YOUR
    FIRST ";
140 INPUT "VISIT";A$: PRINT
    : IF LEFT$(A$,1)="N" THEN 270
150 INPUT "WHAT IS YOUR NAME";B$
    : PRINT
160 PRINT "YOU ARE IN LUCK "B$"."
    : PRINT "DR. SIGMUND WILL SEE
    YOU NOW."
170 FOR T=1 TO 4000: NEXT T
    : PRINT "[CLR]"
180 PRINT "PLEASE, MAKE YOURSELF
    COMFORTABLE ON[SPACE 4]THE
    COUCH.": PRINT
190 PRINT "I SEE THAT YOU ARE A
    LITTLE BIT": PRINT "NERVOUS,
    "B$","
200 PRINT "NOW TRY TO RELAX."
    : FOR T=1 TO 4000: NEXT T: PRINT
    MORE ->

```



Alteration—Change the number of questions, variable **S**, in line 260.

Alteration—Change any word from the analyst's list and the corresponding interpretation. For example, change **SPIDER** in lines 320 and 360 to a friend's name and enter something about that person in lines 520 and 530. Customize the program to suit your purposes.

```
210 PRINT "I HAVE SOME PRETTY
    PICTURES TO SHOW YOU."
220 PRINT "JUST LET YOUR MIND GO
    TO REST. LOOK AT[SPACE 2]THE
    PICTURES AND PICK FROM"
230 PRINT "THE LIST WHAT"
    : PRINT "IT REMINDS YOU OF.
    IF SOMETHING"
240 PRINT "ELSE, JUST ENTER THAT."
    : FOR T=1 TO 7000: NEXT T:S=0
250 FOR T=1 TO 5000: NEXT T
260 S=S+1: IF S=14 THEN 740
270 PRINT "[CLR]OK, HERE IS A PICTURE
    :": PRINT
280 FOR X=0 TO 9: PRINT TAB( 6);
    : FOR Y=1 TO 20:C= INT ( RND
    (7)*8+1): PRINT MID$( C$,C,1);
290 K= INT ( RND (8)*6+1)
    : PRINT MID$( K$,K,1);
300 NEXT Y: PRINT " ": NEXT X
310 PRINT
320 PRINT "[WHITE][SPACE 3]SPIDER
    [SPACE 3]MOTHER[SPACE 6]CAT"
330 PRINT "[SPACE 3]SNAKE[SPACE 4]
    BUTTERFLY[SPACE 3]ROPE"
340 PRINT "[SPACE 3]FATHER[SPACE 3]
    REINDEER[SPACE 4]FOOD"
350 PRINT : INPUT O$: PRINT "[CLR]"
360 IF O$="SPIDER" THEN 520
370 IF O$="MOTHER" THEN 540
380 IF O$="CAT" THEN 560
390 IF O$="SNAKE" THEN 580
400 IF O$="BUTTERFLY" THEN 600
410 IF O$="ROPE" THEN 620
420 IF O$="REINDEER" THEN 640
430 IF O$="FOOD" THEN 660
440 IF O$="FATHER" THEN 670
450 IF O$="INK BLOT" THEN 680
460 IF O$="FACE" THEN 700
470 IF O$="CAMEL" THEN 720
480 PRINT "TELL ME ABOUT YOUR
    CHILDHOOD.": INPUT P$
490 PRINT "[CLR]GO ON...": INPUT P$
    : PRINT "[CLR]HMMM...TELL ME
    MORE.": INPUT P$
500 PRINT "[CLR]THAT IS ENOUGH FOR
    NOW. CALL ME TOMORROW ABOUT
    THIS."
510 GOTO 250
520 PRINT "OBVIOUSLY YOU FEEL
    ENTANGLED. THE"
```

```
530 PRINT "SPIDER IS A SYMBOL OF
      YOUR FEAR OF[SPACE 6]
      FRUSTRATION.": GOTO 250
540 PRINT "YOUR DEEP FEELINGS FOR
      YOUR MOTHER ARE"
550 PRINT "ONLY NATURAL. DO NOT
      BE AFRAID TO LET[SPACE 3]THEM
      SHOW.": GOTO 250
560 PRINT "SOMEWHERE IN YOUR
      SUB-CONSCIOUS,[SPACE 8]YOU HAVE
      A FEAR OF CATS."
570 PRINT "YOU STARTLE WHEN ONE
      CROSSES YOUR PATH[SPACE 2]AT
      NIGHT.": GOTO 250
580 PRINT "BEWARE THE SNAKE. YOU
      NEED TO CONTROL[SPACE 3]YOURSELF
      IN TENSE SITUATIONS."
590 GOTO 250
600 PRINT "THE BUTTERFLY SYMBOLIZES
      YOUR FREE[SPACE 6]SPIRIT AND
      GREAT INNER JOY."
610 PRINT "LET YOURSELF HAVE FUN."
      : GOTO 250
620 PRINT "THE ROPE IS A SIGN OF
      THE STRONG SUPPORT YOU OFFER
      YOUR FRIENDS."
630 PRINT "YOU HAVE GREAT INNER
      RESERVES.": GOTO 250
640 PRINT "YOU ARE BLESSED WITH
      A CHILDLIKE LOVE[SPACE 3]FOR
      NATURE. YOUR HEART IS ";
650 PRINT "PURE.": GOTO 250
660 PRINT "EITHER YOU ARE HUNGRY
      OR YOU DESIRE TO[SPACE 2]FEED
      OTHERS.": GOTO 250
670 PRINT "YOUR FEELINGS FOR YOUR
      FATHER MUST[SPACE 6]SOMEDAY
      BE EXPLORED.": GOTO 250
680 PRINT "YOU ARE A REALIST.
      UNFORTUNATELY, YOU[SPACE 3]ARE
      WITHOUT IMAGINATION."
690 GOTO 250
700 PRINT "YOU ARE LIKE A NEWBORN
      CHILD, YOU ARE"
710 PRINT "ATTRACTED TO FACES. YOU
      SEEK ACCEPTANCE AND LOVE."
      : GOTO 250
720 PRINT "ONE HUMP OR TWO?"
      : PRINT "SERIOUSLY, YOU ARE
      IN A SPIRITUAL DESERT"
730 PRINT "ALONE YOU CAN PREPARE
      TO REENTER THE[SPACE 4]WORLD."
      : GOTO 250
```

MORE →

```
740 PRINT "[CLR]THANK YOU FOR COMING.  
I WILL SEE YOU[SPACE 4]AGAIN  
[SPACE 2]NEXT WEEK."  
750 PRINT "PLEASE LEAVE $50 ON THE  
TV."
```

Spatial Concepts

This fun program is enjoyed by young children. It combines picture-making and learning the concepts of up, down, left, right and in.

The child chooses from a list of objects: bird, child, duck and star. Then the child chooses where to place the object in the colorful castle and landscape displayed on the screen. The object can also be erased.


The child should play along with an older person—especially if the child can't read. Together they can create pictures and a story!


```

100 REM SPATIAL CONCEPTS
110 L$(1)="[HOME][DOWN 16][RIGHT 11]"
120 L$(2)=L$(1)+"[DOWN 4][RIGHT 18]"
130 L$(3)="[HOME][DOWN][RIGHT 16]"
140 L$(4)="[HOME][DOWN 12][RIGHT 2]"
:L$(5)=L$(4)+"[RIGHT 29]"

```

MORE →





```

150 PRINT "[CLR]": PRINT TAB(
      12)"SPATIAL CONCEPTS": PRINT
      : PRINT
160 PRINT "WE ARE ABOUT TO PLAY
      A GAME TO LEARN UP,DOWN, LEFT,
      RIGHT AND IN."
170 PRINT
180 PRINT "YOU WILL SE A CASTLE
      WITH THE WORDS UP, DOWN, LEFT,
      RIGHT AND IN."
190 PRINT
200 PRINT "THESE WORDS ARE PLACES
      NEAR THE CASTLE[SPACE 2]THAT
      SHOW WHAT THE WORDS MEAN."
210 PRINT : PRINT "AT THE BOTTOM
      UF THE SCREEN YOU WILL"
220 PRINT "SEE THE NAMES OF THINGS
      TO PAINT WITH.": PRINT
230 PRINT "YOU CAN USE THESE THINGS
      TO FINISH[SPACE 3]THE CASTLE
      PICTURE.": PRINT
240 PRINT "PRESS ANY KEY TO
      CONTINUE.": PRINT
250 GET Q$: IF Q$="" THEN 250
260 PRINT "[CLR]": PRINT "I WILL
      ASK WHAT THING YOU WANT TO
      ADD"
270 PRINT "TO THE PICTURE, AND YOU
      MUST ENTER ITS[SPACE 2]NAME."
      : PRINT
280 PRINT "THEN I WILL ASK YOU WHERE
      YOU WANT TO"
290 PRINT "PUT THE THING NEAR THE
      CASTLE, AND YOU[SPACE 2]MUST
      TELL ME: UP, DOWN, LEFT";
300 PRINT ", RIGHT, OR IN.": PRINT
310 PRINT "OK, LET US BEGIN..."
      : PRINT
320 PRINT "PRESS ANY KEY TO START."
      : PRINT
330 GET Q$: IF Q$="" THEN 330
340 FOR X=1 TO 5: READ T$(X): NEXT X
350 DATA BIRD,CHILD,DOG,STAR,ERASE
360 FOR X=1 TO 5: READ W$(X): NEXT X
370 DATA IN,DOWN,UP,LEFT,RIGHT
380 PRINT "[CLR]": PRINT : PRINT
      : PRINT
390 FOR X=55896 TO 56295: POKE X,5
      : NEXT X
400 FOR X=1624 TO 2023: POKE X,160
      : NEXT X
410 POKE 53281,1: POKE 53280,6

```

Alteration—Change the castle design by altering lines 420-590.



```

420 PRINT TAB( 14)"[MED.GRAY][RVSON]
    [RVSOFF] [RVSON][SPACE 3]
    [RVSOFF] [RVSON] [RVSOFF]"
430 PRINT TAB( 14)"[RVSON][SPACE 7]
    [RVSOFF]"
440 PRINT TAB( 15)"[RVSON][SPACE 5]
    [RVSOFF]"
450 PRINT TAB( 15)"[RVSON][SPACE 5]
    [RVSOFF]"
460 PRINT TAB( 15)"[RVSON][SPACE 2]
    [BLK] [MED.GRAY][SPACE 2][RVSOFF]
    "
470 PRINT TAB( 9)"[RVSON][SPACE 2]
    [RVSOFF] [RVSON] [RVSOFF][SPACE 2]
    [RVSON][SPACE 2][BLK] [MED.GRAY]
    [SPACE 2][RVSOFF][SPACE 2][RVSON]
    [SPACE 2][RVSOFF] [RVSON][SPACE 2]
    [RVSOFF]"
480 PRINT TAB( 9)"[RVSON][SPACE 4]
    [RVSOFF][SPACE 2][RVSON][SPACE 5]
    [RVSOFF][SPACE 3][RVSON][SPACE 3]
    [RVSOFF]"
490 PRINT TAB( 9)"[RVSON][SPACE 3]
    [RVSOFF][SPACE 3][RVSON][SPACE 5]
    [RVSOFF][SPACE 3][RVSON] [BLK]
    [MED.GRAY] [RVSOFF]"
500 PRINT TAB( 9)"[RVSON] [BLK]
    [MED.GRAY][SPACE 15][RVSOFF]"
510 PRINT TAB( 9)"[RVSON][SPACE 17]
    [RVSOFF]"
520 PRINT TAB( 9)"[RVSON][SPACE 10]
    [BLK] [MED.GRAY][SPACE 2][BLK]
    [MED.GRAY][SPACE 3][RVSOFF]"
530 PRINT TAB( 9)"[RVSON][SPACE 17]
    [RVSOFF]"
540 PRINT TAB( 8)"[RVSON][BLU]
    [MED.GRAY][SPACE 2][WHITE][SPACE 6]
    [MED.GRAY][SPACE 9][BLU]
    [MED.GRAY][RVSOFF]"
550 PRINT TAB( 7)"[RVSON][BLU][SPACE 2]
    [MED.GRAY][SPACE 2][WHITE][SPACE 6]
    [MED.GRAY][SPACE 9][BLU][SPACE 2]
    [RVSOFF]"
560 PRINT TAB( 6)"[RVSON][BLU][SPACE 3]
    [MED.GRAY][SPACE 2][WHITE][SPACE 6]
    [MED.GRAY][SPACE 9][BLU][SPACE 3]
    [RVSOFF]"
570 PRINT TAB( 7)"[RVSON][BLU][SPACE 5]
    [MED.GRAY]f[SPACE 3]c[BLU][SPACE 11]
    "
580 PRINT TAB( 8)"[RVSON][BLU][SPACE 5]
    [MED.GRAY]f[SPACE 3]c[BLU][SPACE 8]
    "

```

MORE →

```
590 PRINT TAB( 20)"[RVSON][SPACE 5]
   [RVSOFF]": PRINT : PRINT
600 PRINT "[BLK][RIGHT 5]BIRD[SPACE 2]
   CHILD[SPACE 2]DOG[SPACE 2]STAR
   [SPACE 2]ERASE[HOME]"
610 PRINT L$(1); "[RVSON][BLK]IN
   [RVSOFF]": PRINT L$(2); "[RVSON]
   DOWN[HOME]"
620 PRINT L$(3); "[RVSON]UP[RVSOFF]"
   : PRINT L$(4); "[RVSON]LEFT
   [RVSOFF]": PRINT L$(5); "[RVSON]
   RIGHT[RVSOFF]"
630 PRINT "[HOME][BLK]";
   : INPUT "PAINT WITH"; P$
640 PRINT "[HOME][SPACE 20]"
650 FOR X=1 TO 5: IF P$=T$(X) THEN
   670
660 NEXT X: GOTO 630
670 PRINT "[HOME]"; : INPUT "PAINT
   WHERE"; W$
680 PRINT "[HOME][SPACE 28]"
690 FOR X=1 TO 5: IF W$=W$(X) THEN
   710
700 NEXT X: GOTO 670
710 C$="[WHITE]"
720 IF W$="DOWN" THEN C$="[GRN]"
730 PRINT L$(X); C$; "[RVSON][SPACE 5]"
   : PRINT L$(X); "[DOWN][RVSON]
   [SPACE 5]": PRINT L$(X); "[DOWN 2]
   [RVSON][SPACE 5][HOME]"
740 C$="[RVSON][GRN]": IF W$<>"DOWN"
   THEN C$="[BLK]"
750 IF P$="STAR" THEN PRINT L$(X); "
   [DOWN][RIGHT][BLK]*"
760 IF P$="CHILD" THEN PRINT L$(X);
   C$; "[RIGHT]W": PRINT L$(X); C$; "
   [DOWN]N±M": PRINT L$(X); C$; "
   [DOWN 2][RIGHT]H"
770 IF P$="BIRD" THEN PRINT L$(X); C$;
   " )": PRINT L$(X); C$; "[DOWN]=±>"
   : PRINT L$(X); C$; "[DOWN 2] )"
780 IF P$="DOG" THEN PRINT L$(X); C$; "
   [DOWN] Q N": PRINT L$(X); C$; "
   [DOWN 2] luh"
790 IF P$="ERASE" THEN PRINT L$(X); "
   [RVSON][BLK][DOWN]"; W$(X)
800 GOTO 630
```


Mind Reader

Here's a fun guessing game in which you and the computer take turns guessing numbers. The program begins with an animated man (you can see his lips move) asking for the user's name. The name is used throughout the game.

The user has five chances to guess a number between 1 and 20. The computer man helps by giving *higher* and *lower* clues. The user then chooses either to think of a number or let the computer man think of a number. The user enters a number between 1 and 5, and the computer man has one chance to guess the number. After this process is repeated five times, the computer man reports the score.


Alteration—Changing the value of **TL** in line 150 changes the speaking rate of the animated man. Decreasing **TL** increases speed.

```

100 REM  MIND READER
110 DIM  W$(81),G$(5)
120 FOR X=0 TO 81: READ W$(X): NEXT
130 FOR X=1 TO 5: READ G$(X): NEXT
140 L$="[HOME][DOWN 11][RIGHT 12]"
150 B$="[SPACE 9]":TL=250
    : POKE 53281,1
160 M1$="[HOME][DOWN 11] [RVSON]k
    [RVSOFF][SPACE 2][RVSON]diiif
    [RVSOFF] k":M2$=" cf cooov k"
170 M3$="[HOME][DOWN 11] [RVSON]k
    [RVSOFF][SPACE 2][BLK][RVSON]
    [SPACE 5][RVSOFF] k":M4$=" cf
    [RVSON]iiiiii[RVSOFF] k"

```

MORE →



```

180 PRINT "[CLR][BLK]"
190 PRINT "[SPACE 3][RVSON]v[SPACE 4]
   ": PRINT "[SPACE 2][RVSON]k
   [SPACE 2]di[RVSOFF]vb"
   : PRINT " d[RVSON]i[RVSOFF]vi
   [SPACE 2]i b"
200 PRINT " [RVSON]k[RVSOFF] c vc
   v k": PRINT "d[RVSON]i[RVSOFF]
   d[RVSON]i[RVSOFF]fd[RVSON]i
   [RVSOFF]f k"
210 PRINT "c[RVSON]b[RVSOFF] c[RVSON]
   v[RVSOFF]vc[RVSON]v[RVSOFF]v
   k": PRINT " cf[SPACE 6][RVSON]k"
   : PRINT "[SPACE 2]k[SPACE 3]b
   [SPACE 2][RVSON]k"
220 PRINT " [RVSON]k[RVSOFF][SPACE 4]
   dv cf": PRINT " [RVSON]k[RVSOFF]
   [SPACE 8]k"
230 PRINT " [RVSON]k[RVSOFF][SPACE 2]
   [RVSON]diiif[RVSOFF] k"
   : PRINT " cf cooov k"
   : PRINT "[SPACE 2]k[SPACE 6]dv
240 PRINT "[SPACE 2]cbiiiiiv"
   : PRINT "[SPACE 6][RVSON] "
250 PRINT "[SPACE 4][RED]++++"
   : PRINT "[SPACE 4][BLK]k[RED]+++
   [BLK][RVSON]k": PRINT "[SPACE 4]
   [RVSON]c[RED]+++[BLK]v"
260 PRINT "[SPACE 5][CYN][RVSON]
   [SPACE 3]": PRINT "[SPACE 5]
   [RVSON] [RVSOFF] [RVSON] "
   : PRINT "[SPACE 4][BLK][RVSON]
   [SPACE 2][RVSOFF] [RVSON][SPACE 2]
   "
270 A=0:B=3: GOSUB 580
280 PRINT L$;"[DOWN 3]";: INPUT N$
   : PRINT L$;"[DOWN 3]";"[SPACE 10]
   "
290 FOR T=1 TO 2222: NEXT
300 A=4:B=39: GOSUB 580
310 FOR T=1 TO 2222: NEXT
320 C=0: FOR G=1 TO 5:K=0
   :M= INT ( RND (8)*20)+1
330 A=9:B=10: GOSUB 580: GOSUB 600
   :K=K+1
340 A=11:B=11: GOSUB 580
350 PRINT L$;"[DOWN 3]";: INPUT Q$
   :N= VAL (Q$)
360 IF M>N THEN A=40:B=40: GOSUB 580
370 IF M<N THEN A=41:B=41: GOSUB 580
380 FOR T=1 TO 500: NEXT
   : PRINT L$;"[DOWN 3]";"[SPACE 5]"

```

Alteration—By changing **M** in line 320 to **M=INT(RND(7)*10)+1**, you make the computer think of a number between 1 and 10, instead of 1 and 20.

```
390 IF M=N THEN W$(48)= STR$(K):A=42
   :B=49: GOSUB 580:C=C+1: GOTO 430
400 IF K<5 THEN 330
410 W$(54)= STR$(M):A=50:B=54
   : GOSUB 580
420 FOR T=1 TO 1111: NEXT T
430 NEXT G: FOR T=1 TO 1000: NEXT T
440 W$(55)=N$:W$(58)= STR$(100*C/5)
   :A=55:B=60: GOSUB 580
450 A=61:B=61: GOSUB 580:A=27:B=39
   : GOSUB 580
460 C=0: FOR G=1 TO 5
470 W$(64)=G$(G):A=62:B=67: GOSUB 580
480 PRINT L$;"[DOWN 3]";: INPUT Q$
   :N= VAL(Q$): IF N<1 OR N>5 THEN
PRINT L$;"[DOWN 3]";"[SPACE 3]"
   : GOTO 470
490 M= INT ( RND (8)*5)+1
   : FOR T=1 TO 900: NEXT
   : PRINT L$;"[DOWN 3]";"[SPACE 5]"
500 IF M=N THEN A=68:B=71: GOSUB 580
   :C=C+1: GOTO 520
510 A=72:B=74: GOSUB 580
520 FOR T=1 TO 900: NEXT T,G
530 FOR T=1 TO 1000: NEXT T:A=75:B=79
   :W$(77)= STR$(C*20): GOSUB 580
540 FOR T=1 TO 1000: NEXT T:A=80:B=81
   : GOSUB 580
550 PRINT L$;"[DOWN 3]";: INPUT N$
   : PRINT L$;"[DOWN 3]";"[SPACE 6]"
560 IF LEFT$(N$,1)="Y" THEN 180
570 PRINT "[CLR]": END
580 FOR X=A TO B: GOSUB 620
   : PRINT L$;B$: PRINT L$;W$(X)
   : FOR T=1 TO TL: NEXT T
590 GOSUB 630: FOR T=1 TO TL
   : NEXT T,X: PRINT L$;B$: RETURN
600 GOSUB 620: PRINT L$;B$
   : PRINT L$;G$(G): FOR T=1 TO TL
   : NEXT T
610 GOSUB 630: FOR T=1 TO TL: NEXT T
   : PRINT L$;B$: RETURN
620 PRINT M1$: PRINT M2$: RETURN
630 PRINT M3$: PRINT M4$: RETURN
640 DATA HELLO,WHO,ARE,YOU?,YOU,ARE,
  ABOUT,TO,PLAY,GUESS,THE,NUMBER.
650 DATA FIRST,I,WILL,THINK,OF,FIVE,
  NUMBERS,AND,YOU,WILL,TRY,TO,
  GUESS,THEM.
660 DATA LATER,YOU,WILL,THINK,OF,
  FIVE,NUMBERS,AND,I,WILL,TRY,TO,
  GUESS,THEM.
```

670 DATA HIGHER,LOWER,THAT,IS,RIGHT!,
IT,TOOK,YOU, ,TRIES.
680 DATA TOO,BAD,IT,WAS, , ,YOU,GOT,
 ,PERCENT,CORRECT,NOW
690 DATA ENTER,THE, ,NUMBER,FROM,1
 -5.,MY,GENIUS,OVERCOMES,ME.,I,WAS
WRONG.
700 DATA I,GOT, ,PERCENT,CORRECT,
PLAY,AGAIN Y/N
710 DATA FIRST,SECOND,THIRD,FOURTH,
FIFTH

Poetry

The program creates *haiku*, an ancient form of Japanese poetry. The traditional form calls for three lines of five, seven and five syllables. The poems usually refer to seasonal changes and moods. Therefore, the phrases selected randomly by the program contain related vocabulary. The program can write some very nice haiku.

```
100 REM "POETRY"  
110 POKE 53281,1  
120 FOR X=0 TO 9: READ F$(X): NEXT  
    : FOR X=0 TO 7: READ S$(X)  
    : NEXT X  
130 FOR X=0 TO 7: READ T$(X): NEXT X  
140 PRINT "[CLR]": PRINT TAB( 17)"  
    [RVSON][DK.GRAY]POETRY[RVSOFF]"  
    : PRINT  
150 PRINT : PRINT "HELLO HONORABLE  
    USER.": PRINT  
160 PRINT "I AM A JAPANESE HAIKU  
    PROGRAM.": PRINT  
170 PRINT "PRESS ANY KEY AND I WILL  
    BE PLEASED TO[SPACE 2]WRITE  
    A POEM FOR YOU."  
180 PRINT : PRINT "HAIKU IS AN  
    ANCIENT FORM OF POETRY."  
190 PRINT "IT IS AN HONOR TO WRITE  
    FOR YOU."  
200 PRINT : PRINT "HIT ANY KEY TO  
    START."
```

MORE →



Alteration—Change the phrases used to build the haiku by changing the **DATA** statements in lines 300-390. Each new phrase should have the same number of syllables as the phrase it replaces.

```

210 GET Q$: IF Q$="" THEN 210
220 A= INT ( RND (8)*10)
    :B= INT ( RND (8)*8)
    :C= INT ( RND (6)*8)
230 PRINT "[CLR][DOWN 8]"
240 PRINT TAB( 20- LEN (F$(A))/2);
    F$(A)
250 PRINT TAB( 20- LEN (S$(B))/2);
    S$(B)
260 PRINT TAB( 20- LEN (T$(C))/2);
    T$(C)
270 PRINT : PRINT : PRINT
    : PRINT "[SPACE 4]HIT [RVSON]R
    [RVSOFF] KEY TO RECEIVE A NEW
    POEM. "
280 GET Q$: IF Q$="" THEN 280
290 IF Q$="R" THEN 220
300 DATA THE GRAY MOUNTAINS HOST,
    TODAY I SAW THE,SPLASHES IN
    THE POND
310 DATA NOW CHERRY BLOSSOMS,DROPPING
    FROM BRANCHES,FRESHLY DRAWN
    CHALK LINES
320 DATA WINTER HILLSIDES HOST,NOW
    REDFACED CHILDREN,A MOONLIT
    QUIET
330 DATA A GARDEN CHANGES,MOMENTARY
    FLASHES OF,SQUIRREL GATHERING
    ACORNS
340 DATA DISTURB THE REFLECTION
    OF,SWIRL IN THE TWILIGHT
    COOLNESS
350 DATA RED LEAVES PROVIDE RAKE
    FODDER,REFLECT IMAGINATION
360 DATA THE LAUGHTER AND SNOWY
    PLAY,CATCH RAYS OF BROKEN
    MOONLIGHT
370 DATA THE SUMMER LIGHTNING,WINTER
    IS COMING,A MOONLIT QUIET,A
    GARDEN CHANGES
380 DATA FUEL FOR CHILDREN'S PLAY,OF
    SUMMER CHILDREN,OF REDFACED
    CHILDREN
390 DATA SPLASHES IN THE POND

```

No Free Lunch

Put on your thinking cap for this arithmetic game. You're out to lunch along the ocean and have to hop rocks across hazardous tidepools to reach your food. A correct answer to the arithmetic problem gets you to the next rock, closer to your lunch. With three correct answers in a row, you can eat. But watch out! A wrong answer gets you all wet.

```

100 REM NO FREE LUNCH
110 POKE 53281,14: POKE 53280,6
120 PRINT "[CLR]"; FOR X=1 TO 17
    : PRINT : NEXT X
130 PRINT TAB( 34);"[RVSON][BLK]
    LUNCH"
140 PRINT "[RVSON][LT.BLU][SPACE 2]
    [DK.GRAY][SPACE 2][LT.BLU][SPACE 4]
    [DK.GRAY] [LT.BLU][SPACE 6]
    [DK.GRAY][SPACE 3][LT.BLU][SPACE 4]
    [DK.GRAY][SPACE 2][LT.BLU][SPACE 4]
    [DK.GRAY][SPACE 2][LT.BLU][SPACE 3]
    [DK.GRAY][SPACE 6][LT.BLU] ";
150 PRINT "[RVSON][BLU] [DK.GRAY]
    [SPACE 3][BLU][SPACE 3][DK.GRAY]
    [SPACE 2][BLU][SPACE 5][DK.GRAY]
    [SPACE 4][BLU][SPACE 3][DK.GRAY]
    [SPACE 3][BLU][SPACE 3][DK.GRAY]
    [SPACE 3][BLU][SPACE 3][DK.GRAY]
    [SPACE 7][BLU]";

```

MORE →



```

160 PRINT "[RVSON][BLU] [DK.GRAY]
[SPACE 4][BLU][SPACE 2][DK.GRAY]
[SPACE 2][BLU][SPACE 4][DK.GRAY]
[SPACE 5][BLU][SPACE 3][DK.GRAY]
[SPACE 3][BLU][SPACE 3][DK.GRAY]
[SPACE 4][BLU][SPACE 2][DK.GRAY]
[SPACE 7][BLU]";
170 PRINT "[RVSON][DK.GRAY][SPACE 5]
[BLU] [DK.GRAY][SPACE 4][BLU]
[SPACE 2][DK.GRAY][SPACE 7][BLU]
[DK.GRAY][SPACE 4][BLU][SPACE 2]
[DK.GRAY][SPACE 5][BLU][SPACE 2]
[DK.GRAY][SPACE 7][BLU]";
180 PRINT "[RVSON][DK.GRAY][SPACE 40]
";
190 PRINT "[LT.BLU][SPACE 13][RVSON]
[BLK]ND FREE LUNCH[RVSOFF][HOME]"
200 PRINT "[RIGHT 3][YEL][RVSON]
[POUND] *"
210 PRINT "[RIGHT 3][YEL][RVSON]
[SPACE 3]"
220 PRINT "[RIGHT 3][YEL]*[RVSON]
[RVSOFF][POUND]"
230 D$="[HOME][DOWN 16]"
240 J$(1)="[RIGHT 2]";J$(2)="[RIGHT 7]
";J$(3)=J$(2)+"[RIGHT 9]"
250 J$(4)=J$(3)+"[RIGHT 5]"
:J$(5)=J$(4)+"[RIGHT 6]"
:J$(6)=J$(5)+"[RIGHT 5]"
260 U$(1)="[WHITE] 0";U$(2)="[N][RVSON]
[RVSOFF]M";U$(3)="[RIGHT]H"
270 E$(1)="[SPACE 2]";E$(2)="[SPACE 3]
";E$(3)="[SPACE 2]"
280 J=1
290 A= INT ( RND (8)*9)+1
300 B= INT ( RND (8)*9)+1
310 PRINT D$;: FOR X=1 TO 3
: PRINT J$(J);U$(X): NEXT X
320 PRINT "[HOME][DOWN 6][RIGHT 6]
[SPACE 23]"
330 PRINT "[RIGHT 6][SPACE 23]"
340 PRINT "[HOME][DOWN 6][RIGHT 6]HOW
MUCH IS"A"+"B";
350 INPUT C
360 IF A+B=C THEN 430
370 PRINT "[RIGHT 6]SORRY, THE ANSWER
IS"A+B
380 PRINT D$;: FOR X=1 TO 3
: PRINT J$(J);E$(X): NEXT X
390 PRINT D$;"[DOWN 2]";
: FOR X=1 TO 2: PRINT J$(J);"
[RIGHT 2]";U$(X): NEXT X

```

Alteration—Alter the equations for **A** and **B** in lines 290 and 300 to change the range of numbers in the questions. For example, **A=INT(RND(9)*20)+1** will include numbers up to 20 in the problems.

Alteration—Make the game more challenging by changing the equations into multiplication problems. Change the addition sign in lines 340, 360 and 370 to the multiplication sign.


```
400 SO=54272: FOR Z=SO TO SO+24
    : POKE Z,0: NEXT Z: POKE SO+5,11
    : POKE SO+6,31: POKE SO+24,15
410 POKE SO+1,65: POKE SO,227
    : POKE SO+4,129
420 PRINT "[HOME][DOWN 9][RIGHT 6]
    SPLASH!!!": FOR T=1 TO 500
    : NEXT T: GOTO 490
430 PRINT "[RIGHT 6]CORRECT!"
440 FOR T=0 TO 999: NEXT T
450 PRINT D$;: FOR X=1 TO 3
    : PRINT J$(J);E$(X): NEXT X
460 J=J+1: IF J<6 THEN 290
470 PRINT D$;: FOR X=1 TO 3
    : PRINT J$(J);U$(X): NEXT X
480 PRINT "[HOME][DOWN 8][RIGHT 6]
    YUM! YUM! EAT WELL!!!"
490 PRINT : PRINT "[RIGHT 6]TRY
    AGAIN";: INPUT Q$
500 IF LEFT$(Q$,1)="Y" THEN 110
510 END
```

Spelling Bee

Here's an educational game of great versatility, providing entertainment and instruction. The program features a bee that moves across the screen, leaving behind it a scrambled word. The object is to unscramble the word and enter the correct spelling.

```

100 REM  SPELLING BEE
110 W=10
120 DIM W$(W),S(20)
130 POKE 53280,6: POKE 53281,14
    : PRINT "[CLR][BLK]": PRINT
140 PRINT "WELCOME TO SPELLING BEE."
150 PRINT : PRINT "YOU WILL SEE
    THE BEE FLY ACROSS THE"
160 PRINT "SCREEN, LEAVING LETTERS
    AS IT GOES."
170 PRINT : PRINT "THESE LETTERS
    ARE THE LETTERS OF A[SPACE 6]
    SCRAMBLED WORD."
180 PRINT "YOU MUST UNSCRAMBLE THE
    WORD AND ENTER"
190 PRINT "THE CORRECT SPELLING
    TO RECEIVE POINTS.": PRINT
210 PRINT : PRINT "GOOD LUCK."
    : PRINT
220 B$(1)="[SPACE 5]V[SPACE 2]NN"
230 B$(2)="[SPACE 5][RVSON][BLK]
    [YEL] [BLK] [YEL][RVSOFF][BLK]W"
240 B$(3)="[SPACE 3][RVSON][BLK]
    [YEL] [BLK] [YEL] [BLK] [YEL]
    [RVSOFF]"
250 B$(4)="[BLK][SPACE 2]<[RVSON]
    [BLK] [YEL] [BLK] [YEL] [BLK]
    [YEL][RVSOFF][BLK]="
260 B$(5)="[BLK][SPACE 6];;"
270 D$="[HOME][DOWN 8]"
    :P$="[RIGHT 34]"
280 FOR X=1 TO W: READ W$(X): NEXT
290 PRINT : PRINT "HIT ANY KEY TO
    START."
300 GET Q$: IF Q$="" THEN 300
310 T=0:S=0
320 FOR X=1 TO 8
330 A= INT ( RND (8)*W)+1:W#=W$(A)
    :S$=""
340 L= LEN (W#): FOR Y=1 TO L:S(Y)=0
    : NEXT Y
350 FOR Y=1 TO LEN (W#)
360 K= INT ( RND (3)*L)+1
    : IF S(K)=1 THEN 360
370 S#=S#+ MID$ (W#,K,1):S(K)=1
    : NEXT Y
380 IF S#=W# THEN 340

```

Alteration—Change the number of words presented by altering line 320. For example, **FOR X=1 TO 10** will ask for 10 questions.



Alteration—Change or expand the vocabulary to fit the user by adding more **DATA** statements after line 600. Change **W** in line 110 to match the number of words in the program's vocabulary.

```

390 PRINT "[CLR]WORD #"X" SCORE="S
400 F=1: FOR Y=1 TO 28
      :B$= LEFT$ (P$,Y)
410 IF LEN (B$)>17 AND F<L+1 THEN
      PRINT D$;B$;"[DOWN 3][LEFT 2]"
      MID$ (S$,F,1):F=F+1
420 PRINT D$
430 FOR Z=1 TO 5: PRINT B$;B$(Z)
      : NEXT Z
440 FOR T=1 TO 49: NEXT T,Y
450 PRINT D$;
460 PRINT D$;"[DOWN 6]"
470 INPUT "ENTER CORRECT SPELLING";C$
      : PRINT
480 IF C$=W$ THEN PRINT TAB( 13)"
      [RVSON]THAT IS CORRECT!":S=S+125
      : GOTO 510
490 PRINT TAB( 10)"[RVSON]WRONG,
      IT WAS "W$"."
500 FOR T=1 TO 1000: NEXT T
510 FOR T=1 TO 2000: NEXT T
520 NEXT X
530 PRINT "[CLR]": PRINT
540 IF S>900 THEN PRINT "YOU WIN
      THE SPELLING BEE!": GOTO 580
550 IF S>800 THEN PRINT "VERY WELL
      DONE.": GOTO 580
560 IF S>700 THEN PRINT "NOT BAD."
      : GOTO 580
570 PRINT "YOU NEED PRACTICE."
580 PRINT : INPUT "PLAY AGAIN";Q$
590 IF LEFT$ (Q$,1)="Y" THEN 290
600 DATA THAT,THEIR,HOUSE,BOAT,
      PEACE,DRAGON,ORANGE,GAME,LOVE,
      HOPE

```

Speed Reading

A fundamental principle of speed reading is that we retain more if we read words in phrases, rather than one at a time. This program helps develop that skill. It flashes phrases quickly then tests retention of the information with true/false questions.

Two players are needed. When prompted by the computer, player 1 types in five statements and five related true/false questions along with the answer. Then player 2 decides on the time duration he'll see each phrase. Time can be as short as .02 seconds. After the phrases are flashed, player 2 is tested with the questions. Score is based on correct answers and the time the player had to read questions.

Then the players change roles and player 2 does the speed-reading test.

Alteration—Change the game into an arithmetic flashcard game. Enter math problems and answers instead of sentences. Change appropriate **PRINT** statements if you do this.

```

100 REM  SPEED READING
110 Q=5: DIM P$(2,Q),Q$(2,Q),A$(2,Q)
120 POKE 53281,6: PRINT "[CLR][WHITE]"
130 PRINT TAB( 14)"[RVSON]SPEED
    READING": PRINT
140 PRINT "WELCOME TO THE GAME OF
    SPEED AND"
150 PRINT "CONCENTRATION.": PRINT
160 PRINT "YOU MUST BE ABLE TO GRASP
    WHOLE PHRASES IN A MOMENT."
    : PRINT
170 PRINT "AFTER THE PHRASES ARE
    DISPLAYED, THERE"
180 PRINT "WILL BE QUESTIONS ABOUT
    THE CONTENT."
190 PRINT "CORRECT ANSWERS UP YOUR
    SCORE."
200 PRINT "THE MORE DIFFICULT THE
    TIME FACTOR, THE MORE POINTS
    ARE EARNED."
210 PRINT : PRINT "ONE PLAYER ENTERS
    FIVE PHRASES, FIVE"
220 PRINT "QUESTIONS, AND THE FIVE
    ANSWERS."
230 PRINT "THE SECOND PLAYER ENTERS
    THE TIME"
240 PRINT "FACTOR BY ENTERING THE
    NUMBER OF"
250 PRINT "SECONDS FOR WHICH THE
    PHRASES ARE[SPACE 7]DISPLAYED."
260 PRINT : PRINT "HIT ANY KEY FOR
    MORE INSTRUCTIONS."
270 GET Q$: IF Q$="" THEN 270
280 PRINT "[CLR]ALL FIVE PHRASES
    WILL BE SHOWN AND THEN"
290 PRINT "ALL FIVE QUESTIONS WILL
    BE ASKED.": PRINT
300 PRINT "THEN THE SECOND PLAYER
    WILL HAVE A TURN."

```

```
310 PRINT "ALL QUESTIONS WILL BE
      TRUE OR FALSE"
320 PRINT "QUESTIONS. TRUE WILL BE
      INDICATED BY A"
330 PRINT "T AND FALSE WILL BE
      INDICATED BY AN F.": PRINT
340 PRINT "HERE IS AN EXAMPLE:"
      : PRINT
350 PRINT "[RVSON]PHRASE
      :[RVSOFF] THIS IS A COMPUTER
      GAME.": PRINT
360 PRINT "[RVSON]QUESTION
      :[RVSOFF] T OR F": PRINT "THIS
      IS A CARD GAME.": PRINT
370 PRINT "[RVSON]ANSWER
      :[RVSOFF] F"
380 PRINT : PRINT "HIT ANY KEY TO
      START."
390 GET Q$: IF Q$="" THEN 390
400 PRINT "[CLR]":P=1
410 INPUT "PLAYER ONE'S NAME";N$(1)
      : PRINT
420 INPUT "PLAYER TWO'S NAME";N$(2)
      : PRINT
```

Alteration—Change the game into a translation exercise. Use foreign and English equivalents instead of sentences. Change appropriate **PRINT** statements if you do this.

MORE →



```
430 PRINT "[CLR]BEGIN NOW "N$(P)"."  
440 FOR X=1 TO Q: PRINT  
: PRINT "ENTER PHRASE #"X  
450 PRINT : PRINT "PHRASES MAY NOT  
BE MORE THAN 1 LINE LONG(40  
CHARACTERS)."  
460 PRINT : INPUT P$(P,X)  
470 IF LEN (P$(P,X))>40 THEN 450  
480 PRINT : PRINT "ENTER QUESTION  
#"X  
490 PRINT : PRINT "QUESTIONS MAY  
NOT BE MORE THAN 1 LINE[SPACE 3]  
LONG (40 CHARACTERS)."  
500 PRINT : INPUT Q$(P,X)  
510 IF LEN (Q$(P,X))>40 THEN 450  
520 PRINT : PRINT "ENTER ANSWER #"X  
530 PRINT : INPUT A$(P,X)  
540 IF A$(P,X)<>"T" AND A$(P,X)<>"F"  
THEN PRINT "ANSWER MUST BE T  
OR F.": GOTO 530  
550 PRINT "[CLR]": NEXT X  
560 K=1: IF P=1 THEN K=2  
570 PRINT "OK "N$(K)".": PRINT  
580 W(K)=0:S(K)=0  
590 PRINT "ENTER THE NUMBER OF  
SECONDS EACH PHRASE"  
600 INPUT "WILL BE SHOWN";S  
: IF S>10 THEN PRINT "TOO LONG!"  
: GOTO 590  
610 FOR X=1 TO Q: PRINT "[CLR][RVSON]  
PHRASE #"X: PRINT : PRINT : PRINT  
: PRINT : PRINT P$(P,X)  
620 TI$="000000"  
630 PRINT "[HOME][RIGHT 14]"TI$  
: IF TI<S*60 THEN 630  
640 PRINT "[CLR]": FOR T=0 TO 999  
: NEXT T,X  
650 FOR X=1 TO 5: PRINT "[CLR][RVSON]  
QUESTION #[RVSOFF]"X: PRINT  
660 PRINT : PRINT Q$(P,X): PRINT  
670 INPUT "YOUR ANSWER";A$  
:A$= LEFT$(A$,1)  
680 IF A$<>"T" AND A$<>"F" THEN PRINT  
"THESE ARE T OR F QUESTIONS."  
: GOTO 670  
690 IF A$=A$(P,X) THEN S(K)=S(K)+700-S  
: PRINT "[RVSON]CORRECT": GOTO 720  
700 W(K)=W(K)+1: PRINT "[RVSON]  
INCORRECT": PRINT P$(P,X)  
710 PRINT : PRINT : FOR T=1 TO 1000  
: NEXT T  
720 FOR T=0 TO 999: NEXT T,X  
730 IF P=1 THEN P=2: GOTO 430  
740 PRINT "[CLR]": PRINT
```

```
750 PRINT N$(1)" SCORE
      : "S(1),5-W(1)"CORRECT"
760 PRINT : PRINT
770 PRINT N$(2)" SCORE
      : "S(2),5-W(2)"CORRECT"
780 IF S(1)>S(2) THEN PRINT N$(1)"
      WINS!"
790 IF S(2)>S(1) THEN PRINT N$(2)"
      WINS!"
```

Pre-School Counter

Here's a useful educational game that helps young children learn counting up to 10. It's also good for introducing a computer to the child. It's easy to run and play.

The computer greets the player and then asks him to count the squares displayed. The user answers by entering a number. The computer then responds. If the answer is correct, the user is rewarded with a smiling face drawn by the computer. If the answer is incorrect, the face's smile becomes a frown. The correct answer is then displayed.

After 10 exercises, the computer gives the user both a raw and a percentage score.

Alteration—Improve on the smiling face by adding eyebrows, hair, ears, etc. Use **PRINT** statements between lines 150 and 260.

Alteration—Change **Q** in lines 270 to alter the number of questions the user gets. For example, **C=0: FOR Q=1 TO 5** gives the user 5 tries.

```

100 REM PRE-SCHOOL COUNTER
110 PRINT "[CLR][WHITE]": PRINT
    : PRINT
120 A$="[HOME][DOWN 10]"
    : B$=A$+"[DOWN 6]": D$=A$+"[RIGHT 17]"
    : "
130 M1$="[RVSON]diiif[RVSOFF]"
    : M2$="[DOWN]c000v"
140 M3$="[RVSON][SPACE 5][RVSOFF]"
    : M4$="[DOWN][RVSON]iiiiii[RVSOFF]"
150 M5$="diiif": M6$="[DOWN]c[SPACE 3]"
    : v"
160 PRINT TAB( 17)"f u f"
170 PRINT TAB( 15)"d[SPACE 7]f"
180 PRINT TAB( 14)"d[SPACE 9]f"
190 PRINT TAB( 14)"f[SPACE 2]q[SPACE 3]"
    : q[SPACE 2]d"
200 PRINT TAB( 13)"d[SPACE 11]f"
210 PRINT TAB( 13)"d[SPACE 11]f"
220 PRINT TAB( 13)"d[SPACE 11]f"
230 PRINT TAB( 14)"f[SPACE 9]d"
240 PRINT TAB( 14)"d[SPACE 9]f"
250 PRINT TAB( 15)"d[SPACE 7]f"
260 PRINT TAB( 17)"v i c"
270 C=0: FOR Q=1 TO 10
280 FOR X=1 TO 3: GOSUB 480: READ T$
    : PRINT A$;"[SPACE 2]";T$
    : FOR T=1 TO 700: NEXT T
290 GOSUB 490: PRINT A$;"[SPACE 12]"
    : FOR T=1 TO 400: NEXT T: NEXT X
300 PRINT B$;: SQ= INT ( RND (8)*10)+1
    : FOR Y=1 TO SQ: PRINT " [RVSON]"
    : [RVSOFF]";: NEXT Y
310 PRINT : PRINT : INPUT "[SPACE 3]"
    : ";A: IF A<>SQ THEN 330
320 C=C+1: PRINT A$;" [RVSON]CORRECT"
    : [RVSOFF]": GOSUB 480
    : FOR T=1 TO 2000: NEXT T
    : GOTO 350
330 PRINT A$;" [RVSON]INCORRECT"
    : [RVSOFF]": GOSUB 480
    : FOR T=1 TO 1000: NEXT T
    : GOSUB 500


```



```

340 PRINT A$ " THE ANSWER"
    : PRINT " IS "SQ: FOR T=1 TO 3500
    : NEXT T
350 GOSUB 490: PRINT A$;"[SPACE 13]"
360 PRINT A$;"[DOWN][SPACE 12]"
    : PRINT B$"[SPACE 20]"
370 PRINT B$"[DOWN 2][SPACE 15]"
380 FOR T=1 TO 1000: NEXT T: RESTORE
390 NEXT Q
400 PRINT A$;" OUT OF TEN"
    : FOR T=1 TO 700: NEXT T
410 GOSUB 490: FOR T=1 TO 400: NEXT T
420 PRINT A$;"[DOWN] TRIES, YOU"
    : FOR T=1 TO 700: NEXT T
430 GOSUB 490: FOR T=1 TO 400: NEXT T
440 PRINT A$;"[DOWN 2] GOT "C"
    RIGHT.": FOR T=1 TO 700: NEXT T
450 GOSUB 490: FOR T=1 TO 400: NEXT T
460 PRINT B$: INPUT "PLAY AGAIN";Q$
    : IF LEFT$(Q$,1)="Y" THEN 100
470 END
480 PRINT D$;M1$: PRINT D$;M2$
    : RETURN
490 PRINT D$;M3$: PRINT D$;M4$
    : RETURN
500 PRINT D$;M5$: PRINT D$;M6$
    : RETURN
510 PRINT D$;M3$: PRINT D$;M4$
    : RETURN
520 DATA HOW,MANY,SQUARES?

```



Keyboard Shooting Gallery

This game improves keyboard skills. You're at a carnival shooting gallery containing four animal targets. In each target a letter appears for a short time. You score points by typing letters displayed in the animals.

The game has three levels of difficulty, so as you become a better typist, you can make the game more challenging. At the expert level, "hunt-and-peck" typists will have trouble scoring well.

Alteration—The game can also include number keys, commas, periods, etc. if you change lines 110, 120 and 430 as follows:

```
110 DIM A$(55)
120 FOR X=0 TO 55:
A$(X)=CHR$(X+35):NEXT
430 FOR Y=1 TO 4:
L$(Y)=A$(INT(RND(8)*56)):
PRINT K$(Y); L$(Y): F(Y)=0: NEXT
```

```
100 REM KEYBOARD SHOOTING GALLERY
110 DIM A$(25)
120 FOR X=0 TO 25:A$(X)=CHR$(X+65)
: NEXT X
130 PRINT "[CLR][WHITE]"
140 PRINT "WELCOME TO [RVSON]KEYBOARD
CARNIVAL[RVSOFF]."

```

```

330 PRINT TAB( 9)" [BRN][RVSON]
    [SPACE 2][RVSOFF][SPACE 11][YEL]
    [RVSON][SPACE 2][RVSOFF]vk"
340 PRINT TAB( 9)"[BRN]c[RVSON]ii
    [RVSOFF]v"
350 PRINT TAB( 9)"[WHITE][RVSON]BEAR
    [RVSOFF][SPACE 9][RVSON]ROOSTER"
360 K$(1)="[HOME][DOWN 4][RIGHT 11]"
    :K$(2)=K$(1)+"[RIGHT 14]"
370 K$(3)=K$(1)+"[DOWN 10][RIGHT 2]"
    :K$(4)=K$(2)+"[DOWN 10][RIGHT 3]"
380 IF L=1 THEN T=20
390 IF L=2 THEN T=10
400 IF L=3 THEN T=5
410 X=0
420 I=0
430 FOR Y=1 TO 4:L$(Y)=A$( INT ( RND
    (8)*26)): PRINT K$(Y);L$(Y)
    :F(Y)=0: NEXT Y
440 I=0:C=0
450 GET Q$: IF Q$="" THEN 450
460 FOR Y=1 TO 4: IF Q$=L$(Y) AND
    F(Y)=0 THEN S=S+10:C=C+1:F(Y)=1
470 PRINT "[HOME]"; TAB( 19);S
    : IF C=4 THEN I=T:Y=4
480 NEXT Y
490 I=I+1: IF I<T THEN 450
500 X=X+1: IF X<6 THEN 430
510 PRINT "[CLR]"
520 PRINT TAB( 15)"[RVSON]GAME OVER."
    : PRINT : PRINT
530 PRINT TAB( 16)"[RVSON]YOU WIN."
540 PRINT : PRINT
550 PRINT TAB( 13)"[RVSON]FINAL
    SCORE:[RVSOFF]"S

```



Translator

This program is an educational quiz. As is, the program is set up to act as a Spanish-language translator. The user picks the number of questions to be asked. An English word is shown and the user is asked to type in the Spanish equivalent.

Correct and incorrect answers are indicated. When the answer is incorrect, the correct word is displayed. When all questions have been asked, the computer displays the user's score.

```
100 REM  TRANSLATOR
110 DIM  E$(10),O$(10)
120 FOR X=1 TO 10: READ E$(X),O$(X)
      : NEXT X
130 PRINT "[CLR]"
140 PRINT "WELCOME TO THE CYBERNETIC
      SCHOOL OF[SPACE 5]COMPUTERIZED
      LANGUAGE."
150 PRINT : PRINT "I AM AMANDA,
      A SPELLING AND TRANSLATOR[SPACE 2]
      PROGRAM."
160 PRINT : PRINT "I WILL ASSIST
      YOU IN LEARNING SPANISH[SPACE 3]
      TODAY.": PRINT
170 PRINT : INPUT "WHAT IS YOUR
      NAME";N$: PRINT "[CLR]"
180 PRINT "ALL RIGHT ";N$;","; PRINT
190 PRINT "WE SHALL BEGIN BY
      DISCUSSING THE RULES.": PRINT
200 PRINT "1) WHENEVER YOU ARE ASKED
      FOR AN ANSWER[SPACE 4]TYPE
      ONE IN.": PRINT
210 PRINT "2) WHENEVER YOU TYPE
      A RESPONSE, YOU"
220 PRINT "[SPACE 3]MUST ENTER IT
      BY PRESSING [RVSON]RETURN
      [RVSOFF].": PRINT
230 PRINT "3) I WILL TELL YOU TO
      GIVE THE CORRECT[SPACE 5]SPANISH
      FOR A WORD. TYPE"
240 PRINT "[SPACE 3]IN YOUR ANSWER
      AND PRESS [RVSON]RETURN[RVSOFF]
      .": PRINT
250 PRINT "[SPACE 3]WHEN THE GAME
      IS OVER I WILL TELL"
260 PRINT "[SPACE 3]HOW WELL YOU
      DID OVERALL.": PRINT : PRINT
270 INPUT "HOW MANY QUESTIONS SHALL
      I ASK YOU";Q
280 S=0: FOR C=1 TO Q: PRINT "[CLR]"
      : PRINT "QUESTION #"C
290 X= INT ( RND (8)*10)+1
      : PRINT "GIVE THE CORRECT SPANISH
      FOR: "E$(X)
300 PRINT : INPUT A$
```


```

310 IF A$=0$(X) THEN S=S+1
    : PRINT "CORRECT": GOTO 330
320 PRINT "INCORRECT. THEN CORRECT
    TRANSLATION IS:[SPACE 2]"0$(X)
330 PRINT : PRINT "PRESS [RVSON]
    RETURN[RVSOFF] FOR NEXT
    QUESTION."
340 GET Q$: IF Q$="" THEN 340
350 NEXT C
360 PRINT "[CLR]"N$","
    : PRINT " OUT OF"Q"QUESTIONS
    YOU HAD"S"RIGHT:"
370 PRINT INT (S*100/Q+.5)"PERCENT."
    : PRINT
380 INPUT "WOULD YOU LIKE TO PLAY
    AGAIN";Q$
390 IF LEFT$(Q$,1)="Y" THEN PRINT "
    [CLR]": GOTO 270
400 PRINT : PRINT "THANKS FOR
    PLAYING!"
410 DATA TO DO,HACER,TO TALK,HABLAR,
    TO PUT,PONER,TO EAT,COMER
420 DATA DOOR,PUERTA,MOUTH,BOCA,COW,
    VACA,FORK,TENEDOR,SPOON,CUCHARA,
    PEACE,PAZ

```

Alteration—Enlarge the computer's vocabulary by adding more **DATA** statements after line 420. Each English word must be followed by a comma and its foreign translation. You will also need to change lines 110 and 120 to accommodate the larger list. For example, if you have 50 words, change the **DIM** statements in line 110 to **DIM E\$(50), O\$(50)**. Change the maximum value of **X** in line 120 to that same number of words: **FOR X=1 TO 50**. It is possible to add several thousand words to a Commodore 64.

Alteration—Change the Spanish to any other language using the Roman alphabet. Change the vocabulary in lines 410-420.



UNDERSTANDING PROGRAM PRINTOUTS

The program lines in this book are actual computer printouts of the working games. To make the lines easier for you to copy, a special notation system is used for graphics characters, color and cursor keys.

An underlined, upper-case letter means that you should hold down the SHIFT key while entering that letter. For example, if you see Z, shift while entering Z. A diamond will appear on the screen. An underlined, lower-case letter indicates that you should hold down the Commodore key (⌘) while entering that letter.

Some keys (+, -, £, @, *) will be underlined twice if the Commodore key (⌘) is to be held down; underlined once if the SHIFT key is to be held down.

When you see a word in brackets, it indicates using special control keys. For example, [DOWN] means press the cursor-up/down key. [DOWN 7] means press the cursor-up/down key seven times. The only exception is with spaces. A single space looks that way in the program line. If two or more spaces are required, an expression will be in brackets, such as [SPACE 2].

The table on the next page lists some more key combinations. For example, if the program shows [WHT] press the CTRL and WHT keys at the same time.

KEY-COMBINATION TABLE

Listing	Press Key(s)
[UP]	SHIFT-CURSOR UP/DOWN
[DOWN]	CURSOR UP/DOWN
[LEFT]	SHIFT-CURSOR LEFT/RIGHT
[RIGHT]	CURSOR LEFT/RIGHT
[HOME]	CLR HOME
[CLR]	SHIFT-CLR HOME
[RVSON]	CTRL-RVS ON
[RVSOFF]	CTRL-RVS OFF
[BLK]	CTRL-BLK
[WHT]	CTRL-WHT
[RED]	CTRL-RED
[CYN]	CTRL-CYN
[PUR]	CTRL-PUR
[GRN]	CTRL-GRN
[BLU]	CTRL-BLU
[YEL]	CTRL-YEL
[ORN]	⌘ 1
[BRN]	⌘ 2
[LT. RED]	⌘ 3
[DARK GRAY]	⌘ 4
[MED. GRAY]	⌘ 5
[LT. GRN]	⌘ 6
[LT. BLU]	⌘ 7
[LT. GRAY]	⌘ 8
[F-1]	f1
[F-2]	f2
[F-3]	f3
[F-4]	f4
[F-5]	f5
[F-6]	f6
[F-7]	f7
[F-8]	f8
[POUND]	£

KEY-COMBINATION TABLE

Function	Key Combination
Find	F7
Shift-Cursor Up/Down	↑ / ↓
Cursor Up/Down	↑ / ↓
Shift-Cursor Left/Right	← / →
Cursor Left/Right	← / →
Home	F5
End	F6
Page Up	F7
Page Down	F8
Print Screen	F9
Insert	F10
Delete	F11
Help	F12
Windows	Windows Key
Alt	Alt Key
Ctrl	Ctrl Key
Shift	Shift Key
Tab	Tab Key
Esc	Esc Key
Space	Space Key
Enter	Enter Key
Backspace	Backspace Key
Num Lock	Num Lock Key
Num Row	Num Row Keys
Arrow Keys	Arrow Keys
Function Keys	F1-F12
Modifier Keys	Ctrl, Alt, Shift, Windows
Special Keys	Esc, Tab, Caps Lock, Num Lock
Standard Keys	QWERTY, AZWSX, YUIOP, MNBVC, ZXCVB

1000000000


7.321654881

Great Games You Can Program!

Create 35 amazing games for your Commodore 64. Simply type the complete program listing into your computer and start playing. Games include color, graphics, sound, instructions to players, and scoring.

Games are written in Commodore 64 BASIC. While you're having fun, you'll learn more about programming. Each game features suggestions for programming variations—change the games to do exactly what you want!

Choose from Arcade, Board and Educational games. Your Commodore 64 has never been so much fun!



Moon Lander



Spatial Concepts



Alien

\$9.95 US
\$12.95 Canada

ISBN 0-89586-316-2
BIPAD 68349

